Matemática discreta en Haskell

María Dolores Valverde Rodríguez

Grupo de Lógica Computacional
Dpto. de Ciencias de la Computación e Inteligencia Artificial
Universidad de Sevilla

Sevilla, 21 de junio de 2016 (Versión de 20 de junio de 2017)

Índice general

Agradecimientos **7**

Abstract **9**

Introducción **11**

1 Conjuntos **13**

1.1 El TAD de los conjuntos 14

1.2 Representaciones de conjuntos 15

1.2.1 Conjuntos como listas ordenadas sin repetición 15

1.2.2 Definición de conjunto 17

1.2.3 Subconjuntos 18

1.2.4 Igualdad de conjuntos 19

1.2.5 Subconjuntos propios 19

1.2.6 Complementario de un conjunto 20

1.2.7 Cardinal de un conjunto 20

1.2.8 Conjunto unitario 21

1.2.9 Unión de conjuntos 21

1.2.10 Intersección de conjuntos 22

1.2.11 Producto cartesiano 23

1.2.12 Combinaciones 23

1.2.13 Variaciones con repetición 24

1.2.14 Conjuntos como listas sin repetición 24

1.2.15 Definición de conjunto 26

1.2.16 Subconjuntos 27

1.2.17 Igualdad de conjuntos 28

1.2.18 Subconjuntos propios 28

1.2.19 Complementario de un conjunto 29

1.2.20 Cardinal de un conjunto 29

1.2.21 Conjunto unitario 30

1.2.22 Unión de conjuntos 30

1.2.23 Intersección de conjuntos . . . 31
1.2.24 Producto cartesiano . . . 32
1.2.25 Combinaciones . . . 32
1.2.26 Variaciones con repetición . . . 33
1.3 Elección de la representación de conjuntos . . . 33

2 Relaciones y funciones **35**
2.1 Relaciones . . . 35
2.1.1 Relación binaria . . . 35
2.1.2 Imagen por una relación . . . 35
2.1.3 Dominio de una relación . . . 36
2.1.4 Rango de una relación . . . 36
2.1.5 Antiimagen por una relación . . . 36
2.1.6 Relación funcional . . . 37
2.2 Relaciones homogéneas . . . 37
2.2.1 Relaciones reflexivas . . . 38
2.2.2 Relaciones simétricas . . . 38
2.2.3 Relaciones antisimétricas . . . 39
2.2.4 Relaciones transitivas . . . 39
2.2.5 Relaciones de equivalencia . . . 40
2.2.6 Relaciones de orden . . . 40
2.2.7 Clases de equivalencia . . . 41
2.3 Funciones . . . 41
2.3.1 Imagen por una función . . . 42
2.3.2 Funciones inyectivas . . . 43
2.3.3 Funciones sobreyectivas . . . 43
2.3.4 Funciones biyectivas . . . 44
2.3.5 Inversa de una función . . . 45

3 Introducción a la teoría de grafos **47**
3.1 Definición de grafo . . . 48
3.2 El TAD de los grafos . . . 48
3.2.1 Grafos como listas de aristas . . . 49
3.3 Generador de grafos . . . 52
3.4 Ejemplos de grafos . . . 53
3.4.1 Grafo nulo . . . 53
3.4.2 Grafo ciclo . . . 54
3.4.3 Grafo de la amistad . . . 54

3.4.4 Grafo completo . . . 55
3.4.5 Grafo bipartito . . . 56
3.4.6 Grafo estrella . . . 58
3.4.7 Grafo rueda . . . 58
3.4.8 Grafo circulante . . . 59
3.4.9 Grafo de Petersen generalizado . . . 60
3.4.10 Otros grafos importantes . . . 61
3.5 Definiciones y propiedades . . . 65
3.5.1 Definiciones de grafos . . . 66
3.5.2 Propiedades de grafos . . . 71
3.5.3 Operaciones y propiedades sobre grafos . . . 71
3.6 Morfismos de grafos . . . 76
3.6.1 Morfismos . . . 76
3.6.2 Complejidad del problema de homomorfismo de grafos . . . 78
3.6.3 Isomorfismos . . . 79
3.6.4 Automorfismos . . . 84
3.7 Caminos en grafos . . . 85
3.7.1 Definición de camino . . . 85
3.7.2 Recorridos . . . 87
3.7.3 Caminos simples . . . 87
3.7.4 Conexión . . . 91
3.7.5 Distancia . . . 91
3.7.6 Caminos cerrados . . . 92
3.7.7 Circuitos . . . 93
3.7.8 Ciclos . . . 93
3.7.9 Grafos acíclicos . . . 94
3.8 Conectividad de los grafos . . . 95
3.8.1 Estar conectados por un camino . . . 95
3.8.2 Componentes conexas de un grafo . . . 95
3.8.3 Grafos conexos . . . 98
3.8.4 Excentricidad . . . 100
3.8.5 Diámetro . . . 100
3.8.6 Radio . . . 101
3.8.7 Centro . . . 101
3.8.8 Grosor . . . 102
3.8.9 Propiedades e invariantes por isomorfismos . . . 104

4 Matrices asociadas a grafos **107**

4.1 Generador de grafos simples 107

4.2 Matrices de adyacencia 108

4.2.1 Definición y propiedades 108

4.2.2 Propiedades básicas de las matrices 109

4.2.3 Propiedades de las matrices de adyacencia 109

4.2.4 Caminos y arcos 111

5 Apéndices **115**

5.1 Sistemas utilizados 115

5.2 Mapa de decisiones de diseño en conjuntos 117

5.3 Mapa de decisiones de diseño en grafos 117

Bibliografía **119**

Indice de definiciones **119**

Lista de tareas pendientes **123**

Agradecimientos

Quisiera agradecer a todas las personas que me han apoyado y prestado sus conocimientos a lo largo de estos cuatro años del grado. Gracias a ellos, el esfuerzo ha merecido la pena.

En primer lugar, me gustaría destacar a María José Hidalgo Doblado y José Antonio Alonso Jiménez, mi tutores en este Trabajo de Fin de Grado. Les agradezco de todo corazón la constancia y la atención que me han brindado a pesar de las dificultades que supone realizar el proyecto durante un curso en el que estoy disfrutando la oportunidad de la movilidad internacional Erasmus. Son para mí un referente y un claro ejemplo de lo que se puede conseguir con esfuerzo y trabajo en equipo.

Seguidamente, expresar mi gratitud al resto de profesores y trabajadores de la Universidad de Sevilla, especialmente a aquellos que se esfuerzan por mejorar la institución día a día.

Quiero agradecer también este trabajo a mi familia, en especial a mis padres, que me apoyaron desde el principio cuando tomé la decisión de cambiar de carrera y siempre han intentado brindarnos a mi hermana y a mí con lo mejor.

Finalmente a mis amigos y compañeros, que han hecho de esta primera etapa universitaria un capítulo de mi vida lleno de bonitos recuerdos. En especial a Pedro, con quien he tenido la suerte de compartir todo.

Muchas gracias a todos.

Abstract

Discrete mathematics is characterized as the branch of mathematics dealing with finite and numerable sets. Concepts and notations from discrete mathematics are useful in studying and describing objects and real-life problems. In particular the graph theory has numerous applications in logistics.

Throughout this project some of the knowlegde adquired from the course "Matemática Discreta" will be given a computational implementation. The code will be written in Haskell language and a free version of it will be available in GitHub under the name: *MDenHaskell*. The work will focus on the graph theory and will provide some examples and algorithms in order to give an introduction of it and how it can be implemented in Haskell.

At first, two chapters will be presented as a gentle reminder of basic concepts related to the set theory and the relations that can be established among them. The third chapter will introduce the main topic, graph theory, with different representations, definitions and examples on graphs. It will go through aspects such as morphism, connectivity and paths in graphs. Finally some properties and advantages of working with adjacency matrices will be presented in the fourth chapter.

This project leaves the door open for the community of programmers to continue and improve it. It can be used as a self-learning tool as well as to make calculations that by hand would be tedious.

Introducción

El objetivo del trabajo es la implementación de algoritmos de matemática discreta en Haskell. Los puntos de partida son

- los temas de la asignatura "Matemática discreta" ([4]),
- los temas de la asignatura "Informática" ([1]),
- el capítulo 7 del libro "Algorithms: A functional programming approach" ([8]) y
- el artículo "Graph theory" ([11]) de la Wikipedia.

Haskell es un lenguaje de programación funcional desarrollado en los últimos años por la comunidad de programadores con la intención de usarlo como instrumento para la enseñanza de programación funcional. La motivación de este desarrollo es hacer el análisis y diseño de programas más simple y permitir que los algoritmos sean fácilmente adaptables a otros lenguajes de programación funcionales.

En comparación con otros lenguajes de programación imperativos, la sintaxis de Haskell permite definir funciones de forma más clara y compacta. En Haskell las funciones se consideran valores, al mismo nivel que los tipos enteros o cadenas en cualquier lenguaje. Por ello, al igual que es habitual que en todos los lenguajes una función reciba datos de entrada (de tipo entero, flotante, cadena, etc) y devuelva datos (de los mismos tipos), en los lenguajes funcionales una función puede recibir como dato de entrada una función y devolver otra función como salida, que puede ser construida a partir de sus entradas y por operaciones entre funciones, como la composición. Esta capacidad nos proporciona métodos más potentes para construir y combinar los diversos módulos de los que se compone un programa. Por ejemplo, emulando la forma de operar sobre funciones que habitualmente se usa en matemáticas.

La matemática discreta consiste en el estudio de las propiedades de los conjuntos finitos o infinitos numerables, lo que hace posible su directa implementación computacional. En particular, la asignatura "Matemática Discreta" del grado se centra en estudiar propiedades y algoritmos de la combinatoria y la teoría de grafos. A lo largo del trabajo implementaré en Haskell definiciones y algoritmos con los que ya he tenido una primera aproximación teórica durante el grado.

Los dos primeros capítulos del trabajo sientan la base que me permitirá en los dos siguientes aproximarme a la teoría de grafos. En el primero, hago una introducción a la teoría de conjuntos y doy dos posibles representaciones de conjuntos con las que

trabajar en Haskell. En el segundo presento los conceptos de "relación" y de "función" que, al igual que los conjuntos, son necesarios para definir aplicaciones de los grafos. Para su redacción me he apoyado en la primera parte de la asignatura "Álgebra Básica" del grado.

En el tercer capítulo del trabajo hago una introducción a la teoría de grafos. En primer lugar doy una representación con la que trabajar en Haskell y un generador de grafos que nos servirá para comprobar propiedades con `QuickCheck`. Seguidamente hago una galería de grafos conocidos y los utilizo como ejemplos para las definiciones y los algoritmos que doy en las secciones siguientes. Las primeras serán algunas deficiones básicas y después trato los conceptos de "morfismos", "caminos" y "conectividad" en grafos.

Por último, en el cuarto capítulo doy una representación de los grafos a través de la matriz de adyacencia. Implemento algunos de los resultados que Javier Franco Galvín expuso en su Trabajo Fin de Grado: *Aspectos algebraicos en teoría de grafos* y compruebo que concuerdan con el desarrollo hecho en el capítulo anterior.

Me gustaría destacar el trabajo con los diferentes sistemas que he realizado a lo largo del proyecto y con diferentes programas:

- las librerías de Haskell `Data.List`, `Data.Matrix` y `Data.QuickCheck`,
- el paquete *Tikz* de LaTeX, mediante el que he dado la representación gráfica de los ejemplos de grafos,
- *GitHub* como sistema de control de versiones,
- y *Hspec* como sistema de validación de módulos.

Capítulo 1

Conjuntos

El concepto de *conjunto* aparece en todos los campos de las Matemáticas, pero, ¿qué debe entenderse por él? La *Teoría de conjuntos* fue introducida por Georg Cantor (1845-1917); desde 1869, Cantor ejerció como profesor en la Universidad de Halle y entre 1879 y 1884 publicó una serie de seis artículos en el *Mathematische Annalen*, en los que hizo una introducción básica a la teoría de conjuntos. En su *Beiträge zur Begründung der transfiniten Mengenlehre*, Cantor dio la siguiente definición de conjunto:

§ 1

The Conception of Power or Cardinal Number

By an "aggregate" (*Menge*) we are to understand any collection into a whole (*Zusammenfassung zu einem Ganzen*) M of definite and separate objects *m* of our intuition or our thought. These objects are called the "elements" of M.

Figura 1.1: Fragmento del texto traducido al inglés en el que Cantor da la definición de conjunto

> «Debemos entender por "conjunto" (*Menge*) cualquier colección vista como un todo (Zusammenfassung zu einem Ganzen), *M*, de objetos separados y bien definidos, *m*, de nuestra intuición o pensamiento. Estos objetos son los "elementos" de *M*»

Felix Hausdorff, en 1914, dice: «un conjunto es una reunión de cosas que constituyen una totalidad; es decir, una nueva cosa», y añade: «esto puede difícilmente ser una definición, pero sirve como demostración expresiva del concepto de conjunto a través de conjuntos sencillos como el conjunto de habitantes de una ciudad o el de átomos de Hidrógeno del Sol».

Un conjunto así definido no tiene que estar compuesto necesariamente de elementos homogéneos y además, da lugar a cuestiones filosóficas como si podemos llamar

conjunto a aquel que no posee ningún elemento. Matemáticamente, conviene aceptar solo elementos que compartan alguna propiedad y definir el *conjunto vacío* como aquel que no tiene elemento alguno.
El gran mérito de Cantor fue considerar conjuntos *transfinitos* (que tiene infinitos elementos), concepto inaudito hasta avanzado el siglo XIX, hablar del *cardinal* de un conjunto como el número de sus elementos y hablar de *conjuntos equivalentes* cuando puede establecerse una biyección entre ellos; ideas ya apuntadas por Bolzano, quien se centró demasiado en el aspecto filosófico, sin llegar a formalizar sus ideas.

A lo largo de la sección, haremos una pequeña introducción a la Teoría de Conjuntos, presentando formalmente sus conceptos más importantes. A la hora de elaborar el contenido se han utilizado los siguientes recursos bibliográficos:
- el primer tema de la asignatura "Álgebra básica" ([6]),
- los temas de la asignatura "Informática" ([1]),
- el artículo de la Wikipedia "Set (mathematics)" ([10] y
* el artículo "El regalo de Cantor" ([5]).

1.1. El TAD de los conjuntos

En la presente sección, se definen las operaciones básicas necesarias para trabajar con conjuntos en un lenguaje funcional. En nuestro caso, el lenguaje que utilizaremos será Haskell. Daremos la signatura del Tipo Abstracto de Dato (TAD) de los conjuntos y daremos algunos ejemplos de posibles representaciones de conjuntos con las que podríamos trabajar.
A continuación, presentamos las operaciones definidas en el TAD de los conjuntos:

```
vacio          :: Conj a
inserta        :: Eq a => a -> Conj a -> Conj a
elimina        :: Eq a => a -> Conj a -> Conj a
pertenece      :: Eq a => Conj a -> a -> Bool
esVacio        :: Conj a -> Bool
minimoElemento :: Ord a => Conj a -> a
```

donde,
- `(vacio)` es el conjunto vacío.
- `(inserta x c)` es el conjunto obtenido añadiendo el elemento `x` al conjunto `c`.
- `(elimina x c)` es el conjunto obtenido eliminando el elemento `x` del conjunto `c`.
- `(pertenece x c)` se verifica si `x` pertenece al conjunto `c`.
- `(esVacio c)` se verifica si `c` no tiene ningún elemento.
- `(minimoElemento c)` devuelve el mínimo elemento del conjunto `c`.

Hemos de tener en cuenta que a la hora de crear un nuevo tipo de dato con el que representar a los conjuntos, este debe ser compatible con la entidad matemática que representa.

Estas son algunas de las posibles representaciones con las que podríamos trabajar con conjuntos en Haskell:

- En primer lugar, podemos trabajar con la librería `Data.Set`; en este caso, la implementación del tipo de dato `Set` está basado en árboles binarios balanceados.
- Por otra parte, podemos definir un nuevo tipo `Cj xs` con el que trabajar con conjuntos como listas no ordenadas con duplicados, como listas no ordenadas sin duplicados o como listas ordenadas sin duplicados.
- Otra opción sería trabajar directamente con conjuntos como listas; para ello, debemos ignorar las repeticiones y el orden con vista a la igualdad de conjuntos.
- Los conjuntos que sólo contienen números (de tipo `Int`) entre 0 y $n-1$, se pueden representar como números binarios con n bits donde el bit $i (0 \leq i < n)$ es 1 syss el número i pertenece al conjunto.

Nota 1.1.1. Las funciones que aparecen en la especificación del TAD no dependen de la representación que elijamos.

1.2. Representaciones de conjuntos

1.2.1. Conjuntos como listas ordenadas sin repetición

En el módulo `ConjuntosConListasOrdenadasSinRepeticion` se definen las funciones del TAD de los conjuntos dando su representación como listas ordenadas sin repetición.

```
module ConjuntosConListasOrdenadasSinRepeticion
  ( Conj
  , vacio
  , inserta
  , listaAConjunto
  , elimina
  , pertenece
  , esVacio
  , minimoElemento
  , sinRepetidos
  , esUnitario
  , esSubconjunto
  , conjuntosIguales
  , esSubconjuntoPropio
  , complementario
```

```
  , cardinal
  , unionConjuntos
  , unionGeneral
  , interseccion
  , productoCartesiano
  , combinaciones
  , variacionesR
  ) where
```

En las definiciones del presente módulo se usarán algunas funciones de la librería `Data.List`

Vamos a definir un nuevo tipo de dato `(Conj a)`, que representa a los conjuntos como listas ordenadas sin repetición.

```
type Conj a = [a]
```

Las funciones básicas que definiremos a partir de este tipo coincidirán con las indicadas en el TAD de los conjuntos.

- `(vacio)` es el conjunto vacío.

```
-- | Ejemplo
-- >>> vacio
-- []
vacio :: Conj a
vacio = []
```

- `(inserta x c)` es el conjunto obtenido añadiendo el elemento `x` al conjunto `c`.

```
-- | Ejemplo
-- >>> inserta 5 vacio
-- [5]
-- >>> foldr inserta vacio [2,2,1,1,2,4,2]
-- [1,2,4]
inserta :: Ord a => a -> Conj a -> Conj a
inserta x [] = [x]
inserta x ys
  | pertenece ys x = ys
  | otherwise      = insert x ys
```

- `(listaAConjunto xs)` devuelve el conjunto cuyos elementos son los de la lista `xs`.

```
-- | Ejemplo
-- >>> listaAConjunto [2,2,1,1,2,4,2]
-- [1,2,4]
listaAConjunto :: Ord a => [a] -> Conj a
listaAConjunto = sort . nub
```

- `(esVacio c)` se verifica si `c` es el conjunto vacío.

```
-- | Ejemplos
-- >>> esVacio (listaAConjunto [2,5,1,3,7,5,3,2,1,9,0])
-- False
-- >>> esVacio vacio
-- True
esVacio :: Conj a -> Bool
esVacio = null
```

- `(pertenece x c)` se verifica si `x` es un elemento del conjunto `c`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [2,5,1,3,7,5,3,2,1,9,0]
-- >>> pertenece c1 3
-- True
-- >>> pertenece c1 4
-- False
pertenece :: Ord a => Conj a -> a -> Bool
pertenece ys x =
  x == head (dropWhile (<x) ys)
```

- `(elimina x c)` es el conjunto obtenido eliminando el elemento `x` del conjunto `c`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [2,5,1,3,7,5,3,2,1,9,0]
-- >>> elimina 3 c1
-- [0,1,2,5,7,9]
-- >>> elimina 4 c1
-- [0,1,2,3,5,7,9]
elimina :: Ord a => a -> Conj a -> Conj a
elimina x ys = us ++ dropWhile (==x) vs
  where (us,vs) = span (<x) ys
```

- `(minimoElemento c)` devuelve el mínimo elemento del conjunto `c`.

```
-- | Ejemplos
-- >>> minimoElemento (listaAConjunto [2,5,1,3,7,5,3,2,1,9,0])
-- 0
-- >>> minimoElemento (listaAConjunto (['a'..'e'] ++ ['A'..'E']))
-- 'A'
minimoElemento :: Ord a => Conj a -> a
minimoElemento = head
```

1.2.2. Definición de conjunto

Nota 1.2.1. Al trabajar con la representación de conjuntos como listas en Haskell, hemos de cuidar que los ejemplos con los que trabajemos no tengan elementos repetidos. La función `(sinRepetidos xs)` se verifica si la lista `xs` no tiene ningún elemento repetido.

```
-- | Ejemplos
-- >>> sinRepetidos []
-- True
-- >>> sinRepetidos [1,2,3,1]
-- False
sinRepetidos :: Eq a => [a] -> Bool
sinRepetidos xs = nub xs == xs
```

Los conjuntos pueden definirse de manera explícita, citando todos sus elementos entre llaves, de manera implícita, dando una o varias características que determinen si un objeto dado está o no en el conjunto. Por ejemplo, los conjuntos $\{1,2,3,4\}$ y $\{x \in \mathbb{N} | 1 \leq x \leq 4\}$ son el mismo, definido de forma explicita e implícita respectivamente.

Nota 1.2.2. La definición implícita es necesaria cuando el conjunto en cuestión tiene una cantidad infinita de elementos. En general, los conjuntos se notarán con letras mayúsculas: $A, B, \ldots$ y los elementos con letras minúsculas: $a, b, \ldots$.

Cuando trabajamos con conjuntos concretos, siempre existe un contexto donde esos conjuntos existen. Por ejemplo, si $A = \{-1,1,2,3,4,5\}$ y $B = \{x | x \in \mathbb{N} \text{es par}\}$ el contexto donde podemos considerar A y B es el conjunto de los números enteros, $\mathbb{Z}$. En general, a este conjunto se le denomina *conjunto universal*. De una forma algo más precisa, podemos dar la siguiente definición:

Definición 1.2.1. *El **conjunto universal**, que notaremos por U, es un conjunto del que son subconjuntos todos los posibles conjuntos que originan el problema que tratamos.*

1.2.3. Subconjuntos

Definición 1.2.2. *Dados dos conjuntos A y B, si todo elemento de A es a su vez elemento de B diremos que A es un subconjunto de B y lo notaremos $A \subseteq B$. En caso contrario se notará $A \nsubseteq B$.*

La función `(esSubconjunto c1 c2)` se verifica si `c1` es un subconjunto de `c2`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [4,2]
-- >>> let c2 = listaAConjunto [3,2,4]
-- >>> let c3 = listaAConjunto [4,2,1]
-- >>> let c4 = listaAConjunto [1,2,4]
-- >>> c1 `esSubconjunto` c2
-- True
-- >>> c1 `esSubconjunto` vacio
-- False
-- >>> vacio `esSubconjunto` c2
```

```
-- True
-- >>> c3 `esSubconjunto` c4
-- True
-- >>> c2 `esSubconjunto` c1
-- False
esSubconjunto :: Ord a => Conj a -> Conj a -> Bool
esSubconjunto [] _ = True
esSubconjunto (x:xs) ys =
  x == head vs && esSubconjunto xs (tail vs)
  where (us,vs) = span (<x) ys
```

1.2.4. Igualdad de conjuntos

Definición 1.2.3. *Dados dos conjuntos A y B, diremos que son **iguales** si tienen los mismos elementos; es decir, si se verifica que $A \subseteq B$ y $B \subseteq A$. Lo notaremos $A = B$.*

La función `(conjuntosIguales c1 c2)` se verifica si los conjuntos `xs` y `ys` son iguales.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [4,2]
-- >>> let c2 = listaAConjunto [3,2,4]
-- >>> let c3 = listaAConjunto [4,2,1]
-- >>> let c4 = listaAConjunto [1,2,4]
-- >>> let c5 = listaAConjunto [4,4,4,4,4,4,2]
-- >>> conjuntosIguales c1 c2
-- False
-- >>> conjuntosIguales c3 c4
-- True
-- >>> conjuntosIguales c1 c5
-- True
conjuntosIguales :: Ord a => Conj a -> Conj a -> Bool
conjuntosIguales = (==)
```

1.2.5. Subconjuntos propios

Definición 1.2.4. *Los subconjuntos de A distintos del $\emptyset$ y del mismo A se denominan **subconjuntos propios** de A.*

La función `(esSubconjuntoPropio c1 c2)` se verifica si `c1` es un subconjunto propio de `c2`.

```
-- | Ejemplos
-- >>> let u  = listaAConjunto [1..9]
-- >>> let c1 = listaAConjunto [3,2,5,7]
-- >>> esSubconjuntoPropio u u
```

```
-- False
-- >>> esSubconjuntoPropio c1 u
-- True
esSubconjuntoPropio :: Ord a => Conj a -> Conj a -> Bool
esSubconjuntoPropio c1 c2 =
  not (null c1) && esSubconjunto c1 c2 && c1 /= c2
```

1.2.6. Complementario de un conjunto

Definición 1.2.5. *Dado un conjunto A, se define el **complementario** de A, que notaremos por $\overline{A}$ como:*

$$\overline{A} = \{x | x \in U, x \notin A\}$$

La función `(complementario u c)` devuelve el complementario del conjunto `c` y en el universo `u`.

```
-- | Ejemplos
-- >>> let u  = listaAConjunto [1..9]
-- >>> let c1 = listaAConjunto [3,2,5,7]
-- >>> let c2 = listaAConjunto [1,4,6,8,9]
-- >>> complementario u c1
-- [1,4,6,8,9]
-- >>> complementario u u
-- []
-- >>> complementario u vacio
-- [1,2,3,4,5,6,7,8,9]
-- >>> complementario u c2
-- [2,3,5,7]
complementario :: Ord a => Conj a -> Conj a -> Conj a
complementario [] _  = []
complementario xs [] = xs
complementario (x:xs) (y:ys)
  | x < y     = x : complementario xs (y:ys)
  | otherwise = complementario xs ys
```

1.2.7. Cardinal de un conjunto

Definición 1.2.6. *Dado un conjunto finito A, denominaremos **cardinal** de A al número de elementos que tiene y lo notaremos $|A|$.*

La función `(cardinal xs)` devuelve el cardinal del conjunto `xs`.

```
-- | Ejemplos
-- >>> cardinal vacio
-- 0
-- >>> cardinal (listaAConjunto [1..10])
```

```
-- 10
-- >>> cardinal (listaAConjunto "chocolate")
-- 7
cardinal :: Ord a => Conj a -> Int
cardinal = length
```

1.2.8. Conjunto unitario

Definición 1.2.7. *Un conjunto con un único elemento se denomina **unitario**.*

Nota 1.2.3. Notemos que, si $X = \{x\}$ es un conjunto unitario, debemos distinguir entre el conjunto X y el elemento x.

La función `(esUnitario c)` se verifica si el conjunto `c` es unitario.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto (take 10 (repeat 5))
-- >>> let c2 = listaAConjunto "valverde"
-- >>> let c3 = listaAConjunto "coco"
-- >>> esUnitario c1
-- True
-- >>> esUnitario c2
-- False
-- >>> esUnitario c2
-- False
esUnitario :: Ord a => Conj a -> Bool
esUnitario c =
  c == take 1 c
```

1.2.9. Unión de conjuntos

Definición 1.2.8. *Dados dos conjuntos A y B se define la **unión** de A y B, notado $A \cup B$, como el conjunto formado por aquellos elementos que pertenecen al menos a uno de los dos conjuntos, A ó B; es decir,*

$$A \cup B = \{x \mid x \in A \vee x \in B\}$$

La función `(unionConjuntos c1 c2)` devuelve la unión de los conjuntos `xs` y `ys`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [1,3..9]
-- >>> let c2 = listaAConjunto [2,4..9]
-- >>> unionConjuntos c1 c2
-- [1,2,3,4,5,6,7,8,9]
unionConjuntos :: Ord a => Conj a -> Conj a -> Conj a
unionConjuntos [] [] = []
```

```
unionConjuntos [] ys = ys
unionConjuntos xs [] = xs
unionConjuntos (x:xs) (y:ys)
  | x <  y    = x : unionConjuntos xs (y:ys)
  | x == y    = x : unionConjuntos xs ys
  | otherwise = y : unionConjuntos (x:xs) ys
```

Definición 1.2.9. *Dada una familia de conjuntos* $\{A\}_i$ *con* $i \in I$*, se define la* ***unión general*** *de los conjuntos* A_i *notado* $\bigcup_{i\in I} A_i$*, como el conjunto formado por aquellos elementos que pertenecen al menos a uno de los conjuntos de la familia; es decir,*

$$\bigcup_{i\in I} A_i = \{x \mid x \in A_i,\ \forall i \in I\}$$

La función `(unionGeneral xss)` devuelve la unión general de la familia de conjuntos de la lista `xss`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [1,4..15]
-- >>> let c2 = listaAConjunto [2,5..15]
-- >>> let c3 = listaAConjunto [3,6..15]
-- >>> unionGeneral [c1,c2,c3]
-- [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]
unionGeneral :: Ord a => [Conj a] -> Conj a
unionGeneral = foldr unionConjuntos vacio
```

1.2.10. Intersección de conjuntos

Definición 1.2.10. *Dados dos conjuntos A y B se define la* ***intersección*** *de A y B, notado* $A \cap B$*, como el conjunto formado por aquellos elementos que pertenecen a cada uno de los dos conjuntos, A y B, es decir,*

$$A \cap B = \{x \mid x \in A \wedge x \in B\}$$

La función `(interseccion c1 c2)` devuelve la intersección de los conjuntos `c1` y `c2`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [1,3..20]
-- >>> let c2 = listaAConjunto [2,4..20]
-- >>> let c3 = listaAConjunto [2,4..30]
-- >>> let c4 = listaAConjunto [4,8..30]
-- >>> interseccion c1 c2
-- []
-- >>> interseccion c3 c4
```

```
-- [4,8,12,16,20,24,28]
interseccion :: Ord a => Conj a -> Conj a -> Conj a
interseccion (x:xs) (y:ys)
  | x < y     = interseccion xs (y:ys)
  | x > y     = interseccion (x:xs) ys
  | otherwise = x : interseccion xs ys
interseccion _ _ = []
```

1.2.11. Producto cartesiano

Definición 1.2.11. *El **producto cartesiano** [1] de dos conjuntos A y B es una operación sobre ellos que resulta en un nuevo conjunto $A \times B$ que contiene a todos los pares ordenados tales que la primera componente pertenece a A y la segunda pertenece a B; es decir,*

$$A \times B = \{(a, b) \mid a \in A,\ b \in B\}$$

La función `(productoCartesiano c1 c2)` devuelve el producto cartesiano de xs e ys.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [3,1]
-- >>> let c2 = listaAConjunto [2,4,7]
-- >>> productoCartesiano c1 c2
-- [(1,2),(1,4),(1,7),(3,2),(3,4),(3,7)]
-- >>> productoCartesiano c2 c1
-- [(2,1),(2,3),(4,1),(4,3),(7,1),(7,3)]
productoCartesiano :: (Ord a, Ord b) => Conj a -> Conj b -> Conj (a,b)
productoCartesiano xs ys=
  listaAConjunto [(x,y) | x <- xs, y <- ys]
```

1.2.12. Combinaciones

Definición 1.2.12. *Las **combinaciones** de un conjunto S tomados en grupos de n son todos los subconjuntos de S con n elementos.*

La función `(combinaciones n xs)` devuelve las combinaciones de los elementos de `xs` en listas de `n` elementos.

```
-- | Ejemplos
-- >>> combinaciones 3 ['a'..'d']
-- ["abc","abd","acd","bcd"]
-- >>> combinaciones 2 [2,4..8]
-- [[2,4],[2,6],[2,8],[4,6],[4,8],[6,8]]
```

[1] https://en.wikipedia.org/wiki/Cartesian_product

```
combinaciones :: Integer -> [a] -> [[a]]
combinaciones 0 _          = [[]]
combinaciones _ []         = []
combinaciones k (x:xs) =
    [x:ys | ys <- combinaciones (k-1) xs] ++ combinaciones k xs
```

1.2.13. Variaciones con repetición

Definición 1.2.13. *Las **variaciones con repetición** de m elementos tomados en grupos de n es el número de diferentes n–tuplas de un conjunto de m elementos.*

La función `(variacionesR n xs)` devuelve las variaciones con con repetición de los elementos de `xs` en listas de `n` elementos.

```
-- | Ejemplos
-- >>> variacionesR 3 ['a','b']
-- ["aaa","aab","aba","abb","baa","bab","bba","bbb"]
-- >>> variacionesR 2 [2,3,5]
-- [[2,2],[2,3],[2,5],[3,2],[3,3],[3,5],[5,2],[5,3],[5,5]]
variacionesR :: Int -> [a] -> [[a]]
variacionesR _ [] = [[]]
variacionesR 0 _  = [[]]
variacionesR k us =
    [u:vs | u <- us, vs <- variacionesR (k-1) us]
```

1.2.14. Conjuntos como listas sin repetición

En el módulo `ConjuntosConListas` se definen las funciones del TAD de los conjuntos dando su representación como listas sin elementos repetidos.

```
module ConjuntosConListas
    ( Conj
  , vacio
  , inserta
  , listaAConjunto
  , elimina
  , pertenece
  , esVacio
  , minimoElemento
  , sinRepetidos
  , esUnitario
  , esSubconjunto
  , conjuntosIguales
  , esSubconjuntoPropio
  , complementario
  , cardinal
```

```
  , unionConjuntos
  , unionGeneral
  , interseccion
  , productoCartesiano
  , combinaciones
  , variacionesR
  ) where
```

En las definiciones del presente módulo se usarán algunas funciones de la librería `Data.List`

Vamos a definir un nuevo tipo de dato `(Conj a)`, que representa a los conjuntos como listas sin elementos repetidos.

```
type Conj a = [a]
```

Las funciones básicas que definiremos a partir de este tipo coincidirán con las indicadas en el TAD de los conjuntos.

- `(vacio)` es el conjunto vacío.

```
-- | Ejemplo
-- >>> vacio
-- []
vacio :: Conj a
vacio = []
```

- `(inserta x c)` es el conjunto obtenido añadiendo el elemento `x` al conjunto `c`.

```
-- | Ejemplo
-- >>> inserta 5 vacio
-- [5]
-- >>> foldr inserta vacio [2,2,1,1,2,4,2]
-- [1,4,2]
inserta :: Eq a => a -> Conj a -> Conj a
inserta x [] = [x]
inserta x ys | elem x ys = ys
             | otherwise = x:ys
```

- `(listaAConjunto xs)` devuelve el conjunto cuyos elementos son los de la lista `xs`.

```
-- | Ejemplo
-- >>> listaAConjunto [2,2,1,1,2,4,2]
-- [2,1,4]
listaAConjunto :: Eq a => [a] -> Conj a
listaAConjunto = nub
```

- `(esVacio c)` se verifica si `c` es el conjunto vacío.

```
-- | Ejemplos
-- >>> esVacio (listaAConjunto [2,5,1,3,7,5,3,2,1,9,0])
-- False
```

```
-- >>> esVacio vacio
-- True
esVacio :: Conj a -> Bool
esVacio = null
```

- `(pertenece x c)` se verifica si `x` es un elemento del conjunto `c`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [2,5,1,3,7,5,3,2,1,9,0]
-- >>> pertenece 3 c1
-- True
-- >>> pertenece 4 c1
-- False
pertenece :: Eq a => a -> Conj a -> Bool
pertenece = elem
```

- `(elimina x c)` es el conjunto obtenido eliminando el elemento `x` del conjunto `c`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [2,5,1,3,7,5,3,2,1,9,0]
-- >>> elimina 3 c1
-- [2,5,1,7,9,0]
-- >>> elimina 4 c1
-- [2,5,1,3,7,9,0]
elimina :: Eq a => a -> Conj a -> Conj a
elimina = delete
```

- `(minimoElemento c)` devuelve el mínimo elemento del conjunto `c`.

```
-- | Ejemplos
-- >>> minimoElemento (listaAConjunto [2,5,1,3,7,5,3,2,1,9,0])
-- 0
-- >>> minimoElemento (listaAConjunto (['a'..'e'] ++ ['A'..'E']))
-- 'A'
minimoElemento :: Ord a => Conj a -> a
minimoElemento = minimum
```

1.2.15. Definición de conjunto

Definición 1.2.14. *Llamaremos **conjunto** a una colección de objetos, que llamaremos **elementos**, distintos entre sí y que comparten una propiedad. Para que un conjunto esté bien definido debe ser posible discernir si un objeto arbitrario está o no en él.*

Si el elemento a pertenece al conjunto A, escribiremos $a \in A$. En caso contrario escribiremos $a \notin A$.

Nota 1.2.4. En Haskell, para poder discernir si un objeto arbitrario pertenece a un conjunto se necesita que su tipo pertenezca a la clase `Eq`.

Nota 1.2.5. Al trabajar con la representación de conjuntos como listas en Haskell, hemos de cuidar que los ejemplos con los que trabajemos no tengan elementos repetidos. La función `(sinRepetidos xs)` se verifica si la lista `xs` no tiene ningún elemento repetido.

```
-- | Ejemplos
-- >>> sinRepetidos []
-- True
-- >>> sinRepetidos [1,2,3,1]
-- False
sinRepetidos :: Eq a => [a] -> Bool
sinRepetidos xs = nub xs == xs
```

Los conjuntos pueden definirse de manera explícita, citando todos sus elementos entre llaves, de manera implícita, dando una o varias características que determinen si un objeto dado está o no en el conjunto. Por ejemplo, los conjuntos $\{1, 2, 3, 4\}$ y $\{x \in \mathbb{N} | 1 \leq x \leq 4\}$ son el mismo, definido de forma explicita e implícita respectivamente.

Nota 1.2.6. La definición implícita es necesaria cuando el conjunto en cuestión tiene una cantidad infinita de elementos. En general, los conjuntos se notarán con letras mayúsculas: $A, B, \ldots$ y los elementos con letras minúsculas: $a, b, \ldots$.

Cuando trabajamos con conjuntos concretos, siempre existe un contexto donde esos conjuntos existen. Por ejemplo, si $A = \{-1, 1, 2, 3, 4, 5\}$ y $B = \{x | x \in \mathbb{N} \text{es par}\}$ el contexto donde podemos considerar A y B es el conjunto de los números enteros, $\mathbb{Z}$. En general, a este conjunto se le denomina *conjunto universal*. De una forma algo más precisa, podemos dar la siguiente definición:

Definición 1.2.15. *El **conjunto universal**, que notaremos por U, es un conjunto del que son subconjuntos todos los posibles conjuntos que originan el problema que tratamos.*

1.2.16. Subconjuntos

Definición 1.2.16. *Dados dos conjuntos A y B, si todo elemento de A es a su vez elemento de B diremos que A es un subconjunto de B y lo notaremos $A \subseteq B$. En caso contrario se notará $A \nsubseteq B$.*

La función `(esSubconjunto c1 c2)` se verifica si `c1` es un subconjunto de `c2`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [4,2]
-- >>> let c2 = listaAConjunto [3,2,4]
-- >>> let c3 = listaAConjunto [4,2,1]
-- >>> let c4 = listaAConjunto [1,2,4]
```

```
-- >>> c1 `esSubconjunto` c2
-- True
-- >>> c1 `esSubconjunto` vacio
-- False
-- >>> vacio `esSubconjunto` c2
-- True
-- >>> c3 `esSubconjunto` c4
-- True
-- >>> c2 `esSubconjunto` c1
-- False
esSubconjunto :: Eq a => [a] -> [a] -> Bool
esSubconjunto c1 c2 = all (`elem` c2) c1
```

1.2.17. Igualdad de conjuntos

Definición 1.2.17. *Dados dos conjuntos A y B, diremos que son* ***iguales*** *si tienen los mismos elementos; es decir, si se verifica que $A \subseteq B$ y $B \subseteq A$. Lo notaremos $A = B$.*

La función `(conjuntosIguales c1 c2)` se verifica si los conjuntos `xs` y `ys` son iguales.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [4,2]
-- >>> let c2 = listaAConjunto [3,2,4]
-- >>> let c3 = listaAConjunto [4,2,1]
-- >>> let c4 = listaAConjunto [1,2,4]
-- >>> let c5 = listaAConjunto [4,4,4,4,4,4,2]
-- >>> conjuntosIguales c1 c2
-- False
-- >>> conjuntosIguales c3 c4
-- True
-- >>> conjuntosIguales c1 c5
-- True
conjuntosIguales :: Eq a => [a] -> [a] -> Bool
conjuntosIguales c1 c2 =
  esSubconjunto c1 c2 && esSubconjunto c2 c1
```

1.2.18. Subconjuntos propios

Definición 1.2.18. *Los subconjuntos de A distintos del $\emptyset$ y del mismo A se denominan* ***subconjuntos propios*** *de A.*

La función `(esSubconjuntoPropio c1 c2)` se verifica si `c1` es un subconjunto propio de `c2`.

```
-- | Ejemplos
-- >>> let u  = listaAConjunto [1..9]
-- >>> let c1 = listaAConjunto [3,2,5,7]
-- >>> esSubconjuntoPropio u u
-- False
-- >>> esSubconjuntoPropio c1 u
-- True
esSubconjuntoPropio :: Eq a => [a] -> [a] -> Bool
esSubconjuntoPropio c1 c2
  | null c1                = False
  | conjuntosIguales c1 c2 = False
  | otherwise              = esSubconjunto c1 c2
```

1.2.19. Complementario de un conjunto

Definición 1.2.19. *Dado un conjunto A, se define el **complementario** de A, que notaremos por $\overline{A}$ como:*

$$\overline{A} = \{x | x \in U, x \notin A\}$$

La función `(complementario u c)` devuelve el complementario del conjunto `c` y en el universo `u`.

```
-- | Ejemplos
-- >>> let u  = listaAConjunto [1..9]
-- >>> let c1 = listaAConjunto [3,2,5,7]
-- >>> let c2 = listaAConjunto [1,4,6,8,9]
-- >>> complementario u c1
-- [1,4,6,8,9]
-- >>> complementario u u
-- []
-- >>> complementario u vacio
-- [1,2,3,4,5,6,7,8,9]
-- >>> complementario u c2
-- [2,3,5,7]
complementario :: Eq a => [a] -> [a] -> [a]
complementario = (\\)
```

1.2.20. Cardinal de un conjunto

Definición 1.2.20. *Dado un conjunto finito A, denominaremos **cardinal** de A al número de elementos que tiene y lo notaremos $|A|$.*

La función `(cardinal xs)` devuelve el cardinal del conjunto `xs`.

```
-- | Ejemplos
-- >>> cardinal vacio
-- 0
-- >>> cardinal (listaAConjunto [1..10])
-- 10
-- >>> cardinal (listaAConjunto "chocolate")
-- 7
cardinal :: [a] -> Int
cardinal = length
```

1.2.21. Conjunto unitario

Definición 1.2.21. *Un conjunto con un único elemento se denomina **unitario**.*

Nota 1.2.7. Notemos que, si $X = \{x\}$ es un conjunto unitario, debemos distinguir entre el conjunto X y el elemento x.

La función `(esUnitario c)` se verifica si el conjunto `c` es unitario.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto (take 10 (repeat 5))
-- >>> let c2 = listaAConjunto "valverde"
-- >>> let c3 = listaAConjunto "coco"
-- >>> esUnitario c1
-- True
-- >>> esUnitario c2
-- False
-- >>> esUnitario c2
-- False
esUnitario :: Eq a => [a] -> Bool
esUnitario c = c == take 1 c
```

1.2.22. Unión de conjuntos

Definición 1.2.22. *Dados dos conjuntos A y B se define la **unión** de A y B, notado $A \cup B$, como el conjunto formado por aquellos elementos que pertenecen al menos a uno de los dos conjuntos, A ó B; es decir,*

$$A \cup B = \{x \mid x \in A \vee x \in B\}$$

La función `(unionConjuntos c1 c2)` devuelve la unión de los conjuntos `xs` y `ys`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [1,3..9]
-- >>> let c2 = listaAConjunto [2,4..9]
```

```
-- >>> unionConjuntos c1 c2
-- [1,3,5,7,9,2,4,6,8]
unionConjuntos :: Eq a => [a] -> [a] -> [a]
unionConjuntos = union
```

Definición 1.2.23. *Dada una familia de conjuntos* $\{A\}_i$ *con* $i \in I$*, se define la* ***unión general*** *de los conjuntos* A_i *notado* $\bigcup_{i \in I} A_i$*, como el conjunto formado por aquellos elementos que pertenecen al menos a uno de los conjuntos de la familia; es decir,*

$$\bigcup_{i \in I} A_i = \{x \mid x \in A_i,\ \forall i \in I\}$$

La función `(unionGeneral xss)` devuelve la unión general de la familia de conjuntos de la lista `xss`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [1,4..15]
-- >>> let c2 = listaAConjunto [2,5..15]
-- >>> let c3 = listaAConjunto [3,6..15]
-- >>> unionGeneral [c1,c2,c3]
-- [1,4,7,10,13,2,5,8,11,14,3,6,9,12,15]
unionGeneral :: Eq a => [[a]] -> [a]
unionGeneral = foldr unionConjuntos []
```

1.2.23. Intersección de conjuntos

Definición 1.2.24. *Dados dos conjuntos A y B se define la* ***intersección*** *de A y B, notado* $A \cap B$*, como el conjunto formado por aquellos elementos que pertenecen a cada uno de los dos conjuntos, A y B, es decir,*

$$A \cap B = \{x \mid x \in A \wedge x \in B\}$$

La función `(interseccion c1 c2)` devuelve la intersección de los conjuntos `c1` y `c2`.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [1,3..20]
-- >>> let c2 = listaAConjunto [2,4..20]
-- >>> let c3 = listaAConjunto [2,4..30]
-- >>> let c4 = listaAConjunto [4,8..30]
-- >>> interseccion c1 c2
-- []
-- >>> interseccion c3 c4
-- [4,8,12,16,20,24,28]
```

```
interseccion :: Eq a => [a] -> [a] -> [a]
interseccion = intersect
```

1.2.24. Producto cartesiano

Definición 1.2.25. *El **producto cartesiano** [2] de dos conjuntos A y B es una operación sobre ellos que resulta en un nuevo conjunto $A \times B$ que contiene a todos los pares ordenados tales que la primera componente pertenece a A y la segunda pertenece a B; es decir,*

$$A \times B = \{(a, b) \mid a \in A,\ b \in B\}$$

La función `(productoCartesiano c1 c2)` devuelve el producto cartesiano de xs e ys.

```
-- | Ejemplos
-- >>> let c1 = listaAConjunto [3,1]
-- >>> let c2 = listaAConjunto [2,4,7]
-- >>> productoCartesiano c1 c2
-- [(3,2),(3,4),(3,7),(1,2),(1,4),(1,7)]
-- >>> productoCartesiano c2 c1
-- [(2,3),(2,1),(4,3),(4,1),(7,3),(7,1)]
productoCartesiano :: [a] -> [b] -> [(a,b)]
productoCartesiano xs ys = [ (x,y) | x <- xs , y <- ys]
```

1.2.25. Combinaciones

Definición 1.2.26. *Las **combinaciones** de un conjunto S tomados en grupos de n son todos los subconjuntos de S con n elementos.*

La función `(combinaciones n xs)` devuelve las combinaciones de los elementos de `xs` en listas de `n` elementos.

```
-- | Ejemplos
-- >>> combinaciones 3 ['a'..'d']
-- ["abc","abd","acd","bcd"]
-- >>> combinaciones 2 [2,4..8]
-- [[2,4],[2,6],[2,8],[4,6],[4,8],[6,8]]
combinaciones :: Integer -> [a] -> [[a]]
combinaciones 0 _          = [[]]
combinaciones _ []         = []
combinaciones k (x:xs) =
    [x:ys | ys <- combinaciones (k-1) xs] ++ combinaciones k xs
```

[2] https://en.wikipedia.org/wiki/Cartesian_product

1.2.26. Variaciones con repetición

Definición 1.2.27. *Las **variaciones con repetición** de m elementos tomados en grupos de n es el número de diferentes n–tuplas de un conjunto de m elementos.*

La función `(variacionesR n xs)` devuelve las variaciones con con repetición de los elementos de `xs` en listas de `n` elementos.

```
-- | Ejemplos
-- >>> variacionesR 3 ['a','b']
-- ["aaa","aab","aba","abb","baa","bab","bba","bbb"]
-- >>> variacionesR 2 [2,3,5]
-- [[2,2],[2,3],[2,5],[3,2],[3,3],[3,5],[5,2],[5,3],[5,5]]
variacionesR :: Int -> [a] -> [[a]]
variacionesR _ [] = [[]]
variacionesR 0 _  = [[]]
variacionesR k us =
    [u:vs | u <- us, vs <- variacionesR (k-1) us]
```

1.3. Elección de la representación de conjuntos

Nota 1.3.1. En el módulo `Conjuntos` se elige la representación de conjunto con la que se trabajará.

```
-- Seleccionar para trabajar con los conjuntos como listas sin elementos
-- repetidos.

module Conjuntos
       (module ConjuntosConListas)
       where
import ConjuntosConListas

-- Seleccionar para trabajar con los conjuntos como listas ordenadas sin
-- elementos repetidos.

-- module Conjuntos
--        (module ConjuntosConListasOrdenadasSinRepeticion)
--        where
-- import ConjuntosConListasOrdenadasSinRepeticion
```

Capítulo 2

Relaciones y funciones

2.1. Relaciones

Las relaciones que existen entre personas, números, conjuntos y muchas otras entidades pueden formalizarse en la idea de relación binaria. En esta sección se define y desarrolla este concepto, particularizando en el caso de las relaciones homogéneas y en el de las funciones.

2.1.1. Relación binaria

Definición 2.1.1. *Una **relación binaria*** [1] *(o **correspondencia**) entre dos conjuntos A y B es un subconjunto del producto cartesiano $A \times B$.*

La función `(esRelacion xs ys r)` se verifica si `r` es una relación binaria de `xs` en `ys`. Por ejemplo,

```
-- | Ejemplos
-- >>> esRelacion [3,1] [2,4,7] [(1,2),(3,4)]
-- True
-- >>> esRelacion [3,1] [2,4,7] [(1,2),(3,1)]
-- False
esRelacion :: (Ord a, Ord b) => [a] -> [b] -> [(a,b)] -> Bool
esRelacion xs ys r =
  r `esSubconjunto` productoCartesiano xs ys
```

2.1.2. Imagen por una relación

Definición 2.1.2. *Si R es una relación binaria, la **imagen del elemento** x en la relación R es el conjunto de los valores correspondientes a x en R.*

[1] https://en.wikipedia.org/wiki/Binary_relation

La función `(imagenRelacion r x)` es la imagen de `x` en la relación `r`.

```
-- | Ejemplos
-- >>> imagenRelacion [(1,3),(2,5),(1,4)] 1
-- [3,4]
-- >>> imagenRelacion [(1,3),(2,5),(1,4)] 2
-- [5]
-- >>> imagenRelacion [(1,3),(2,5),(1,4)] 3
-- []
imagenRelacion :: (Ord a, Ord b) => [(a,b)] -> a -> [b]
imagenRelacion r x =
  nub [y | (z,y) <- r, z == x]
```

2.1.3. Dominio de una relación

Definición 2.1.3. *Dada una relación binaria R, su **dominio** es el conjunto que contiene a todos los valores que se toman en la relación R.*

La función `(dominio r)` devuelve el dominio de la relación r.

```
-- | Ejemplo
-- >>> dominio [(3,2),(5,1),(3,4)]
-- [3,5]
dominio :: Ord a => [(a,b)] -> [a]
dominio r = listaAConjunto (map fst r)
```

2.1.4. Rango de una relación

Definición 2.1.4. *El **rango** de una relación binaria R es el conjunto de las imágenes de mediante R.*

La función `(rango r)` devuelve el rango de la relación binaria `r`.

```
-- | Ejemplo
-- >>> rango [(3,2),(5,2),(3,4)]
-- [2,4]
rango :: Ord b => [(a,b)] -> [b]
rango r = listaAConjunto (map snd r)
```

2.1.5. Antiimagen por una relación

Definición 2.1.5. *La **antiimagen del elemento** y por una relación r es el conjunto de los elementos cuya imagen es y.*

La `(antiImagenRelacion r y)` es la antiimagen del elemento y en la relación binaria `r`.

```
-- | Ejemplo
-- >>> antiImagenRelacion [(1,3),(2,3),(7,4)] 3
-- [1,2]
antiImagenRelacion :: (Ord a, Ord b) => [(a,b)] -> b -> [a]
antiImagenRelacion r y =
  nub [x | (x,z) <- r, z == y]
```

2.1.6. Relación funcional

Definición 2.1.6. *Dada una relación binaria R, se dice* ***funcional*** *si todos los elementos de su dominio tienen una única imagen en R.*

La función `(esFuncional r)` se verifica si la relación `r` es funcional.

```
-- | Ejemplos
-- >>> esFuncional [(3,2),(5,1),(7,9)]
-- True
-- >>> esFuncional [(3,2),(5,1),(3,4)]
-- False
-- >>> esFuncional [(3,2),(5,1),(3,2)]
-- True
esFuncional :: (Ord a, Ord b) => [(a,b)] -> Bool
esFuncional r =
  and [esUnitario (imagenRelacion r x) | x <- dominio r]
```

2.2. Relaciones homogéneas

Para elaborar la presente sección, se han consultado los apuntes de "Álgebra básica" ([6]), asignatura del primer curso del Grado en Matemáticas.

Definición 2.2.1. *Una relación binaria entre dos conjuntos A y B se dice que es* ***homogénea*** *si los conjuntos son iguales; es decir, si $A = B$. Si el par $(x, y) \in AA$ está en la relación homogénea R, diremos que x está R–relacionado con y, o relacionado con y por R. Esto se notará frecuentemente xRy (nótese que el orden es importante).*

La función `(esRelacionHomogenea xs r)` se verifica si `r` es una relación binaria homogénea en el conjunto `xs`.

```
-- | Ejemplos
-- >>> esRelacionHomogenea [1..4] [(1,2),(2,4),(3,4),(4,1)]
-- True
-- >>> esRelacionHomogenea [1..4] [(1,2),(2,5),(3,4),(4,1)]
-- False
-- >>> esRelacionHomogenea [1..4] [(1,2),(3,4),(4,1)]
```

```
-- True
esRelacionHomogenea :: Ord a => [a] -> [(a,a)] -> Bool
esRelacionHomogenea xs = esRelacion xs xs
```

Nota 2.2.1. El segundo argumento que recibe la función ha de ser una lista de pares con ambas componentes del mismo tipo.

La función `(estaRelacionado r x y)` se verifica si `x` está relacionado con `y` en la relación homogénea `r`.

```
-- | Ejemplos
-- >>> estaRelacionado [(1,3),(2,5),(4,6)] 2 5
-- True
-- >>> estaRelacionado [(1,3),(2,5),(4,6)] 2 3
-- False
estaRelacionado :: Ord a => [(a,a)] -> a -> a -> Bool
estaRelacionado r x y = (x,y) `elem` r
```

2.2.1. Relaciones reflexivas

Definición 2.2.2. *Sea R una relación binaria homogénea en el conjunto A. Diremos que R es **reflexiva** cuando todos los elementos de A están relacionados por R consigo mismos; es decir, cuando $\forall x \in A$ se tiene que xRx.*

La función `(esReflexiva xs r)` se verifica si la relación `r` en `xs` es reflexiva.

```
-- | Ejemplos
-- >>> esReflexiva [1,2] [(1,1),(1,2),(2,2)]
-- True
-- >>> esReflexiva [1,2] [(1,1),(1,2)]
-- False
esReflexiva :: Ord a => [a] -> [(a,a)] -> Bool
esReflexiva xs r = zip xs xs `esSubconjunto` r
```

Nota 2.2.2. En el conjunto $\mathbb{N}$, las relaciones caracterizadas por:

- $xRy \longleftrightarrow x \leq y$,
- $xSy \longleftrightarrow x - y$ es par,
- $xTy \longleftrightarrow x$ divide a y,

son relaciones binarias homogéneas reflexivas.

2.2.2. Relaciones simétricas

Definición 2.2.3. *Diremos que una relación homogénea R es **simétrica** cuando*

$$\forall (x,y) \in R \longrightarrow (y,x) \in R$$

La función `(esSimetrica r)` se verifica si la relación `r` es simétrica.

```
-- | Ejemplos
-- >>> esSimetrica [(1,1),(1,2),(2,1)]
-- True
-- >>> esSimetrica [(1,1),(1,2),(2,2)]
-- False
esSimetrica :: Ord a => [(a,a)] -> Bool
esSimetrica r =
  listaAConjunto [(y,x) | (x,y) <- r] `esSubconjunto` r
```

Nota 2.2.3. En el conjunto $\mathbb{N}$, la relación caracterizada por $xSy \longleftrightarrow x - y$ es par, es una relación binaria homogénea simétrica.

2.2.3. Relaciones antisimétricas

Definición 2.2.4. *Sea R una relación binaria homogénea en el conjunto A. Diremos que R es **antisimétrica** cuando*

$$\forall (x,y)[(x,y) \in R \wedge (y,x) \in R \longrightarrow x = y$$

La función `(esAntisimetrica r)` se verifica si la relación `r` es antisimétrica.

```
-- | Ejemplos
-- >>> esAntisimetrica [(1,2),(3,1)]
-- True
-- >>> esAntisimetrica [(1,2),(2,1)]
-- False
esAntisimetrica :: Ord a => [(a,a)] -> Bool
esAntisimetrica r =
  and [x == y | (x,y) <- r, (y,x) `elem` r]
```

Nota 2.2.4. En el conjunto $\mathbb{N}$, las relaciones caracterizadas por:

- $xRy \longleftrightarrow x \leq y$,
- $xTy \longleftrightarrow x$ divide a y,
- $xRy \longleftrightarrow x < y$,

son relaciones binarias homogéneas antisimétricas.

2.2.4. Relaciones transitivas

Definición 2.2.5. *Sea R una relación binaria homogénea en el conjunto A. Diremos que R es **transitiva** cuando $\forall (x,y),(y,z) \in R$ se tiene que xRy e $yRz \longrightarrow xRz$.*

La función `(esTransitiva r)` se verifica si la relación `r` es transitiva.

```
-- | Ejemplos
-- >>> esTransitiva [(1,2),(1,3),(2,3)]
-- True
-- >>> esTransitiva [(1,2),(2,3)]
-- False
esTransitiva :: Ord a => [(a,a)] -> Bool
esTransitiva r =
  listaAConjunto [(x,z) | (x,y) <- r, (w,z) <- r, y == w] `esSubconjunto` r
```

Nota 2.2.5. En el conjunto $\mathbb{N}$, las relaciones caracterizadas por:

- $xRy \longleftrightarrow x \leq y$,
- $xSy \longleftrightarrow x - y$ es par,
- $xTy \longleftrightarrow x$ divide a y,
- $xRy \longleftrightarrow x < y$,

son relaciones binarias homogéneas transitivas.

2.2.5. Relaciones de equivalencia

Definición 2.2.6. *Las relaciones homogéneas que son a la vez reflexivas, simétricas y transitivas se denominan* ***relaciones de equivalencia****.*

La función `(esRelacionEquivalencia xs r)` se verifica si `r` es una relación de equivalencia en `xs`.

```
-- | Ejemplos
-- >>> esRelacionEquivalencia [1..3] [(1,1),(1,2),(2,1),(2,2),(3,3)]
-- True
-- >>> esRelacionEquivalencia [1..3] [(1,2),(2,1),(2,2),(3,3)]
-- False
-- >>> esRelacionEquivalencia [1..3] [(1,1),(1,2),(2,2),(3,3)]
-- False
esRelacionEquivalencia :: Ord a => [a] -> [(a,a)] -> Bool
esRelacionEquivalencia xs r =
  esReflexiva xs r   &&
  esSimetrica r      &&
  esTransitiva r
```

Nota 2.2.6. En el conjunto $\mathbb{N}$, la relación caracterizada por $xSy \longleftrightarrow x - y$ es par, es una relación de equivalencia.

2.2.6. Relaciones de orden

Definición 2.2.7. *Las relaciones homogéneas que son a la vez reflexivas, antisimétricas y transitivas se denominan* ***relaciones de orden****.*

La función `(esRelacionOrden xs r)` se verifica si `r` es una relación de orden en `xs`.

```
-- | Ejemplo
-- >>> esRelacionOrden [1..3] [(1,1),(1,2),(1,3),(2,2),(2,3),(3,3)]
-- True
esRelacionOrden :: Ord a => [a] -> [(a,a)] -> Bool
esRelacionOrden xs r =
  esReflexiva xs r  &&
  esAntisimetrica r &&
  esTransitiva r
```

Nota 2.2.7. En el conjunto $\mathbb{N}$, las relaciones caracterizadas por:

- $xRy \longleftrightarrow x \leq y$,
- $xTy \longleftrightarrow x$ divide a y,

son relaciones de orden.

2.2.7. Clases de equivalencia

Definición 2.2.8. *Si R es una relación de equivalencia en A, denominamos **clase de equivalencia** de un elemento $x \in A$ al conjunto de todos los elementos de A relacionados con x; es decir, $\overline{x} = R(x) = \{y \in A | xRy\}$ donde la primera notación se usa si la relación con la que se está tratando se sobreentiende, y la segunda si no es así.*

La función `(clasesEquivalencia xs r)` devuelve las clases de la relación de equivalencia `r` en `xs`.

```
-- | Ejemplo
-- >>> let r = [(x,y) | x <- [1..5], y <- [1..5], even (x-y)]
-- >>> clasesEquivalencia [1..5] r
-- [[1,3,5],[2,4]]
clasesEquivalencia :: Ord a => [a] -> [(a,a)] -> [[a]]
clasesEquivalencia _ [] = []
clasesEquivalencia [] _ = []
clasesEquivalencia (x:xs) r = (x:c) : clasesEquivalencia (xs \\ c) r
  where c = filter (estaRelacionado r x) xs
```

2.3. Funciones

Definición 2.3.1. *Dada una relación F entre A y B, se dirá que es una **función** si es una relación binaria, es funcional y todos los elementos de A están en el dominio.*

La función `(esFuncion xs ys f)` se verifica si `f` es una función de `xs` en `ys`.

```
-- | Ejemplos
-- >>> esFuncion [1,3] [2,4,7] [(1,7),(3,2)]
```

```
-- True
-- >>> esFuncion [1,3] [2,4,7] [(1,7)]
-- False
-- >>> esFuncion [1,3] [2,4,7] [(1,4),(1,7),(3,2)]
-- False
esFuncion :: (Ord a, Ord b) => [a] -> [b] -> [(a,b)] -> Bool
esFuncion xs ys f =
  esRelacion xs ys f &&
  xs `esSubconjunto` dominio f &&
  esFuncional f
```

Nota 2.3.1. A lo largo de la sección representaremos a las funciones como listas de pares.

```
type Funcion a b = [(a,b)]
```

La función `(funciones xs ys)` devuelve todas las posibles funciones del conjunto `xs` en `ys`.

```
-- | Ejemplos
-- >>> pp $ funciones [1,2] [3,4]
-- [[(1, 3),(2, 3)],[(1, 3),(2, 4)],[(1, 4),(2, 3)],
--  [(1, 4),(2, 4)]]
-- >>> pp $ funciones [1,2] [3,4,5]
-- [[(1, 3),(2, 3)],[(1, 3),(2, 4)],[(1, 3),(2, 5)],
--  [(1, 4),(2, 3)],[(1, 4),(2, 4)],[(1, 4),(2, 5)],
--  [(1, 5),(2, 3)],[(1, 5),(2, 4)],[(1, 5),(2, 5)]]
-- >>> pp $ funciones [0,1,2] [3,4]
-- [[(0, 3),(1, 3),(2, 3)],[(0, 3),(1, 3),(2, 4)],
--  [(0, 3),(1, 4),(2, 3)],[(0, 3),(1, 4),(2, 4)],
--  [(0, 4),(1, 3),(2, 3)],[(0, 4),(1, 3),(2, 4)],
--  [(0, 4),(1, 4),(2, 3)],[(0, 4),(1, 4),(2, 4)]]
funciones :: [a] -> [b] -> [Funcion a b]
funciones xs ys =
  [zip xs zs | zs <- variacionesR (length xs) ys]
```

2.3.1. Imagen por una función

Definición 2.3.2. *Si f es una función entre A y B y x es un elemento del conjunto A, la **imagen del elemento** x por la función f es el valor asociado a x por la función f.*

La función `(imagen f x)` es la imagen del elemento `x` en la función `f`.

```
-- | Ejemplos
-- >>> imagen [(1,7),(3,2)] 1
-- 7
-- >>> imagen [(1,7),(3,2)] 3
```

```
-- 2
imagen :: (Ord a, Ord b) => Funcion a b -> a -> b
imagen f x = head (imagenRelacion f x)
```

La función `(imagenConjunto f xs)` es la imagen del conjunto `xs` en la función `f`.

```
-- | Ejemplos
-- >>> imagenConjunto [(1,7),(3,2),(4,3)] [1,4]
-- [7,3]
-- >>> imagenConjunto [(1,7),(3,2)] [3,1]
-- [2,7]
imagenConjunto :: (Ord a, Ord b) => Funcion a b -> [a] -> [b]
imagenConjunto f xs = nub (map (imagen f) xs)
```

2.3.2. Funciones inyectivas

Definición 2.3.3. *Diremos que una función f entre dos conjuntos es* ***inyectiva*** [2] *si a elementos distintos del dominio le corresponden elementos distintos de la imagen; es decir, si $\forall a, b \in dominio(f)$ tales que $a \neq b$, $f(a) \neq f(b)$.*

La función `(esInyectiva fs)` se verifica si la función `fs` es inyectiva.

```
-- | Ejemplos
-- >>> esInyectiva [(1,4),(2,5),(3,6)]
-- True
-- >>> esInyectiva [(1,4),(2,5),(3,4)]
-- False
-- >>> esInyectiva [(1,4),(2,5),(3,6),(3,6)]
-- True
esInyectiva :: (Ord a, Ord b) => Funcion a b -> Bool
esInyectiva f =
  all esUnitario [antiImagenRelacion f y | y <- rango f]
```

2.3.3. Funciones sobreyectivas

Definición 2.3.4. *Diremos que una función f entre dos conjuntos A y B es* ***sobreyectiva*** [3] *si todos los elementos de B son imagen de algún elemento de A.*

La función `(esSobreyectiva xs ys f)` se verifica si la función `f` es sobreyectiva. A la hora de definirla, estamos contando con que `f` es una función entre `xs` y `ys`.

```
-- | Ejemplos
-- >>> esSobreyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,6)]
```

[2] https://en.wikipedia.org/wiki/Injective_function
[3] https://en.wikipedia.org/wiki/Surjective_function

```
-- True
-- >>> esSobreyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,4)]
-- False
-- >>> esSobreyectiva [1,2,3] [4,5,6] [(1,4),(2,4),(3,6),(3,6)]
-- False
esSobreyectiva :: (Ord a,Ord b) => [a] -> [b] -> Funcion a b -> Bool
esSobreyectiva _ ys f = ys `esSubconjunto` rango f
```

2.3.4. Funciones biyectivas

Definición 2.3.5. *Diremos que una función f entre dos conjuntos A y B es* ***biyectiva*** [4] *si cada elementos de B es imagen de un único elemento de A.*

La función `(esBiyectiva xs ys f)` se verifica si la función `f` es biyectiva.

```
-- | Ejemplos
-- >>> esBiyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,6),(3,6)]
-- True
-- >>> esBiyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,4)]
-- False
-- >>> esBiyectiva [1,2,3] [4,5,6,7] [(1,4),(2,5),(3,6)]
-- False
esBiyectiva :: (Ord a, Ord b) => [a] -> [b] -> Funcion a b -> Bool
esBiyectiva xs ys f =
  esInyectiva f && esSobreyectiva xs ys f
```

La funciones `biyecciones1 xs ys` y `biyecciones2 xs ys` devuelven la lista de todas las biyecciones entre los conjuntos `xs` y `ys`. La primera lo hace filtrando las funciones entre los conjuntos que son biyectivas y la segunda lo hace construyendo únicamente las funciones biyectivas entre los conjuntos, con el consecuente ahorro computacional.

```
ghci> length (biyecciones1 [1..7] ['a'..'g'])
5040
(16.75 secs, 4,146,744,104 bytes)
ghci> length (biyecciones2 [1..7] ['a'..'g'])
5040
(0.02 secs, 0 bytes)
ghci> length (biyecciones1 [1..6] ['a'..'g'])
0
(2.53 secs, 592,625,824 bytes)
ghci> length (biyecciones2 [1..6] ['a'..'g'])
0
(0.01 secs, 0 bytes)
```

[4] https://en.wikipedia.org/wiki/Bijective_function

```
biyecciones1 :: (Ord a, Ord b) => [a] -> [b] -> [Funcion a b]
biyecciones1 xs ys =
  filter (esBiyectiva xs ys) (funciones xs ys)

biyecciones2 :: (Ord a, Ord b) => [a] -> [b] -> [Funcion a b]
biyecciones2 xs ys
  | length xs /= length ys = []
  | otherwise              = [zip xs zs | zs <- permutations ys]
```

Nota 2.3.2. En lo que sigue trabajaremos con la función `biyecciones2` así que la redefiniremos como `biyecciones`.

```
biyecciones :: (Ord a, Ord b) => [a] -> [b] -> [Funcion a b]
biyecciones = biyecciones2
```

2.3.5. Inversa de una función

Definición 2.3.6. *Si f es una función biyectiva entre los conjuntos A y B, definimos la* ***función inversa*** [5] *como la función que a cada elemento de B le hace corresponder el elemento de A del que es imagen en B.*

El valor de `(inversa f)` es la función inversa de `f`.

```
-- | Ejemplos
-- >>> inversa [(1,4),(2,5),(3,6)]
-- [(4,1),(5,2),(6,3)]
-- >>> sort (inversa [(1,'f'),(2,'m'),(3,'a')])
-- [('a',3),('f',1),('m',2)]
inversa :: (Ord a, Ord b) => Funcion a b -> Funcion b a
inversa f = listaAConjunto [(y,x) | (x,y) <- f]
```

Nota 2.3.3. Para considerar la inversa de una función, esta tiene que ser biyectiva. Luego `(inversa f)` asigna a cada elemento del conjunto imagen (que en este caso coincide con la imagen) uno y solo uno del conjunto de salida.

La función `(imagenInversa f y)` devuelve el elemento del conjunto de salida de la función `f` tal que su imagen es `y`.

```
-- | Ejemplos
-- >>> imagenInversa [(1,4),(2,5),(3,6)] 5
-- 2
-- >>> imagenInversa [(1,'f'),(2,'m'),(3,'a')] 'a'
-- 3
imagenInversa :: (Ord a, Ord b) => Funcion a b -> b -> a
imagenInversa f = imagen (inversa f)
```

[5] https://en.wikipedia.org/wiki/Inverse_function

Capítulo 3

Introducción a la teoría de grafos

Se dice que la Teoría de Grafos tiene su origen en 1736, cuando Euler dio una solución al problema (hasta entonces no resuelto) de los siete puentes de Königsberg: ¿existe un camino que atraviese cada uno de los puentes exactamente una vez?
Para probar que no era posible, Euler sustituyó cada región por un nodo y cada puente por una arista, creando el primer grafo que fuera modelo de un problema matemático.



Figura 3.1: Dibujo de los puentes de Königsberg



Figura 3.2: Modelo de los puentes de Königsberg

Desde entonces, se ha ido desarrollando esta metodología hasta convertise en los últimos años en una herramienta importante en áreas del conocimiento muy variadas como, por ejemplo: la Investigación Operativa, la Computación, la Ingeniería Eléctrica, la Geografía y la Química. Es por ello que, además, se ha erigido como una nueva disciplina matemática, que generalmente asociada a las ramas de Topología y Álgebra.
La utilidad de los grafos se basa en su gran poder de abstracción y una representación muy clara de cualquier relación, lo que facilita enormemente tanto la fase de modelado como la de resolución de cualquier problema. Gracias a la Teoría de Grafos se han desarrollado una gran variedad de algoritmos y métodos de decisión que podemos implementar a través de lenguajes funcionales y permiten automatizar la resolución de muchos problemas, a menudo tediosos de resolver a mano.

3.1. Definición de grafo

En primer lugar, vamos a introducir terminología básica en el desarrollo de la Teoría de Grafos.

Definición 3.1.1. *Un **grafo** G es un par (V, A), donde V es el conjunto cuyos elementos llamamos **vértices** (o **nodos**) y A es un conjunto cuyos elementos llamamos **aristas**.*

Definición 3.1.2. *Una **arista** de un grafo $G = (V, A)$, es un conjunto de dos elementos de V. Es decir, para dos vértices v, v' de G, (v, v') y (v', v) representa la misma arista.*

Definición 3.1.3. *Dado un grafo $G = (V, A)$, diremos que un vértice $v \in V$ es **adyacente** a $v' \in V$ si $(v', v) \in A$.*

Definición 3.1.4. *Si en un grafo dirigido se permiten aristas repetidas, lo llamaremos **multigrafo**. Si no se permiten, lo llamaremos **grafo regular**.*

Nota 3.1.1. Denotaremos por $|V|$ al número de vértices y por $|A|$ al número de aristas del grafo (V, A).

Ejemplo 3.1.2. Sea $G = (V, A)$ un grafo con $V = \{a, b, c, d\}$ y $A = \{(a, b), (a, c), (b, d), (d, d)\}$. En este grafo, los vértices a, d son adyacentes a b.



,

3.2. El TAD de los grafos

En esta sección, nos planteamos la tarea de implementar las definiciones presentadas anteriormente en un lenguaje funcional. En nuestro caso, el lenguaje que utilizaremos será Haskell. Definiremos el Tipo Abstracto de Dato (TAD) de los grafos y daremos algunos ejemplos de posibles representaciones de grafos con las que podremos trabajar.
Si consideramos un grafo finito cualquiera $G = (V, A)$, podemos ordenar el conjunto de los vértices y representarlo como $V = \{v_1, \ldots, v_n\}$ con $n = |V|$.
En primer lugar, necesitaremos crear un tipo `(Grafo)` cuya definición sea compatible con la entidad matemática que representa y que nos permita definir las operaciones que necesitamos para trabajar con los grafos. Estas operaciones son:

```
creaGrafo  -- [a] -> [(a,a)] -> Grafo a
vertices   -- Grafo a -> [a]
adyacentes -- Grafo a -> a -> [a]
aristaEn   -- (a,a) -> Grafo a -> Bool
aristas    -- Grafo a -> [(a,a)]
```

donde:

- `(creaGrafo vs as)` es un grafo tal que el conjunto de sus vértices es `vs` y el de sus aristas es `as`.
- `(vertices g)` es la lista de todos los vértices del grafo `g`.
- `(adyacentes g v)` es la lista de los vértices adyacentes al vértice `v` en el grafo `g`.
- `(aristaEn a g)` se verifica si `a` es una arista del grafo `g`.
- `(aristas g)` es la lista de las aristas del grafo `g`.

Nota 3.2.1. Las funciones que aparecen en la especificación del TAD no dependen de la representación que elijamos.

Ejemplo 3.2.2. Veamos un ejemplo de creación de grafo y su representación gráfica

```
creaGrafo [1..5] [(1,2),(1,3),(1,5),(2,4),
                  (2,5),(3,4),(3,5),(4,5)]
```



3.2.1. Grafos como listas de aristas

En el módulo `GrafoConListaDeAristas` se definen las funciones del TAD de los grafos dando su representación como conjuntos de aristas; es decir, representando a un grafo como dos conjuntos, la primera será la lista ordenada de los vértices y la segunda la lista ordenada de las aristas (en ambas listas se excluye la posibilidad de repeticiones).

Nota 3.2.3. Las ventajas de usar arrays frente a usar listas es que los array tienen acceso constante ($O(1)$) a sus elementos mientras que las listas tienen acceso lineal ($O(n)$) y que la actualización de un elemento en un array no supone espacio extra. Sin embargo, los arrays son representaciones muy rígidas: cualquier modificación en su estructura, como cambiar su tamaño, supone un gran coste computacional pues se tendría que crear de nuevo el array y, además, sus índices deben pertenecer

a la clase de los objetos indexables (`Ix`), luego perdemos mucha flexibilidad en la representación.

```
{-# LANGUAGE DeriveGeneric #-}

module GrafoConListaDeAristas
    ( Grafo
    , creaGrafo  -- [a] -> [(a,a)] -> Grafo a
    , vertices   -- Grafo a -> [a]
    , adyacentes -- Grafo a -> a -> [a]
    , aristaEn   -- (a,a) -> Grafo a -> Bool
    , aristas    -- Grafo a -> [(a,a)]
    ) where
```

En las definiciones del presente módulo se usarán las funciones `nub` y `sort` de la librería `Data.List`

Vamos a definir un nuevo tipo de dato `(Grafo a)`, que representará un grafo a partir de la lista de sus vértices (donde los vértices son de tipo `a`) y de aristas (que son pares de vértices).

```
data Grafo a = G [a] [(a,a)]
  deriving (Eq, Show, Generic)

instance (Out a) => Out (Grafo a)
```

Las funciones básicas que definiremos a partir de este tipo coincidirán con las indicadas en el TAD de los grafos.

- `(creaGrafo vs as)` es el grafo cuyo conjunto de vértices es `cs` y el de sus aristas es `as`.

```
-- | Ejemplo
-- >>> creaGrafo [1..5] [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
-- G [1,2,3,4,5] [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
creaGrafo :: Ord a => [a] -> [(a,a)] -> Grafo a
creaGrafo vs as =
  G (sort vs) (nub (sort [parOrdenado a | a <- as]))

parOrdenado :: Ord a => (a,a) -> (a,a)
parOrdenado (x,y) | x <= y    = (x,y)
                  | otherwise = (y,x)
```

Ejemplo 3.2.4. `ejGrafo` es el grafo



Los ejemplos usarán el siguiente grafo

```
ejGrafo :: Grafo Int
ejGrafo = creaGrafo [1..5]
                    [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
```

- `(vertices g)` es la lista de los vértices del grafo `g`.

```
-- | Ejemplo
-- >>> vertices ejGrafo
-- [1,2,3,4,5]
vertices :: Grafo a -> [a]
vertices (G vs _) = vs
```

- `(adyacentes g v)` es la lista de los vértices adyacentes al vértice `v` en el grafo `g`.

```
-- | Ejemplos
-- >>> adyacentes ejGrafo 4
-- [1,3,5]
-- >>> adyacentes ejGrafo 3
-- [2,4,5]
adyacentes :: Eq a => Grafo a -> a -> [a]
adyacentes (G _ as) v =
  [u | (u,x) <- as, x == v] `union` [u | (x,u) <- as, x == v]
```

- `(aristaEn a g)` se verifica si `a` es una arista del grafo `g`.

```
-- | Ejemplos
-- >>> (5,1) `aristaEn` ejGrafo
-- True
-- >>> (3,1) `aristaEn` ejGrafo
-- False
aristaEn :: Ord a => (a,a) -> Grafo a -> Bool
aristaEn a (G _ as) = parOrdenado a `elem` as
```

- `(aristas g)` es la lista de las aristas del grafo `g`.

```
-- | Ejemplo
-- >>> aristas ejGrafo
-- [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
aristas :: Grafo a -> [(a,a)]
aristas (G _ as) = as
```

3.3. Generador de grafos

En esta sección, presentaremos el generador de grafos que nos permitirá generar grafos como listas de aristas arbitrariamente y usarlos como ejemplos o para comprobar propiedades.
Para aprender a controlar el tamaño de los grafos generados, he consultado las siguientes fuentes:

* **QuickCheck: A Lightweight Tool for Random Testing of Haskell Programs** [1] ([3])
* **Property Testing using QuickCheck** [2] ([9])

`(generaGrafos n)` es un generador de grafos de hasta `n` vértices. Por ejemplo,

```
ghci> sample (generaGrafo 5)
G [1,2] []
G [1] [(1,1)]
G [] []
G [1,2,3,4] [(2,2)]
G [1,2,3] [(1,1),(1,2),(1,3),(2,2)]
G [1,2,3,4,5] [(1,2),(1,4),(1,5),(2,2),(2,3),(2,4),(3,3),(3,4),(4,4),(4,5)]
G [1] []
G [1,2,3] [(1,2),(2,2),(3,3)]
G [1,2,3,4] [(1,1),(1,4),(2,3),(2,4),(3,3),(3,4),(4,4)]
G [1,2] []
G [1,2,3] [(1,1),(1,2),(2,2),(3,3)]

ghci> sample (generaGrafo 2)
G [1,2] [(1,2),(2,2)]
G [1,2] [(1,1)]
G [1,2] [(1,1),(1,2)]
G [] []
G [1] [(1,1)]
G [1] []
G [1] []
G [] []
G [] []
G [] []
G [1] [(1,1)]
```

```
generaGrafo :: Int -> Gen (Grafo Int)
generaGrafo s = do
  let m = s  `mod` 11
  n  <- choose (0,m)
  as <- sublistOf [(x,y) | x <- [1..n], y <- [x..n]]
  return (creaGrafo [1..n] as)
```

[1] https://www.eecs.northwestern.edu/~robby/courses/395-495-2009-fall/quick.pdf
[2] https://www.dcc.fc.up.pt/~pbv/aulas/tapf/slides/quickcheck.html

Nota 3.3.1. Los grafos están contenidos en la clase de los objetos generables aleatoriamente.

```
instance Arbitrary (Grafo Int) where
  arbitrary = sized generaGrafo
```

En el siguiente ejemplo se pueden observar algunos grafos generados

```
ghci> sample (arbitrary :: Gen (Grafo Int))
G [] []
G [1] [(1,1)]
G [1,2] [(1,1),(1,2)]
G [1] [(1,1)]
G [1,2,3,4,5,6,7,8] [(1,4),(1,7),(2,2),(2,3),(2,5),(2,8),(3,5),
   (3,6),(3,8),(4,4),(4,6),(4,7),(4,8),(5,6),(6,6),(6,7),(7,8)]
G [1,2,3,4,5] [(1,1),(1,2),(1,4),(1,5),(2,4),(2,5),(3,4),(5,5)]
G [] []
G [] []
G [1,2,3] [(1,1),(2,2),(2,3)]
G [1,2,3,4] [(1,1),(1,2),(1,4),(2,2),(3,3)]
G [1,2,3,4,5,6,7] [(1,1),(1,5),(1,6),(1,7),(2,2),(2,4),(2,5),
   (2,6),(2,7),(3,3),(3,5),(4,6),(4,7),(5,5),(5,7),(6,6),(7,7)]
```

3.4. Ejemplos de grafos

El objetivo de esta sección es reunir una colección de grafos lo suficientemente extensa y variada como para poder utilizarla como recurso a la hora de comprobar las propiedades y definiciones de funciones que implementaremos más adelante.
En el proceso de recopilación de ejemplos, se ha trabajado con diversas fuentes:

- los apuntes de la asignatura "Matemática discreta" ([4]),
- los temas de la asignatura "Informática" ([1]) y
- el artículo "Graph theory" ([11]) de la Wikipedia.

Nota 3.4.1. Se utilizará la representación de los grafos como listas de aristas.

3.4.1. Grafo nulo

Definición 3.4.1. *Un **grafo nulo** es un grafo que no tiene ni vértices ni aristas.*

La función `(grafoNulo)` devuelve un grafo nulo.

```
grafoNulo :: Ord a => Grafo a
grafoNulo = creaGrafo [] []
```

La función `(esGrafoNulo g)` se verifica si `g` es un grafo nulo.

```
-- | Ejemplos
-- >>> esGrafoNulo grafoNulo
-- True
-- >>> esGrafoNulo (creaGrafo [] [(1,2)])
-- False
-- >>> esGrafoNulo (creaGrafo [1,2] [(1,2)])
-- False
esGrafoNulo :: Grafo a -> Bool
esGrafoNulo g =
  null (vertices g) && null (aristas g)
```

3.4.2. Grafo ciclo

Definición 3.4.2. *Un **ciclo**,* [3] *de orden n, $C(n)$, es un grafo no dirigido y no ponderado cuyo conjunto de vértices viene dado por $V = \{1, \ldots, n\}$ y el de las aristas por $A = \{(0,1), (1,2), \ldots, (n-2, n-1), (n-1, 0)\}$*

La función `(grafoCiclo n)` nos genera el ciclo de orden `n`.




```
-- | Ejemplos
-- >>> grafoCiclo 5
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
grafoCiclo :: Int -> Grafo Int
grafoCiclo 0 = grafoNulo
grafoCiclo 1 = creaGrafo [1] []
grafoCiclo n = creaGrafo [1..n]
                         ([(u,u+1) | u <- [1..n-1]] ++ [(n,1)])
```

3.4.3. Grafo de la amistad

Definición 3.4.3. *Un **grafo de la amistad*** [4] *de orden n es un grafo con $2n + 1$ vértices y $3n$ aristas formado uniendo n copias del ciclo C_3 por un vértice común. Lo denotamos por F_n.*

[3] https://es.wikipedia.org/wiki/Grafo_completo
[4] https://es.wikipedia.org/wiki/Grafo_de_la_amistad

La función `(grafoAmistad n)` genera el grafo de la amistad de orden n. Por ejemplo,




```
-- | Ejemplos
-- >>> pp $ grafoAmistad 2
-- G [1,2,3,4,5]
--   [(1, 2),(1, 3),(1, 4),(1, 5),(2, 3),(4, 5)]
-- >>> pp $ grafoAmistad 3
-- G [1,2,3,4,5,6,7]
--   [(1, 2),(1, 3),(1, 4),(1, 5),(1, 6),(1, 7),(2, 3),
--    (4, 5),(6, 7)]
grafoAmistad :: Int -> Grafo Int
grafoAmistad n =
  creaGrafo [1..2*n+1]
            ([(1,a) | a <- [2..2*n+1]] ++
             [(a,b) | (a,b) <-zip [2,4..2*n] [3,5..2*n+1]])
```

3.4.4. Grafo completo

Definición 3.4.4. *El **grafo completo**,* [5] *de orden n, $K(n)$, es un grafo no dirigido cuyo conjunto de vértices viene dado por $V = \{1,\ldots,n\}$ y tiene una arista entre cada par de vértices distintos.*

La función `(completo n)` nos genera el grafo completo de orden n. Por ejemplo,

[5] https://es.wikipedia.org/wiki/Grafo_completo




```
-- | Ejemplo
-- >>> completo 4
-- G [1,2,3,4] [(1,2),(1,3),(1,4),(2,3),(2,4),(3,4)]
completo :: Int -> Grafo Int
completo n =
  creaGrafo [1..n]
            [(a,b) | a <- [1..n], b <- [1..a-1]]
```

3.4.5. Grafo bipartito

Definición 3.4.5. *Un **grafo bipartito** [6] es un grafo $G = (V, A)$ verificando que el conjunto de sus vértices se puede dividir en dos subconjuntos disjuntos V_1, V_2 tales que $V_1 \cup V_2 = V$ de manera que $\forall u_1, u_2 \in V_1[(u_1, u_2) \notin A]$ y $\forall v_1, v_2 \in V_2[(v_1, v_2) \notin A]$. Un **grafo bipartito completo** [7] será entonces un grafo bipartito $G = (V_1 \cup V_2, A)$ en el que todos los vértices de una partición están conectados a los de la otra. Si $n = |V_1|, m = |V_2|$ denotamos al grafo bipartito $G = (V_1 \cup V_2, A)$ por $K_{n,m}$.*

La función `(bipartitoCompleto n m)` nos genera el grafo bipartito $K_{n,m}$. Por ejemplo,




```
-- | Ejemplo
-- >>> bipartitoCompleto 2 3
-- G [1,2,3,4,5] [(1,3),(1,4),(1,5),(2,3),(2,4),(2,5)]
bipartitoCompleto :: Int -> Int -> Grafo Int
bipartitoCompleto n m =
  creaGrafo [1..n+m]
            [(a,b) | a <- [1..n], b <- [n+1..n+m]]
```

[6] https://es.wikipedia.org/wiki/Grafo_bipartito
[7] https://es.wikipedia.org/wiki/Grafo_bipartito_completo

La función `(conjuntosVerticesDisjuntos g)` devuelve `Nothing` si el grafo `g` no es bipartito y `Just(xs,ys)` si lo es, donde `xs`, `ys` es una partición disjunta de los vértices de `g`.

```
-- | Ejemplo
-- >>> conjuntosVerticesDisjuntos (bipartitoCompleto 3 4)
-- Just ([3,2,1],[7,6,5,4])
-- >>> conjuntosVerticesDisjuntos (grafoCiclo 5)
-- Nothing
-- >>> conjuntosVerticesDisjuntos (grafoCiclo 6)
-- Just ([5,3,1],[6,4,2])
conjuntosVerticesDisjuntos :: Ord a => Grafo a -> Maybe ([a],[a])
conjuntosVerticesDisjuntos g
    | null (vertices g) = Just ([],[])
    | otherwise = aux (vertices g) [] [] [] []
    where u = union
          a = adyacentes g
          d xs x = xs \\ [x]
          aux [] _ _ r b  = if (null (intersect r b))
                            then (Just (r,b))
                            else Nothing
          aux (v:vs) [] [] [] [] = aux vs (a v) [v] [v] []
          aux (v:vs) [] c r b
              | null (a v) = aux vs [] (v:c) (v:r) b
              | null ((a v) \\ c) =
                  if (or [elem u r | u <- (a v)])
                  then (if (or [elem u b | u <- (a v)])
                        then Nothing
                        else (aux vs [] (v:c) r (v:b)))
                  else (aux vs [] (v:c) (v:r) b)
              | otherwise =
                  if (or [elem u r | u <- (a v)])
                  then (if (or [elem u b | u <- (a v)])
                        then Nothing
                        else (aux vs ((a v) \\ c) (v:c) r (v:b)))
                  else (aux vs ((a v) \\ c) (v:c) (v:r) b)
          aux vs (q:qs) c r b
              | null ((a q) \\ c) =
                  if (or [elem u r | u <- (a q)])
                  then (if (or [elem u b | u <- (a q)])
                        then Nothing
                        else (aux (d vs q) qs (q:c) r (q:b)))
                  else (aux (d vs q) qs (q:c) (q:r) b)
              | otherwise =
                  if (or [elem u r | u <- (a q)])
                  then (if (or [elem u b | u <- (a q)])
                        then Nothing
                        else (aux (d vs q) (u ((a q) \\ c) qs)
```

```
                                (q:c) r (q:b)))
                    else (aux (d vs q) (u ((a q) \\ c) qs)
                              (q:c) (q:r) b)
```

La función `(esBipartito g)` se verifica si el grafo `g` es bipartito.

```
-- | Ejemplo
-- >>> esBipartito (bipartitoCompleto 3 4)
-- True
-- >>> esBipartito (grafoCiclo 5)
-- False
-- >>> esBipartito (grafoCiclo 6)
-- True
esBipartito :: Ord a => Grafo a -> Bool
esBipartito g = isJust (conjuntosVerticesDisjuntos g)
```

3.4.6. Grafo estrella

Definición 3.4.6. *Una **estrella*** [8] *de orden n es el grafo bipartito completo $K_{1,n}$. Denotaremos a una estrella de orden n por S_n. Una estrella con 3 aristas se conoce en inglés como **claw** (garra o garfio).*

La función `(grafoEstrella n)` crea un grafo circulante a partir de su orden n.




```
-- | Ejemplo
-- >>> grafoEstrella 5
-- G [1,2,3,4,5,6] [(1,2),(1,3),(1,4),(1,5),(1,6)]
grafoEstrella :: Int -> Grafo Int
grafoEstrella = bipartitoCompleto 1
```

3.4.7. Grafo rueda

Definición 3.4.7. *Un **grafo rueda*** [9] *de orden n es un grafo no dirigido y no ponderado con n vértices que se forma conectando un único vértice a todos los vértices de un ciclo C_{n-1}. Lo denotaremos por W_n.*

[8] https://en.wikipedia.org/wiki/Star_(graph_theory))
[9] https://es.wikipedia.org/wiki/Grafo_rueda

La función `(grafoRueda n)` crea un grafo rueda a partir de su orden `n`.




```
-- | Ejemplo
-- >>> pp $ grafoRueda 6
-- G [1,2,3,4,5,6]
--   [(1, 2),(1, 3),(1, 4),(1, 5),(1, 6),(2, 3),(2, 6),
--    (3, 4),(4, 5),(5, 6)]
grafoRueda :: Int -> Grafo Int
grafoRueda n =
  creaGrafo [1..n]
            ([(1,a) | a <- [2..n]] ++
             [(a,b) | (a,b) <- zip (2:[2..n-1]) (3:n:[4..n])])
```

3.4.8. Grafo circulante

Definición 3.4.8. *Un **grafo circulante** [10] de orden $n \geq 3$ y saltos $\{s_1, \ldots, s_k\}$ es un grafo no dirigido y no ponderado $G = (\{1, \ldots, n\}, A)$ en el que cada nodo $\forall i \in V$ es adyacente a los $2k$ nodos $i \pm s_1, \ldots, i \pm s_k$ mód n. Lo denotaremos por $Cir_n^{s_1,\ldots,s_k}$.*

La función `(grafoCirculante n ss)` crea un grafo circulante a partir de su orden `n` y de la lista de sus saltos `ss`. Por ejemplo,




```
-- | Ejemplo
-- >>> pp $ grafoCirculante 6 [1,2]
-- G [1,2,3,4,5,6]
--   [(1, 2),(1, 3),(1, 5),(1, 6),(2, 3),(2, 4),(2, 6),
--    (3, 4),(3, 5),(4, 5),(4, 6),(5, 6)]
grafoCirculante :: Int -> [Int] -> Grafo Int
grafoCirculante n ss =
```

[10] https://en.wikipedia.org/wiki/Circulant_graph

```
  creaGrafo [1..n]
            [(a,b) | a <- [1..n]
                   , b <- sort (auxCir a ss n)
                   , a < b]
  where auxCir v ss1 k =
          concat [[fun (v+s) k, fun (v-s) k] | s <- ss1]
        fun a b = if mod a b == 0 then b else mod a b
```

3.4.9. Grafo de Petersen generalizado

El **grafo de Petersen generalizado** [11] que denotaremos $GP_{n,k}$ (con $n \geq 3$ y $1 \leq k \leq (n-1)/2$) es un grafo formado por un grafo circulante $Cir^n_{\{k\}}$ en el interior, rodeado por un ciclo C_n al que está conectado por una arista saliendo de cada vértice, de forma que se creen n polígonos regulares. El grafo $GP_{n,k}$ tiene $2n$ vértices y $3n$ aristas. La función `(grafoPetersenGen n k)` devuelve el grafo de Petersen generalizado $GP_{n,k}$.




```
-- | Ejemplo
-- >>> pp $ grafoPetersenGen 4 2
-- G [1,2,3,4,5,6,7,8]
--   [(1, 3),(1, 5),(2, 4),(2, 6),(3, 7),(4, 8),(5, 6),
--    (5, 8),(6, 7),(7, 8)]
grafoPetersenGen :: Int -> Int -> Grafo Int
grafoPetersenGen n k =
  creaGrafo [1..2*n]
            (filter p (aristas (grafoCirculante n [k])) ++
             [(x,x+n) | x <- [1..n]] ++
             (n+1,n+2) : (n+1,2*n) : [(x,x+1) | x <- [n+2..2*n-1]])
  where p (a,b) = a < b
```

[11] https://en.wikipedia.org/wiki/Generalized_Petersen_graph

3.4.10. Otros grafos importantes

Grafo de Thomson

Definición 3.4.9. *El grafo bipartito completo $K_{3,3}$ es conocido como el **grafo de Thomson** y, como veremos más adelante, será clave a la hora de analizar propiedades topológicas de los grafos.*



La función `(grafoThomson)` genera el grafo de Thomson.

```
-- | Ejemplo
-- >>> pp $ grafoThomson
-- G [1,2,3,4,5,6]
--   [(1, 4),(1, 5),(1, 6),(2, 4),(2, 5),(2, 6),(3, 4),
--    (3, 5),(3, 6)]
grafoThomson :: Grafo Int
grafoThomson = bipartitoCompleto 3 3
```

Grafo de Heawood

El **grafo de Heawood** [12] es un grafo no dirigido, regular con 14 vértices y 21 aristas. Todos sus vértices son incidentes a exactamente 3 aristas; es decir, es un grafo **cúbico**. Tiene importantes propiedades geométricas y topológicas.



[12] https://en.wikipedia.org/wiki/Heawood_graph

La función `grafoHeawood` genera el grafo de Heawood.

```
-- | Ejemplo
-- >>> pp $ grafoHeawood
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14]
--   [(1, 2),(1, 6),(1, 14),(2, 3),(2, 11),(3, 4),(3, 8),
--    (4, 5),(4, 13),(5, 6),(5, 10),(6, 7),(7, 8),(7, 12),
--    (8, 9),(9, 10),(9, 14),(10, 11),(11, 12),(12, 13),
--    (13, 14)]
grafoHeawood :: Grafo Int
grafoHeawood =
  creaGrafo [1..14]
            (aristas (grafoCiclo 14) ++
             zip [1, 2,3, 4, 5, 7, 9]
                 [6,11,8,13,10,12,14])
```

Grafo de McGee

El **grafo de McGee** [13] es un grafo no dirigido, cúbico, con 24 vértices y 36 aristas. Tiene importantes propiedades geométricas y topológicas.



La función `grafoMcGee` genera el grafo de McGee.

```
-- | Ejemplo
-- >>> pp $ grafoMcGee
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,
--    20,21,22,23,24]
--   [(1, 2),(1, 13),(1, 24),(2, 3),(2, 9),(3, 4),(3, 20),
```

[13] https://en.wikipedia.org/wiki/McGee_graph

```
--    (4, 5),(4, 16),(5, 6),(5, 12),(6, 7),(6, 23),(7, 8),
--    (7, 19),(8, 9),(8, 15),(9, 10),(10, 11),(10, 22),
--    (11, 12),(11, 18),(12, 13),(13, 14),(14, 15),
--    (14, 21),(15, 16),(16, 17),(17, 18),(17, 24),
--    (18, 19),(19, 20),(20, 21),(21, 22),(22, 23),
--    (23, 24)]
grafoMcGee :: Grafo Int
grafoMcGee =
  creaGrafo [1..24]
            (aristas (grafoCiclo 24) ++
             zip [ 1,2, 3, 4, 5, 6, 7, 8,10,11,14,17]
                 [13,9,20,16,12,23,19,15,22,18,21,24])
```

Grafo Tutte–Coxeter

El **grafo Tutte–Coxeter** [14] es un grafo no dirigido, cúbico, con 30 vértices y 45 aristas. Tiene importantes propiedades geométricas y topológicas.



La función `grafoTutteCoxeter` genera el grafo Tutte–Coxeter.

```
-- | Ejemplo
-- >>> pp $ grafoTutteCoxeter
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,
--    20,21,22,23,24,25,26,27,28,29,30]
```

[14] https://en.wikipedia.org/wiki/Tutte\T1\textendashCoxeter_graph

```
--   [(1, 2),(1, 18),(1, 30),(2, 3),(2, 23),(3, 4),
--    (3, 10),(4, 5),(4, 27),(5, 6),(5, 14),(6, 7),(6, 19),
--    (7, 8),(7, 24),(8, 9),(8, 29),(9, 10),(9, 16),
--    (10, 11),(11, 12),(11, 20),(12, 13),(12, 25),
--    (13, 14),(13, 30),(14, 15),(15, 16),(15, 22),
--    (16, 17),(17, 18),(17, 26),(18, 19),(19, 20),
--    (20, 21),(21, 22),(21, 28),(22, 23),(23, 24),
--    (24, 25),(25, 26),(26, 27),(27, 28),(28, 29),
--    (29, 30)]
grafoTutteCoxeter :: Grafo Int
grafoTutteCoxeter =
  creaGrafo [1..30]
            (aristas (grafoCiclo 30) ++
             zip [ 1, 2, 3, 4, 5, 6, 7, 8, 9,11,12,13,15,17,21]
                 [18,23,10,27,14,19,24,29,16,20,25,30,22,26,28])
```

Grafo de Petersen

El **grafo de Petersen** [15] se define como el grafo de Petersen generalizado $GP_{5,2}$; es decir, es un grafo cúbico formado por los vértices de un pentágono, conectados a los vértices de una estrella de cinco puntas en la que cada nodo es adyacente a los nodos que están a un salto 2 de él. Es usado como ejemplo y como contraejemplo en muchos problemas de la Teoría de grafos.



La función `grafoPetersen` devuelve el grafo de Petersen.

```
-- | Ejemplo
-- >>> pp $ grafoPetersen
-- G [1,2,3,4,5,6,7,8,9,10]
--   [(1, 3),(1, 4),(1, 6),(2, 4),(2, 5),(2, 7),(3, 5),
--    (3, 8),(4, 9),(5, 10),(6, 7),(6, 10),(7, 8),(8, 9),
--    (9, 10)]
grafoPetersen :: Grafo Int
grafoPetersen = grafoPetersenGen 5 2
```

[15] https://en.wikipedia.org/wiki/Petersen_graph

Grafo de Moëbius–Cantor

El **grafo de Moëbius–Cantor** [16] se define como el grafo de Petersen generalizado $GP_{8,3}$; es decir, es un grafo cúbico formado por los vértices de un octógono, conectados a los vértices de una estrella de ocho puntas en la que cada nodo es adyacente a los nodos que están a un salto 3 de él. Al igual que el grafo de Petersen, tiene importantes propiedades que lo hacen ser ejemplo y contraejemplo de muchos problemas de la Teoría de Grafos.



La función `grafoMoebiusCantor` genera el grafo de Moëbius–Cantor

```
-- | Ejemplo
-- >>> pp $ grafoMoebiusCantor
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16]
--   [(1, 4),(1, 6),(1, 9),(2, 5),(2, 7),(2, 10),(3, 6),
--    (3, 8),(3, 11),(4, 7),(4, 12),(5, 8),(5, 13),(6, 14),
--    (7, 15),(8, 16),(9, 10),(9, 16),(10, 11),(11, 12),
--    (12, 13),(13, 14),(14, 15),(15, 16)]
grafoMoebiusCantor :: Grafo Int
grafoMoebiusCantor = grafoPetersenGen 8 3
```

3.5. Definiciones y propiedades

Una vez construida una pequeña fuente de ejemplos, estamos en condiciones de implementar las definiciones sobre grafos en Haskell y ver que funcionan correctamente. Además, comprobaremos que se cumplen las propiedades básicas que se han presentado en el tema *Introducción a la teoría de grafos* de "Matemática discreta" ([4]).

[16] https://en.wikipedia.org/wiki/Moëbius-Kantor_graph

Nota 3.5.1. Se utilizará el tipo abstracto de grafos presentados en la sección 3.2 y se utilizarán las librerias `Data.List` y `Test.QuickCheck`.

3.5.1. Definiciones de grafos

Definición 3.5.1. *El **orden** de un grafo $G = (V, A)$ se define como su número de vértices. Lo denotaremos por $|V(G)|$.*

La función `(orden g)` devuelve el orden del grafo `g`.

```
-- | Ejemplos
-- >>> orden (grafoCiclo 4)
-- 4
-- >>> orden (grafoEstrella 4)
-- 5
orden :: Grafo a -> Int
orden = length . vertices
```

Definición 3.5.2. *El **tamaño** de un grafo $G = (V, A)$ se define como su número de aristas. Lo denotaremos por $|A(G)|$.*

La función `(tamaño g)` devuelve el orden del grafo `g`.

```
-- | Ejemplos
-- >>> tamaño (grafoCiclo 4)
-- 4
-- >>> tamaño grafoPetersen
-- 15
tamaño :: Grafo a -> Int
tamaño = length . aristas
```

Definición 3.5.3. *Diremos que dos aristas a, a' son **incidentes** si tienen intersección no vacía; es decir, si tienen algún vértice en común.*

La función `(sonIncidentes a a')` se verifica si las aristas `a` y `a'` son incidentes.

```
-- | Ejemplos
-- >>> sonIncidentes (1,2) (2,4)
-- True
-- >>> sonIncidentes (1,2) (3,4)
-- False
sonIncidentes :: Eq a => (a,a) -> (a,a) -> Bool
sonIncidentes (u1,u2) (v1,v2) =
  or [u1 == v1, u1 == v2, u2 == v1, u2 == v2]
```

Definición 3.5.4. *Diremos que una arista de un grafo G es un **lazo** si va de un vértice en sí mismo.*

La función `(lazos g)` devuelve los lazos del grafo `g`.

```
-- | Ejemplos
-- >>> lazos (creaGrafo [1,2] [(1,1),(1,2),(2,2)])
-- [(1,1),(2,2)]
-- >>> lazos (grafoCiclo 5)
-- []
lazos :: Eq a => Grafo a -> [(a,a)]
lazos g = [(u,v) | (u,v) <- aristas g, u == v]
```

La función `(esLazo a)` se verifica si la arista `a` es un lazo.

```
-- | Ejemplos
-- >>> esLazo (4,4)
-- True
-- >>> esLazo (1,2)
-- False
esLazo :: Eq a => (a,a) -> Bool
esLazo (u,v) = u == v
```

Definición 3.5.5. *Dado un grafo $G = (V, A)$, fijado un vértice $v \in V$, al conjunto de vértices que son adyacentes a v lo llamaremos* ***entorno*** *de v y lo denotaremos por $N(v) = \{u \in V | (u, v) \in A\}$.*

La función `(entorno g v)` devuelve el entorno del vértice `v` en el grafo `g`.

```
-- | Ejemplo
-- >>> entorno (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 2
-- [1,3]
entorno :: Eq a => Grafo a -> a -> [a]
entorno = adyacentes
```

Definición 3.5.6. *Sea $G = (V, A)$ un grafo. El* ***grado*** *(o* ***valencia****) de $v \in V$ es $grad(v) = |N(v)|$.*

La función `(grado g v)` devuelve el grado del vértice `v` en el grafo `g`.

```
-- | Ejemplos
-- >>> grado (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 2
-- 2
-- >>> grado (grafoEstrella 5) 1
-- 5
grado :: Eq a => Grafo a -> a -> Int
grado g v = length (entorno g v)
```

Definición 3.5.7. *Un vértice v de un grafo es* ***aislado*** *si su grado es 0.*

La función `(esAislado g v)` se verifica si el vértice `v` es aislado en el grafo `g`.

```
-- | Ejemplos
-- >>> esAislado (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 4
-- True
-- >>> esAislado (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 3
-- False
esAislado :: Eq a => Grafo a -> a -> Bool
esAislado g v = grado g v == 0
```

Definición 3.5.8. *Un grafo es **regular** si todos sus vértices tienen el mismo grado.*

La función `(esRegular g)` se verifica si el grafo g es regular.

```
-- | Ejemplos
-- >>> esRegular (creaGrafo [1..3] [(1,2),(1,3),(2,3)])
-- True
-- >>> esRegular (creaGrafo [1..4] [(1,2),(1,3),(2,4)])
-- False
esRegular :: Eq a => Grafo a -> Bool
esRegular g = all (==x) xs
  where (x:xs) = [grado g v | v <- vertices g]
```

Definición 3.5.9. *Dado un grafo $G = (V, A)$ llamamos **valencia mínima** o **grado mínimo** de G al valor $\delta(G) = \text{mín}\{grad(v)|v \in V\}$*

La función `(valenciaMin g)` devuelve la valencia mínima del grafo g.

```
-- | Ejemplo
-- >>> valenciaMin (creaGrafo [1..4] [(1,2),(1,3),(2,4)])
-- 1
valenciaMin :: Ord a => Grafo a -> Int
valenciaMin g = minimum [grado g v | v <- vertices g]
```

Definición 3.5.10. *Dado un grafo $G = (V, A)$ llamamos **valencia máxima** o **grado máximo** de G al valor $\delta(G) = \text{máx}\{grad(v)|v \in V\}$*

La función `(valenciaMax g)` devuelve la valencia máxima del grafo g.

```
-- | Ejemplo
-- >>> valenciaMax (creaGrafo [1..4] [(1,2),(1,3),(2,4)])
-- 2
valenciaMax :: Ord a => Grafo a -> Int
valenciaMax g = maximum [grado g v | v <- vertices g]
```

Definición 3.5.11. *Se dice que un grafo es simple si no contiene lazos ni aristas repetidas.*

La función `(esSimple g)` se verifica si g es un grafo simple.

```
-- | Ejemplos
-- >>> esSimple (creaGrafo [1..3] [(1,2),(1,3),(2,3)])
-- True
-- >>> esSimple (creaGrafo [1..3] [(1,1),(1,2),(2,3)])
-- False
esSimple :: Ord a => Grafo a -> Bool
esSimple g =
  and [not ((x,x) `aristaEn` g) | x <- vertices g]
```

Definición 3.5.12. *Sea G un grafo. Llamamos* ***secuencia de grados*** *de G a la lista de grados de sus vértices. La secuencia se suele presentar en orden decreciente: $d_1 \geq d_2 \geq \cdots \geq d_n$.*

La función `(secuenciaGrados g)` devuelve la secuencia de los grados del grafo `g` en orden decreciente.

```
-- | Ejemplo
-- >>> secuenciaGrados (creaGrafo [1..5] [(1,2),(1,3),(1,4),(2,4)])
-- [3,2,2,1,0]
secuenciaGrados :: Eq a => Grafo a -> [Int]
secuenciaGrados g = sortBy (flip compare) [grado g v | v <- vertices g]
```

Nota 3.5.2. ¿Qué listas de n números enteros son secuencias de grafos de n vértices?

- Si $\sum_{i=1}^{n} d_i$ es impar, no hay ninguno.
- Si $\sum_{i=1}^{n} d_i$ es par, entonces siempre hay un grafo con esa secuencia de grados (aunque no necesariamente simple).

Definición 3.5.13. *Una* ***secuencia gráfica*** *es una lista de número enteros no negativos que es la secuencia de grados para algún grafo simple.*

La función `(secuenciaGrafica ss)` se verifica si existe algún grafo con la secuencia de grados `ss`.

```
-- | Ejemplos
-- >>> secuenciaGrafica [3,2,2,1,0]
-- True
-- >>> secuenciaGrafica [3,2,2,2]
-- False
secuenciaGrafica :: [Int] -> Bool
secuenciaGrafica ss = even (sum ss) && all p ss
  where p s = s >= 0 && s <= length ss
```

Definición 3.5.14. *Dado un grafo $G = (V, A)$, diremos que $G' = (V', A')$ es un* ***subgrafo*** *de G si $V' \subseteq V$ y $A' \subseteq A$.*

La función `(esSubgrafo g' g)` se verifica si `g'` es un subgrafo de `g`.

```
-- |Ejemplos
-- >>> esSubgrafo (bipartitoCompleto 3 2) (bipartitoCompleto 3 3)
-- True
-- >>> esSubgrafo (grafoEstrella 4) (grafoEstrella 5)
-- True
-- >>> esSubgrafo (completo 5) (completo 4)
-- False
-- >>> esSubgrafo (completo 3) (completo 4)
-- True
esSubgrafo :: Ord a => Grafo a -> Grafo a -> Bool
esSubgrafo g' g =
  vertices g' `esSubconjunto` vertices g &&
  aristas g'  `esSubconjunto` aristas g
```

Definición 3.5.15. *Si $G' = (V', A')$ es un subgrafo de $G = (V, A)$ tal que $V' = V$, diremos que G' es un* ***subgrafo maximal***, ***grafo recubridor*** *o* ***grafo de expansión*** *(en inglés,* spanning grah*) de G.*

La función `(esSubgrafoMax g' g)` se verifica si `g'` es un subgrafo maximal de `g`.

```
-- | Ejemplos
-- >>> esSubgrafoMax (grafoRueda 3) (grafoRueda 4)
-- False
-- >>> esSubgrafoMax (grafoCiclo 4) (grafoRueda 4)
-- True
-- >>> esSubgrafoMax (creaGrafo [1..3] [(1,2)]) (grafoCiclo 3)
-- True
-- >>> esSubgrafoMax (creaGrafo [1..2] [(1,2)]) (grafoCiclo 3)
-- False
esSubgrafoMax :: Ord a => Grafo a -> Grafo a -> Bool
esSubgrafoMax g' g =
  g' `esSubgrafo` g && conjuntosIguales (vertices g') (vertices g)
```

Definición 3.5.16. *Sean $G' = (V', A')$, $G = (V, A)$ dos grafos si $V' \subset V$, o $A' \subset A$, se dice que G' es un* ***subgrafo propio*** *de G, y se denota por $G' \subset G$.*

La función `(esSubgrafoPropio g' g)` se verifica si `g'` es un subgrafo propio de `g`.

```
-- | Ejemplos
-- >>> esSubgrafoPropio (grafoRueda 3) (grafoRueda 4)
-- True
-- >>> esSubgrafoPropio (grafoRueda 4) (grafoCiclo 5)
-- False
-- >>> esSubgrafoPropio (creaGrafo [1..3] [(1,2)]) (grafoCiclo 3)
-- True
-- >>> esSubgrafoPropio (creaGrafo [1..2] [(1,2)]) (grafoCiclo 3)
```

```
-- True
esSubgrafoPropio :: Ord a => Grafo a -> Grafo a -> Bool
esSubgrafoPropio g' g =
  esSubgrafo g' g &&
  (not (conjuntosIguales (vertices g) (vertices g')) ||
   not (conjuntosIguales (aristas g) (aristas g')))
```

3.5.2. Propiedades de grafos

Teorema 3.5.17 (Lema del apretón de manos). *En todo grafo simple el número de vértices de grado impar es par o cero.*

Vamos a comprobar que se verifica el lema del apretón de manos utilizando la función `prop_LemaApretonDeManos`.

```
ghci> quickCheck prop_LemaApretonDeManos
+++ OK, passed 100 tests.
```

```
prop_LemaApretonDeManos :: Grafo Int -> Bool
prop_LemaApretonDeManos g =
  even (length (filter odd [grado g v | v <- vertices g]))
  where g' = creaGrafo (vertices g)
                       [(x,y) | (x,y) <- aristas g, x /= y]
```

Teorema 3.5.18 (Havel–Hakimi). *Si $n > 1$ y $D = [d_1, \ldots, d_n]$ es una lista de enteros, entonces D es secuencia gráfica si y sólo si la secuencia D' obtenida borrando el mayor elemento d_{max} y restando 1 a los siguientes d_{max} elementos más grandes es gráfica.*

Vamos a comprobar que se verifica el teorema de Havel–Hakimi utilizando la función `prop_HavelHakimi`.

```
ghci> quickCheck prop_HavelHakimi
+++ OK, passed 100 tests.
```

```
prop_HavelHakimi :: [Int] -> Bool
prop_HavelHakimi [] = True
prop_HavelHakimi (s:ss) =
  not (secuenciaGrafica (s:ss) && not (esVacio ss)) ||
  secuenciaGrafica (map (\x -> x-1) (take s ss) ++ drop s ss)
```

3.5.3. Operaciones y propiedades sobre grafos

Eliminación de una arista

Definición 3.5.19. *Sea $G = (V, A)$ un grafo y sea $(u, v) \in A$. Definimos el grafo $G \setminus (u, v)$ como el subgrafo de G, $G' = (V', A')$, con $V' = V$ y $A' = A \setminus \{(u, v)\}$. Esta*

operación se denomina ***eliminar una arista****.*

La función `(eliminaArista g a)` elimina la arista `a` del grafo `g`.

```
-- | Ejemplos
-- >>> eliminaArista (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) (1,4)
-- G [1,2,3,4] [(1,2),(2,4)]
-- >>> eliminaArista (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) (4,1)
-- G [1,2,3,4] [(1,2),(2,4)]
-- >>> eliminaArista (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) (4,3)
-- G [1,2,3,4] [(1,2),(1,4),(2,4)]
eliminaArista :: Ord a => Grafo a -> (a,a) -> Grafo a
eliminaArista g (a,b) =
  creaGrafo (vertices g)
            (aristas g \\ [(a,b),(b,a)])
```

La función `(eliminaLazos g)` devuelve un nuevo grafo, obtenido a partir del grafo `g` eliminando todas las aristas que fueran lazos.

```
-- | Ejemplos
-- >>> eliminaLazos (creaGrafo [1,2] [(1,1),(1,2),(2,2)])
-- G [1,2] [(1,2)]
-- >>> grafoCiclo 5
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
-- >>> eliminaLazos (grafoCiclo 5)
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
eliminaLazos :: Ord a => Grafo a -> Grafo a
eliminaLazos g = creaGrafo (vertices g)
                           [(x,y) | (x,y) <- aristas g, x /= y]
```

Eliminación un vértice

Definición 3.5.20. *Sea $G = (V, A)$ un grafo y sea $v \in V$. Definimos el grafo $G \setminus v$ como el subgrafo de G, $G' = (V', A')$, con $V' = V \setminus \{v\}$ y $A' = A \setminus \{a \in A | v$ es un extremo de $a\}$. Esta operación se denomina* ***eliminar un vértice****.*

La función `(eliminaVertice g v)` elimina el vértice `v` del grafo `g`.

```
-- | Ejemplos
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) 1
-- G [2,3,4] [(2,4)]
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) 4
-- G [1,2,3] [(1,2)]
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) 3
-- G [1,2,4] [(1,2),(1,4),(2,4)]
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(1,3)]) 1
-- G [2,3,4] []
eliminaVertice :: Ord a => Grafo a -> a -> Grafo a
```

```
eliminaVertice g v =
  creaGrafo (vertices g \\ [v])
            (as \\ [(a,b) | (a,b) <- as, a == v || b == v])
  where as = aristas g
```

Suma de aristas

Definición 3.5.21. *Sea $G = (V, A)$ un grafo y sean $u, v \in V$ tales que $(u, v), (v, u) \notin A$. Definimos el grafo $G + (u, v)$ como el grafo $G' = (V, A \cup \{(u, v)\})$. Esta operación se denomina* ***suma de una arista****.*

La función `(sumaArista g a)` suma la arista `a` al grafo `g`.

```
-- | Ejemplos
-- >>> grafoCiclo 5
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
-- >>> sumaArista (grafoCiclo 5) (1,3)
-- G [1,2,3,4,5] [(1,2),(1,3),(1,5),(2,3),(3,4),(4,5)]
sumaArista :: Ord a => Grafo a -> (a,a) -> Grafo a
sumaArista g a =
  creaGrafo (vertices g) (a : aristas g)
```

Suma de vértices

Definición 3.5.22. *Sea $G = (V, A)$ un grafo y sea $v \notin V$. Definimos el grafo $G + v$ como el grafo $G' = (V', A')$, donde $V' = V \cup \{v\}$, $A' = A \cup \{(u, v) | u \in V\}$. Esta operación se denomina* ***suma de un vértice****.*

La función `(sumaVertice g a)` suma el vértice `a` al grafo `g`.

```
-- | Ejemplo
-- >>> grafoCiclo 3
-- G [1,2,3] [(1,2),(1,3),(2,3)]
-- >>> sumaVertice (grafoCiclo 3) 4
-- G [1,2,3,4] [(1,2),(1,3),(1,4),(2,3),(2,4),(3,4)]
sumaVertice :: Ord a => Grafo a -> a -> Grafo a
sumaVertice g v =
  creaGrafo (v : vs) (aristas g ++ [(u,v) | u <- vs])
  where vs = vertices g
```

Propiedad de los grafos completos

Proposición 3.5.23. *La familia de grafos completos K_n verifica que $K_n = K_{n-1} + n$.*

Vamos a ver que se cumple la propiedad utilizando la función `prop_completos`.

```
ghci> quickCheck prop_completos
+++ OK, passed 100 tests.
```

```
prop_completos :: Int -> Property
prop_completos n = n >= 2 ==>
   completo n == sumaVertice (completo (n-1)) n
```

A partir de esta propiedad, se puede dar una definición alternativa de la función `(completo n)`, que devuelve el grafo completo K_n. La nueva función será `(completo2 n)`.

```
completo2 :: Int -> Grafo Int
completo2 0 = grafoNulo
completo2 n = sumaVertice (completo2 (n-1)) n
```

Vamos a comprobar la equivalencia de ambas deficiones:

```
ghci> prop_EquiCompleto
True
```

```
prop_EquiCompleto :: Bool
prop_EquiCompleto = and [completo n == completo2 n | n <- [0..50]]
```

Vamos a ver cuál de las definiciones es más eficiente:

```
ghci> tamaño (completo 100)
4950
(1.17 secs, 0 bytes)
ghci> tamaño (completo2 100)
4950
(15.49 secs, 61,756,056 bytes)
```

La función `(completo n)` es más eficiente, así que será la que seguiremos utilizando.

Suma de grafos

Definición 3.5.24. *Sean $G = (V, A), G' = (V', A)$ dos grafos. Definimos el grafo suma de G y G' como el grafo $G + G' = (V \cup V', A \cup A' \cup \{(u, v) | u \in V, v \in V'\})$. Esta operación se denomina* ***suma de grafos****.*

La función `(sumaGrafos g g')` suma los grafos `g` y `g'`.

```
-- | Ejemplo
-- >>> let g1 = creaGrafo [1..3] [(1,1),(1,3),(2,3)]
-- >>> let g2 = creaGrafo [4..6] [(4,6),(5,6)]
-- >>> pp $ sumaGrafos g1 g2
-- G [1,2,3,4,5,6]
--   [(1, 1),(1, 3),(1, 4),(1, 5),(1, 6),(2, 3),(2, 4),
--    (2, 5),(2, 6),(3, 4),(3, 5),(3, 6),(4, 6),(5, 6)]
sumaGrafos :: Ord a => Grafo a -> Grafo a -> Grafo a
```

```
sumaGrafos g1 g2 =
  creaGrafo (vs1 `union` vs2)
            (aristas g1 `union`
             aristas g2 `union`
             [(u,v) | u <- vs1, v <- vs2])
  where vs1 = vertices g1
        vs2 = vertices g2
```

Unión de grafos

Definición 3.5.25. *Sean $G = (V, A), G' = (V', A)$ dos grafos. Definimos el grafo unión de G y G' como el grafo $G \cup H = (V \cup V', A \cup A')$. Esta operación se denomina **unión de grafos**.*

La función `(unionGrafos g g')` une los grafos `g` y `g'`. Por ejemplo,

```
-- | Ejemplo
-- >>> let g1 = creaGrafo [1..3] [(1,1),(1,3),(2,3)]
-- >>> let g2 = creaGrafo [4..6] [(4,6),(5,6)]
-- >>> unionGrafos g1 g2
-- G [1,2,3,4,5,6] [(1,1),(1,3),(2,3),(4,6),(5,6)]
unionGrafos :: Ord a => Grafo a -> Grafo a -> Grafo a
unionGrafos g1 g2 =
  creaGrafo (vertices g1 `union` vertices g2)
            (aristas  g1 `union` aristas  g2)
```

Grafo complementario

Definición 3.5.26. *Dado un grafo $G = (V, A)$ se define el **grafo complementario** de G como $\overline{G} = (V, \overline{A})$, donde $\overline{A} = \{(u, v) | u, v \in V, (u, v) \notin A\}$.*

La función `(grafoComplementario g)` devuelve el grafo complementario de `g`.

```
-- | Ejemplo
-- >>> grafoComplementario (creaGrafo [1..3] [(1,1),(1,3),(2,3)])
-- G [1,2,3] [(1,2),(2,2),(3,3)]
grafoComplementario :: Ord a => Grafo a -> Grafo a
grafoComplementario  g =
  creaGrafo vs
            [(u,v)| u <- vs, v <- vs, u <= v, not ((u,v) `aristaEn` g)]
  where vs = vertices g
```

Definición 3.5.27. *Dado un grafo, diremos que es **completo** si su complementario sin lazos no tiene aristas.*

La función `(esCompleto g)` se verifica si el grafo `g` es completo.

```
-- Ejemplos
-- >>> esCompleto (completo 4)
-- True
-- >>> esCompleto (grafoCiclo 5)
-- False
esCompleto :: Ord a => Grafo a -> Bool
esCompleto g =
    tamaño (eliminaLazos (grafoComplementario g)) == 0
```

3.6. Morfismos de grafos

Llegados a este punto, es importante resaltar que un grafo se define como una entidad matemática abstracta; es evidente que lo importante de un grafo no son los nombres de sus vértices ni su representación gráfica. La propiedad que caracteriza a un grafo es la forma en que sus vértices están unidos por las aristas.
A priori, los siguientes grafos pueden parecer distintos:



Sin embargo, ambos grafos son grafos de cinco vértices que tienen las mismas relaciones de vecindad entre sus nodos. En esta sección estudiaremos estas relaciones que establecen las aristas entre los vértices de un grafo y presentaremos algoritmos que nos permitan identificar cuándo dos grafos se pueden relacionar mediante aplicaciones entre sus vértices cumpliendo ciertas caracterísicas.

3.6.1. Morfismos

Definición 3.6.1. *Si f es una función entre dos grafos $G = (V, A)$ y $G' = (V', A')$, diremos que* ***conserva la adyacencia*** *si $\forall u, v \in V$ se verifica que si $(u, v) \in A$, entonces $(f(u), f(v)) \in A'$.*

La función `(conservaAdyacencia g h f)` se verifica si la función `f` entre los grafos `g` y `h` conserva las adyacencias.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1..4] [(1,2),(2,3),(3,4)]
-- >>> let g2 = creaGrafo [1..4] [(1,2),(2,3),(2,4)]
-- >>> let g3 = creaGrafo [4,6..10] [(4,8),(6,8),(8,10)]
```

```
-- >>> conservaAdyacencia g1 g3 [(1,4),(2,6),(3,8),(4,10)]
-- False
-- >>> conservaAdyacencia g2 g3 [(1,4),(2,8),(3,6),(4,10)]
-- True
conservaAdyacencia :: (Ord a, Ord b) =>
                      Grafo a -> Grafo b -> Funcion a b -> Bool
conservaAdyacencia g h f =
  and [(imagen f x,imagen f y) `aristaEn` h | (x,y) <- aristas g]
```

Definición 3.6.2. *Dados dos grafos simples $G = (V, A)$ y $G' = (V', A')$, un* ***morfismo*** *entre G y G' es una función $\phi : V \to V'$ que conserva las adyacencias.*

La función `(esMorfismo g h vvs)` se verifica si la función representada por `vvs` es un morfismo entre los grafos `g` y `h`.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,2),(3,2)]
-- >>> let g2 = creaGrafo [4,5,6] [(4,6),(5,6)]
-- >>> esMorfismo g1 g2 [(1,4),(2,6),(3,5)]
-- True
-- >>> esMorfismo g1 g2 [(1,4),(2,5),(3,6)]
-- False
-- >>> esMorfismo g1 g2 [(1,4),(2,6),(3,5),(7,9)]
-- False
esMorfismo :: (Ord a,Ord b) => Grafo a -> Grafo b ->
              Funcion a b -> Bool
esMorfismo g1 g2 f =
  esFuncion (vertices g1) (vertices g2) f &&
  conservaAdyacencia g1 g2 f
```

La función `(morfismos g h)` devuelve todos los posibles morfismos entre los grafos `g` y `h`.

```
-- | Ejemplos
-- >>> grafoCiclo 3
-- G [1,2,3] [(1,2),(1,3),(2,3)]
-- >>> let g = creaGrafo [4,6] [(4,4),(6,6)]
-- >>> morfismos (grafoCiclo 3) g
-- [[(1,4),(2,4),(3,4)],[(1,6),(2,6),(3,6)]]
-- >>> morfismos g (grafoCiclo 3)
-- []
morfismos :: (Ord a, Ord b) => Grafo a -> Grafo b -> [[(a,b)]]
morfismos g h =
  [f | f <- funciones (vertices g) (vertices h)
     , conservaAdyacencia g h f]
```

3.6.2. Complejidad del problema de homomorfismo de grafos

En general en computación un problema algorítmico consta de un conjunto I de todas las posibles entradas para el problema, llamado el conjunto de instancias, y de una pregunta Q sobre esas instancias. Resolver uno de estos problemas consiste en desarrollar un algoritmo cuya entrada es una instancia del problema y cuya salida es una respuesta a la pregunta del problema.
Decimos que un problema es de decisión cuando las posibles respuestas a la pregunta son "sí" o "no". Un ejemplo de problema de este tipo sería, dados dos grafos, decidir si estos son homomorfos o no.
A la hora de implementar un algoritmo que dé solución al problema que estemos considerando, es importante tener en cuenta el número de pasos que efectúa hasta encontrar una respuesta. Un **algoritmo de tiempo polinomial** se define como aquel con función de complejidad temporal en $O(p(n))$ para alguna función polinómica p, donde n denota el tamaño de la entrada. Cualquier algoritmo cuya función de complejidad temporal no pueda ser acotada de esta manera, se denomina **algoritmo de tiempo exponencial**.
La mayoría de los algoritmos de tiempo exponencial son simples variaciones de una búsqueda exhaustiva, mientras que los algoritmos de tiempo polinomial, usualmente se obtienen mediante un análisis más profundo de la estructura del problema. En la Teoría de la complejidad computacional, existe el consenso de que un problema no está "bien resuelto" hasta que se conozca un algoritmo de tiempo polinomial que lo resuelva. Por tanto, nos referiremos a un problema como "intratable", si es tan difícil que no existe algoritmo de tiempo polinomial capaz de resolverlo.
Los problemas de decisión para los que existen algoritmos polinomiales constituyen la clase P y son llamados **polinomiales**. Un problema de decisión es **no-determinístico polinomial** cuando cualquier instancia que produce respuesta "sí" posee una comprobación de correctitud (también llamada certificado) verificable en tiempo polinomia en el tamaño de la instancia. Estos problemas de decisión se dice que pertenecen a la clase NP. Claramente, $P \subseteq NP$. Sin embargo, no se sabe si esta inclusión es estricta: uno de los principales problemas abiertos en Teoría de la computación es saber si $P \neq NP$.
Una reducción es una transformación de un problema en otro problema. Intuitivamente, un problema Q puede ser reducido a otro problema Q', si cualquier instancia del problema Q puede ser "fácilmente" expresada como una instancia del problema Q', y cuya solución proporcione una solución para la instancia de Q.
Las reducciones en tiempo polinomial nos dotan de elementos para probar, de una manera formal, que un problema es al menos tan difícil que otro, con una diferencia de un factor polinomial. Estas son esenciales para definir a los problemas

NP-completos, además de ayudar a comprender los mismos.
La clase de los problemas NP-completos contiene a los problemas más difíciles en NP, en el sentido de que son los que estén más lejos de estar en P. Debido a que el problema $P = NP$ no ha sido resuelto, el hecho de reducir un problema B, a otro problema A, indicaría que no se conoce solución en tiempo polinomial para A. Esto es debido a que una solución en tiempo polinomial para A, tendría como consecuencia la existencia de una solución polinomial para B. De manera similar, debido a que todos los problemas NP pueden ser reducidos a este conjunto, encontrar un problema NP-completo que pueda ser resuelto en un tiempo polinomial significaría que $P = NP$.
El problema de decidir si dos grafos son homomorfos es un problema de decisión NP-completo. La NP-completitud del problema se demuestra mediante la reducción de este problema al **problema de la clique**, [17].
En el proceso de recopilación de imformación acerca de la Teoría de la complejidad, se ha trabajado con dos fuentes:

- la tesis de licenciatura de Pablo Burzyn "Complejidad computacional en problemas de modificación de aristas en grafos" ([2]) y
- el artículo "Teoría de la complejidad computacional" ([12]) de la Wikipedia.

3.6.3. Isomorfismos

Definición 3.6.3. *Dados dos grafos simples $G = (V, A)$ y $G' = (V', A')$, un **isomorfismo** entre G y G' es un morfismo biyectivo cuyo inverso es morfismo entre G' y G.*

La función `(esIsomorfismo g h f)` se verifica si la aplicación `f` es un isomorfismo entre los grafos `g` y `h`.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> esIsomorfismo g1 g2 [(1,2),(2,4),(3,6)]
-- True
-- >>> esIsomorfismo g1 g2 [(1,2),(2,2),(3,6)]
-- False
esIsomorfismo :: (Ord a,Ord b) =>
                 Grafo a -> Grafo b -> Funcion a b -> Bool
esIsomorfismo g h f =
  esBiyectiva vs1 vs2 f      &&
  esMorfismo g h f           &&
  esMorfismo h g (inversa f)
  where vs1 = vertices g
        vs2 = vertices h
```

[17] https://es.wikipedia.org/wiki/Problema_de_la_clique

La función `(isomorfismos1 g h)` devuelve todos los isomorfismos posibles entre los grafos `g` y `h`.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfismos1 g1 g2
-- [[(1,2),(2,4),(3,6)]]
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfismos1 g3 g4
-- []
-- >>> let g5 = creaGrafo [1,2,3] [(1,1),(2,2),(3,3)]
-- >>> let g6 = creaGrafo [2,4,6] [(2,2),(4,4),(6,6)]
-- >>> pp $ isomorfismos1 g5 g6
-- [[(1, 2),(2, 4),(3, 6)],[(1, 4),(2, 2),(3, 6)],
--  [(1, 6),(2, 4),(3, 2)],[(1, 4),(2, 6),(3, 2)],
--  [(1, 6),(2, 2),(3, 4)],[(1, 2),(2, 6),(3, 4)]]
isomorfismos1 :: (Ord a,Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos1 g h =
  [f | f <- biyecciones vs1 vs2
     , conservaAdyacencia g h f
     , conservaAdyacencia h g (inversa f)]
  where vs1 = vertices g
        vs2 = vertices h
```

Definición 3.6.4. *Dos grafos G y H se dicen **isomorfos** si existe algún isomorfismo entre ellos.*

La función `isomorfos1 g h` se verifica si los grafos `g` y `h` son isomorfos.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfos1 g1 g2
-- True
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfos1 g3 g4
-- False
isomorfos1 :: (Ord a,Ord b) => Grafo a -> Grafo b -> Bool
isomorfos1 g = not . null . isomorfismos1 g
```

Nota 3.6.1. Al tener Haskell una evaluación perezosa, la función `(isomorfos g h)` no necesita generar todos los isomorfismos entre los grafos `g` y `h`.

Definición 3.6.5. *Sea $G = (V, A)$ un grafo. Un **invariante por isomorfismos** de G es una propiedad de G que tiene el mismo valor para todos los grafos que son isomorfos a él.*

```
esInvariantePorIsomorfismos ::
  Eq a => (Grafo Int -> a) -> Grafo Int -> Grafo Int -> Bool
esInvariantePorIsomorfismos p g h =
  isomorfos g h --> (p g == p h)
  where (-->) = (<=)
```

Teorema 3.6.6. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, se verifica que $|V(G)| = |V(G')|$; es decir, el orden de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos orden)
+++ OK, passed 100 tests.
```

Teorema 3.6.7. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, se verifica que $|A(G)| = |A(G')|$; es decir, el tamaño de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos tamaño)
+++ OK, passed 100 tests.
```

Teorema 3.6.8. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, tienen la misma secuencia de grados; es decir, la secuencia de grados de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos secuenciaGrados)
+++ OK, passed 100 tests.
```

A partir de las propiedades que hemos demostrado de los isomorfismos, vamos a dar otra definición equivalente de las funciones `(isomorfismos1 g h)` y `(isomorfos1 g h)`.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfismos2 g1 g2
-- [[(1,2),(2,4),(3,6)]]
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfismos2 g3 g4
-- []
-- >>> let g5 = creaGrafo [1,2,3] [(1,1),(2,2),(3,3)]
```

```
-- >>> let g6 = creaGrafo [2,4,6] [(2,2),(4,4),(6,6)]
-- >>> pp $ isomorfismos2 g5 g6
-- [[(1, 2),(2, 4),(3, 6)],[(1, 4),(2, 2),(3, 6)],
--  [(1, 6),(2, 4),(3, 2)],[(1, 4),(2, 6),(3, 2)],
--  [(1, 6),(2, 2),(3, 4)],[(1, 2),(2, 6),(3, 4)]]
isomorfismos2 :: (Ord a, Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos2 g h
  | orden g  /= orden h  = []
  | tamaño g /= tamaño h = []
  | secuenciaGrados g /= secuenciaGrados h = []
  | otherwise = [f | f <- biyecciones vs1 vs2
                   , conservaAdyacencia g h f]
  where vs1 = vertices g
        vs2 = vertices h

-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfos2 g1 g2
-- True
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfos2 g3 g4
-- False
isomorfos2 :: (Ord a, Ord b) => Grafo a -> Grafo b -> Bool
isomorfos2 g =
  not. null . isomorfismos2 g
```

```
isomorfismos3 :: (Ord a, Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos3 g h
  | orden g  /= orden h  = []
  | tamaño g /= tamaño h = []
  | secuenciaGrados g /= secuenciaGrados h = []
  | otherwise = filter (conservaAdyacencia g h) (posibles g h)

verticesPorGrados g =
    [filter (p n) (vertices g) | n <- nub (secuenciaGrados g)]
    where p m v = grado g v == m

aux1 [] _ = []
aux1 (xs:xss) (ys:yss) = (map (zip xs) (permutations ys)):aux1 xss yss

aux2 [] = []
aux2 (xss:[]) = xss
aux2 (xss:yss:[]) = [xs ++ ys | xs <- xss, ys <- yss]
aux2 (xss:yss:xsss) =
    aux2 ([xs ++ ys | xs <- xss, ys <- yss]:xsss)
```

```
posibles g h =
    aux2 (aux1 (verticesPorGrados g) (verticesPorGrados h))
```

Vamos a comparar la eficiencia entre ambas definiciones

Nota 3.6.2. La nueva definición de `(isomorfismos3 g h)` es la más eficiente con grafos "poco regulares", sin embargo, cuando todos los vértices tienen el mismo grado, `(isomorfismos2 g h)` sigue siendo mejor opción (aunque no hay mucha diferencia en el coste computacional).

```
ghci> let n = 6 in (length (isomorfismos1 (completo n) (completo n)))
720
(0.18 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos2 (completo n) (completo n)))
720
(0.18 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos3 (completo n) (completo n)))
720
(0.18 secs, 26,123,800 bytes)

ghci> let n = 6 in (length (isomorfismos1 (grafoCiclo n) (grafoCiclo n)))
12
(0.04 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos2 (grafoCiclo n) (grafoCiclo n)))
12
(0.04 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos3 (grafoCiclo n) (grafoCiclo n)))
12
(0.04 secs, 0 bytes)

ghci> length (isomorfismos1 (grafoCiclo 6) (completo 7))
0
(0.00 secs, 0 bytes)
ghci> length (isomorfismos2 (grafoCiclo 6) (completo 7))
0
(0.01 secs, 0 bytes)
ghci> length (isomorfismos3 (grafoCiclo 6) (completo 7))
0
(0.00 secs, 0 bytes)

ghci> isomorfos1 (completo 10) (grafoCiclo 10)
False
(51.90 secs, 12,841,861,176 bytes)
ghci> isomorfos2 (completo 10) (grafoCiclo 10)
False
(0.00 secs, 0 bytes)
ghci> isomorfos3 (completo 10) (grafoCiclo 10)
False
```

```
(0.00 secs, 0 bytes)

ghci> isomorfos1 (grafoCiclo 10) (grafoRueda 10)
False
(73.90 secs, 16,433,969,976 bytes)
ghci> isomorfos2 (grafoCiclo 100) (grafoRueda 100)
False
(0.00 secs, 0 bytes)
ghci> isomorfos3 (grafoCiclo 100) (grafoRueda 100)
False
(0.00 secs, 0 bytes)

ghci> length (isomorfismos2 (grafoRueda 10) (grafoRueda 10))
18
(101.12 secs, 23,237,139,992 bytes)
ghci> length (isomorfismos3 (grafoRueda 10) (grafoRueda 10))
18
(44.67 secs, 9,021,442,440 bytes)
```

Nota 3.6.3. Cuando los grafos son isomorfos, comprobar que tienen el mismo número de vértices, el mismo número de aristas y la misma secuencia gráfica no requiere mucho tiempo ni espacio, dando lugar a costes muy similares entre los dos pares de definiciones. Sin embargo, cuando los grafos tienen el mismo número de vértices y fallan en alguna de las demás propiedades, el resultado es muy costoso para la primera definición mientras que es inmediato con la segunda.
A partir de ahora utilizaremos la función `isomorfismos2` para calcular los isomorfismos entre dos grafos y la función `isomorfos` para determinar si dos grafos son isomorfos, de modo que las renombraremos por `isomorfismos` y `isomorfismos` respectivamente.

```
isomorfismos :: (Ord a,Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos = isomorfismos2

isomorfos :: (Ord a,Ord b) => Grafo a -> Grafo b -> Bool
isomorfos = isomorfos2
```

3.6.4. Automorfismos

Definición 3.6.9. *Dado un grafo simple $G = (V, A)$, un **automorfismo** de G es un isomorfismo de G en sí mismo.*

La función `(esAutomorfismo g f)` se verifica si la aplicación `f` es un automorfismo de `g`.

```
-- | Ejemplos
-- >>> let g = creaGrafo [1,2,3] [(1,2),(1,3)]
```

```
-- >>> esAutomorfismo g [(1,1),(2,3),(3,2)]
-- True
-- >>> esAutomorfismo g [(1,2),(2,3),(3,1)]
-- False
esAutomorfismo :: Ord a => Grafo a -> Funcion a a -> Bool
esAutomorfismo g = esIsomorfismo g g
```

La función `(automorfismos g)` devuelve la lista de todos los posibles automorfismos en g.

```
-- | Ejemplo
-- >>> automorfismos (creaGrafo [1,2,3] [(1,2),(1,3)])
-- [[(1,1),(2,2),(3,3)],[(1,1),(2,3),(3,2)]]
automorfismos :: Ord a => Grafo a -> [Funcion a a]
automorfismos g = isomorfismos1 g g
```

Nota 3.6.4. Cuando trabajamos con automorfismos, es mejor utilizar en su definición la función `isomorfismos1` en vez de `isomorfismos2`, pues ser isomorfos es una relación reflexiva; es decir, un grafo siempre es isomorfo a sí mismo.

3.7. Caminos en grafos

Una de las aplicaciones de la teoría de grafos es la determinación de trayectos o recorridos en una red de transporte o de distribución de productos. Así, si cada vértice representa un punto de interés y cada arista representa una conexión entre dos puntos, usando grafos como modelos, podemos simplificar el problema de encontrar la ruta más ventajosa en cada caso.

3.7.1. Definición de camino

Definición 3.7.1. *Sea $G = (V, A)$ un grafo simple y sean $u, v \in V$ dos vértices. Un* ***camino*** *entre u y v es una sucesión de vértices de G: $u = v_0, v_1, v_2, \ldots, v_{k-1}, v_k = v$ donde $\forall i \in \{0, \ldots, k-1\}, (v_i, v_{i+1}) \in A$.*

La función `(esCamino g c)` se verifica si la sucesión de vértices `c` es un camino en el grafo g.

```
-- | Ejemplo
-- >>> grafoCiclo 5
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
-- >>> esCamino (grafoCiclo 5) [1,2,3,4,5,1]
-- True
-- >>> esCamino (grafoCiclo 5) [1,2,4,5,3,1]
-- False
```

```
esCamino :: Ord a => Grafo a -> [a] -> Bool
esCamino g c = all (`aristaEn` g) (zip c (tail c))
```

La función `(aristasCamino c)` devuelve la lista de las aristas recorridas en el camino `c`.

```
-- | Ejemplos
-- >>> aristasCamino [1,2,3]
-- [(1,2),(2,3)]
-- >>> aristasCamino [1,2,3,1]
-- [(1,2),(2,3),(1,3)]
-- >>> aristasCamino [1,2,3,2]
-- [(1,2),(2,3),(2,3)]
aristasCamino :: Ord a => [a] -> [(a,a)]
aristasCamino c =
  map parOrdenado (zip c (tail c))
  where parOrdenado (u,v) | u <= v    = (u,v)
                          | otherwise = (v,u)
```

La función `(verticesCamino c)` devuelve la lista de las vertices recorridas en el camino `c`.

```
-- | Ejemplo
-- >>> verticesCamino [1,2,3,1]
-- [1,2,3]
verticesCamino :: Ord a => [a] -> [a]
verticesCamino c = nub c
```

Definición 3.7.2. *Se llama **longitud** de un camino al número de veces que se atraviesa una arista en dicho camino.*

La función `(longitudCamino c)` devuelve la longitud del camino `c`.

```
-- | Ejemplo
-- >>> longitudCamino [4,2,7]
-- 2
longitudCamino :: [a] -> Int
longitudCamino c = length c - 1
```

La función `(todosCaminos g u v k)` devuelve todos los caminos entre los vértices `u` y `v` en el grafo `g` que tienen longitud `k`.

```
-- | Ejemplo
-- >>> todosCaminos (creaGrafo [1,2] [(1,2)]) 1 2 3
-- [[1,2,1,2]]
-- >>> todosCaminos (bipartitoCompleto 2 3) 1 3 3
-- [[1,3,1,3],[1,3,2,3],[1,4,1,3],[1,4,2,3],[1,5,1,3],[1,5,2,3]]
todosCaminos :: Ord a => Grafo a -> a -> a -> Int -> [[a]]
todosCaminos g u v 0 = if u == v && elem u (vertices g)
```

```
                      then [[u]]
                      else []
todosCaminos g u v 1 = if aristaEn (u,v) g
                      then [[u,v]]
                      else []
todosCaminos g u v k =
    filter p [u:vs | n <- [1..k-1], w <- a u, vs <- tC w v n]
       where p xs = longitudCamino xs == k
             a = adyacentes g
             tC = todosCaminos g
```

La función `(numeroCaminosDeLongitud g u v k)` es el número de caminos de longitud `k` uniendo los vértices `u` y `v` en el grafo `g`.

```
-- | Ejemplos
-- >>> numeroCaminosDeLongitud (completo 7) 1 4 5
-- 1111
-- >>> numeroCaminosDeLongitud grafoPetersen 1 4 3
-- 5
numeroCaminosDeLongitud :: Ord a => Grafo a -> a -> a -> Int -> Int
numeroCaminosDeLongitud g u v = length . todosCaminos g u v
```

3.7.2. Recorridos

Definición 3.7.3. *Sea $G = (V, A)$ un grafo y sean $u, v \in V$. Un camino entre u y v que no repite aristas (quizás vértices) se llama* ***recorrido****.*

La función `(esRecorrido g c)` se verifica si el camino `c` es un recorrido en el grafo `g`.

```
-- | Ejemplo
-- >>> esRecorrido (grafoCiclo 4) [2,1,4]
-- True
-- >>> esRecorrido (grafoCiclo 4) [2,1,4,1]
-- False
-- >>> esRecorrido (grafoCiclo 4) [2,1,3]
-- False
esRecorrido :: Ord a => Grafo a -> [a] -> Bool
esRecorrido g c =
  esCamino g c && sinRepetidos (aristasCamino c)
```

3.7.3. Caminos simples

Definición 3.7.4. *Un camino que no repite vértices (y, por tanto, tampoco aristas) se llama* ***camino simple****.*

La función `(esCaminoSimple g c)` se verifica si el camino `c` es un camino simple en el grafo `g`.

```
-- | Ejemplos
-- >>> esCaminoSimple (creaGrafo [1,2] [(1,1),(1,2)]) [1,1,2]
-- False
-- >>> esCaminoSimple (grafoCiclo 4) [2,1,4]
-- True
-- >>> esCaminoSimple (grafoCiclo 4) [1,4,3,2,1]
-- True
-- >>> esCaminoSimple (grafoCiclo 4) [4,3,2,1,2]
-- False
esCaminoSimple :: Ord a => Grafo a -> [a] -> Bool
esCaminoSimple g [] = True
esCaminoSimple g vs =
  esRecorrido g vs && noRepiteVertices vs
  where noRepiteVertices (x:xs)
          | sinRepetidos (x:xs)             =  True
          | x == last xs && sinRepetidos xs =  True
          | otherwise                       =  False
```

La función `(todosArcosBP g u v)` devuelve una lista con todos los caminos simples posibles en el grafo `g` entre los vértices `u` y `v`, utilizando una algoritmo de búsqueda en profundidad sobre el grafo `g`. Este algoritmo recorre el grafo de izquierda a derecha y de forma al visitar un nodo, explora todos los caminos que pueden continuar por él antes de pasar al siquiente.

```
-- | Ejemplos
-- >>> grafoCiclo 4
-- G [1,2,3,4] [(1,2),(1,4),(2,3),(3,4)]
-- >>> todosArcosBP (grafoCiclo 4) 1 4
-- [[1,4],[1,2,3,4]]
-- >>> todosArcosBP (grafoCiclo 4) 4 1
-- [[4,3,2,1],[4,1]]
-- >>> todosArcosBP (creaGrafo [1..4] [(1,2),(3,4)]) 1 4
-- []
-- >>> todosArcosBP (creaGrafo [1,2] [(1,1),(1,2)]) 1 1
-- [[1]]
todosArcosBP :: Ord a => Grafo a -> a -> a -> [[a]]
todosArcosBP g x y = aux [[y]]
  where aux []       = []
        aux ([]:zss) = aux zss
        aux ((z:zs):zss)
          | z == x    = (z:zs) : aux zss
          | otherwise = aux ([v:z:zs | v <- adyacentes g' z \\ zs] ++ zss)
        g' = eliminaLazos g
        eliminaLazos h = creaGrafo (vertices h)
```

```
                                        [(x,y) | (x,y) <- aristas h, x /= y]
```

La función `(todosArcosBA g u v)` devuelve una lista con todos los caminos simples posibles en el grafo `g` entre los vértices `u` y `v`, utilizando una algoritmo de búsqueda en anchura sobre el grafo `g`. Este algoritmo recorre el grafo por niveles, de forma que el primer camino de la lista es de longitud mínima.

```
-- | Ejemplos
-- >>> grafoCiclo 4
-- G [1,2,3,4] [(1,2),(1,4),(2,3),(3,4)]
-- >>> todosArcosBA (grafoCiclo 4) 1 4
-- [[1,4],[1,2,3,4]]
-- >>> todosArcosBA (grafoCiclo 4) 4 1
-- [[4,1],[4,3,2,1]]
-- >>> todosArcosBA (creaGrafo [1..4] [(1,2),(3,4)]) 1 4
-- []
-- >>> todosArcosBA (creaGrafo [1,2] [(1,1),(1,2)]) 1 1
-- [[1]]
todosArcosBA :: Ord a => Grafo a -> a -> a -> [[a]]
todosArcosBA g x y = aux [[y]]
  where aux []       = []
        aux ([]:zss) = aux zss
        aux ((z:zs):zss)
          | z == x    = (z:zs) : aux zss
          | otherwise = aux (zss ++ [v:z:zs | v <- adyacentes g' z \\ zs])
        g' = eliminaLazos g
```

Vamos a comprobar con QuickCheck que el primer elemento de la lista que devuelve la función `(todosArcosBA g u v)` es de longitud mínima. Para ello, vamos a utilizar la función `(parDeVertices g)` que es un generador de pares de vértices del grafo no nulo `g`. Por ejemplo,

```
ghci> sample (parDeVertices (creaGrafo [1..9] []))
(3,9)
(9,3)
(7,4)
(4,3)
(2,8)
(7,2)
(8,4)
(5,3)
(7,2)
(3,1)
(7,2)
```

```
parDeVertices :: Grafo Int -> Gen (Int,Int)
parDeVertices g = do
```

```
  x <- elements vs
  y <- elements vs
  return (x,y)
  where vs = vertices g
```

La propiedad es

```
prop_todosArcosBA :: Grafo Int -> Property
prop_todosArcosBA g =
  not (esGrafoNulo g) ==>
  forAll (parDeVertices g)
         (\(x,y) -> let zss = todosArcosBA g x y
                    in null zss || longitudCamino (head zss) ==
                                   minimum (map longitudCamino zss))
```

La comprobación es

```
ghci> quickCheck prop_todosArcosBA
+++ OK, passed 100 tests:
```

Veamos que la función `(todosArcosBP g u v)` no verifica la propiedad.

```
ghci> quickCheck prop_todosArcosBP
*** Failed! Falsifiable (after 6 tests):
G [1,2,3,4,5] [(1,2),(1,4),(1,5),(2,2),(2,3),(2,4),(4,4),(4,5),(5,5)]
(5,2)
```

```
prop_todosArcosBP :: Grafo Int -> Property
prop_todosArcosBP g =
  not (esGrafoNulo g) ==>
  forAll (parDeVertices g)
         (\(x,y) -> let zss = todosArcosBP g x y
                    in null zss || longitudCamino (head zss) ==
                                   minimum (map longitudCamino zss))
```

La función `(numeroArcosDeLongitud g u v k)` devuelve el número de caminos entre los vértices `u` y `v` en el grafo `g` que tienen longitud `k`.

```
-- | Ejemplos
-- >>> numeroArcosDeLongitud (completo 6) 1 3 4
-- 24
-- >>> numeroArcosDeLongitud grafoPetersen 1 3 4
-- 4
numeroArcosDeLongitud :: Ord a => Grafo a -> a -> a -> Int -> Int
numeroArcosDeLongitud g u v k =
    length (filter p (todosArcosBA g u v))
    where p vs = longitudCamino vs == k
```

3.7.4. Conexión

Definición 3.7.5. *Dado un grafo $G = (V, A)$, sean $u, v \in V$. Si existe algún camino entre u y v en el grafo G diremos que están* ***conectados*** *y lo denotamos por $u\ v$.*

La función `(estanConectados g u v)` se verifica si los vértices `u` y `v` están conectados en el grafo `g`.

```
-- | Ejemplos
-- >>> estanConectados (creaGrafo [1..4] [(1,2),(2,4)]) 1 4
-- True
-- >>> estanConectados (creaGrafo [1..4] [(1,2),(3,4)]) 1 4
-- False
estanConectados :: Ord a => Grafo a -> a -> a -> Bool
estanConectados g u v
  | esGrafoNulo g = False
  | otherwise     = not (null (todosArcosBA g u v))
```

Nota 3.7.1. La función `(estanConectados g u v)` no necesita calcular todos los caminos entre `u` y `v`. Puesto que Haskell utiliza por defecto evaluación perezosa, si existe algún camino entre los dos vértices, basta calcular el primero para saber que la lista de todos los caminos no es vacía

3.7.5. Distancia

Definición 3.7.6. *Se define la* ***distancia*** *entre u y v en el grafo G como la longitud del camino más corto que los une. Si u y v no están conectados, decimos que la distancia es infinita.*

La función `(distancia g u v)` devuelve la distancia entre los vértices `u` y `v` en el grafo `g`. En caso de que los vértices no estén conectados devuelve el valor `Nothing`.

```
-- | Ejemplos
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 1
-- Just 0
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 2
-- Just 1
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 3
-- Just 2
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 4
-- Nothing
distancia :: Ord a => Grafo a -> a -> a -> Maybe Int
distancia g u v
  | estanConectados g u v =
      Just (longitudCamino (head (todosArcosBA g u v)))
  | otherwise = Nothing
```

Definición 3.7.7. *Dado $G = (V, A)$ un grafo, sean $u, v \in V$. Un camino entre u y v cuya longitud coincide con la distancia entre los vértices se llama* ***geodésica*** *entre u y v.*

La función `(esGeodesica g c)` se verifica si el camino `c` es una geodésica entre sus extremos en el grafo `g`.

```
-- | Ejemplos
-- >>> let g = creaGrafo [1..4] [(1,2),(1,3),(2,3),(3,4)]
-- >>> esGeodesica g [1,3,4]
-- True
-- >>> esGeodesica g [1,2,3,4]
-- False
esGeodesica :: Ord a => Grafo a -> [a] -> Bool
esGeodesica g c =
  esCamino g c &&
  longitudCamino c == fromJust (distancia g u v)
  where u = head c
        v = last c
```

Nota 3.7.2. No hace falta imponer en la definición de la función que el camino sea un recorrido, pues el camino de mínima longitud entre dos vértices, es siempre un recorrido.

3.7.6. Caminos cerrados

Definición 3.7.8. *Un camino en un grafo G se dice* ***cerrado*** *si sus extremos son iguales. Diremos que un camino cerrado de longitud tres es un* ***triángulo****.*

La función `(esCerrado g c)` se verifica si el camino `c` es cerrado en el grafo `g`.

```
-- | Ejemplos
-- >>> let g = creaGrafo [1..4] [(1,2),(1,3),(2,3),(3,4)]
-- >>> esCerrado g [1,2,3,1]
-- True
-- >>> esCerrado g [1,2,3]
-- False
-- >>> esCerrado g [1,2,4,1]
-- False
esCerrado :: (Ord a) => Grafo a -> [a] -> Bool
esCerrado _ []    = False
esCerrado g (v:c) =
  esCamino g c && v == last c
```

La función `(triangulos g v)` devuelve la lista de todos los triángulos que pasan por el vértice `v` en el grafo `g`.

```
-- | Ejemplos
-- >>> triangulos (completo 4) 3
```

```
-- [[3,1,2,3],[3,1,4,3],[3,2,1,3],[3,2,4,3],[3,4,1,3],[3,4,2,3]]
-- >>> triangulos (grafoCiclo 6) 1
-- []
triangulos :: Ord a => Grafo a -> a -> [[a]]
triangulos g u = todosCaminos g u u 3
```

3.7.7. Circuitos

Definición 3.7.9. *Un recorrido en un grafo G se dice* ***circuito*** *si sus extremos son iguales.*

La función `(esCircuito g c)` se verifica si la sucesión de vértices `c` es un circuito en el grafo `g`.

```
-- | Ejemplos
-- >>> grafoCiclo 4
-- G [1,2,3,4] [(1,2),(1,4),(2,3),(3,4)]
-- >>> esCircuito (grafoCiclo 4) [1,2,3,4,1]
-- True
-- >>> esCircuito (grafoCiclo 4) [1,2,3,4]
-- False
-- >>> esCircuito (grafoCiclo 4) [1,2,3,4,1,4,1]
-- False
esCircuito :: (Ord a) => Grafo a -> [a] -> Bool
esCircuito g c =
  esRecorrido g c && esCerrado g c
```

3.7.8. Ciclos

Definición 3.7.10. *Un camino simple en un grafo G se dice que es un* ***ciclo*** *si sus extremos son iguales.*

La función `(esCiclo g c)` se verifica si el camino `c` es un ciclo en el grafo `g`.

```
-- | Ejemplos
-- >>> esCiclo (grafoCiclo 4) [1,2,1]
-- False
-- >>> esCiclo (grafoCiclo 4) [1,2,3,4,1]
-- True
-- >>> esCiclo (grafoCiclo 4) [1,2,3,4]
-- False
-- >>> esCiclo (grafoCiclo 4) [1,2,3,4,1,4,1]
-- False
esCiclo :: (Ord a) => Grafo a -> [a] -> Bool
esCiclo g c =
  esCaminoSimple g c && esCerrado g c
```

La función `(todosCiclos g v)` devuelve todos los ciclos en el grafo `g` que pasan por el vértice `v`.

```
-- | Ejemplos
-- >>> todosCiclos (grafoCiclo 4) 3
-- [[3,4,1,2,3],[3,2,1,4,3]]
-- >>> todosCiclos (completo 3) 2
-- [[2,3,1,2],[2,1,3,2]]
-- >>> todosCiclos (creaGrafo [1,2,3] [(1,1),(1,2),(1,3),(2,3)]) 1
-- [[1],[1,3,2,1],[1,2,3,1]]
-- >>> todosCiclos (creaGrafo [1,2] [(1,2),(2,2),(2,3)]) 2
-- [[2]]
todosCiclos :: Ord a => Grafo a -> a -> [[a]]
todosCiclos g x =
    map f (filter p (concat [todosArcosBA g x u | u <- adyacentes g x]))
    where p c = longitudCamino c /= 1
          f c | longitudCamino c == 0 = c
              | otherwise = c ++ [x]
```

Nota 3.7.3. El algoritmo utilizado en la definición de `(todosCiclos g)` es el de búsqueda en anchura. Este algoritmo recorre el grafo por niveles, de forma que el primer camino de la lista es de longitud mínima.

3.7.9. Grafos acíclicos

Definición 3.7.11. *Diremos que un grafo* $G = (V, A)$ *es* ***acíclico*** *si no contiene ningún ciclo; es decir, si* $\forall v \in V$ *no existe ningún camino simple que comience y termine en* v.

La función `(esAciclico g)` se verifica si el grafo `g` es acíclico.

```
-- | Ejemplo
-- >>> esAciclico (creaGrafo [1..4] [(1,2),(2,4)])
-- True
-- >>> esAciclico (grafoCiclo 5)
-- False
-- >>> esAciclico (grafoEstrella 6)
-- True
esAciclico :: Ord a => Grafo a -> Bool
esAciclico g =
  and [null (todosCiclos g x) | x <- vertices g]
```

3.8. Conectividad de los grafos

3.8.1. Estar conectados por un camino

Teorema 3.8.1. *Dado un grafo G, la relación $u \sim v$ (estar conectados por un camino) es una relación de equivalencia.*

La función `(estarConectadosCamino g)` devuelve la relación entre los vértices del grafo g de estar conectados por un camino en él.

```
-- | Ejemplo
-- >>> estarConectadosCamino (creaGrafo [1..4] [(1,2),(2,4)])
-- [(1,1),(1,2),(1,4),(2,1),(2,2),(2,4),(3,3),(4,1),(4,2),(4,4)]
estarConectadosCamino :: Ord a => Grafo a -> [(a,a)]
estarConectadosCamino g =
  [(u,v) | u <- vs, v <- vs , estanConectados g u v]
  where vs = vertices g
```

A continuación, comprobaremos el resultado con QuickCheck.

```
ghci> quickCheck prop_conectadosRelEqui
+++ OK, passed 100 tests.
```

```
prop_conectadosRelEqui :: Grafo Int -> Bool
prop_conectadosRelEqui g =
  esRelacionEquivalencia (vertices g) (estarConectadosCamino g)
```

3.8.2. Componentes conexas de un grafo

Definición 3.8.2. *Las clases de equivalencia obtenidas por la relación $\sim$, estar conectados por un camino en un grafo G, inducen subgrafos en G, los vértices y todas las aristas de los caminos que los conectan, que reciben el nombre de* ***componentes conexas por caminos*** *de G.*

La función `(componentesConexas1 g)` devuelve las componentes conexas por caminos del grafo g.

```
-- | Ejemplos
-- >>> componentesConexas (creaGrafo [1..5] [(1,2),(2,3)])
-- [[1,2,3],[4],[5]]
-- >>> componentesConexas (creaGrafo [1..5] [(1,2),(2,3),(4,5)])
-- [[1,2,3],[4,5]]
-- >>> componentesConexas (creaGrafo [1..3] [(1,2),(2,3)])
-- [[1,2,3]]
componentesConexas1 :: Ord a => Grafo a -> [[a]]
componentesConexas1 g =
  clasesEquivalencia (vertices g) (estarConectadosCamino g)
```

Vamos a definir dos nuevas funciones: una que compruebe si el grafo es nulo o completo, pues en dichos casos, las componentes serán el conjunto vacío y el conjunto del conjunto de vértices respectivamente y otra con un algoritmo algo más complejo.

```
componentesConexas2 :: Ord a => Grafo a -> [[a]]
componentesConexas2 g
  | esGrafoNulo g = []
  | esCompleto g  = [vertices g]
  | otherwise     =
      clasesEquivalencia (vertices g) (estarConectadosCamino g)
```

```
componentesConexas3 :: Ord a => Grafo a -> [[a]]
componentesConexas3 g = aux (vertices g) [] []
    where aux [] [] []     = []
          aux [] xs ys     = [xs]
          aux (v:vs) [] [] =
              aux (vs \\ (a v)) (i v (a v)) (a v)
          aux vs xs ys | null ((ug [a v | v <- ys]) \\ xs) =
                         xs: aux vs [] []
                       | otherwise =
                           aux (vs \\ ug [a v | v <- ys])
                               (u xs (ug [a v | v <- ys]))
                               (ug [a v | v <- ys] \\ ys)
          a  = adyacentes g
          i  = inserta
          ug = unionGeneral
          u  = unionConjuntos
```

La comprobación con `QuickCheck` de la equivalencia de las definiciones es:

```
ghci> quickCheck prop_EquiComponentesConexas
+++ OK, passed 100 tests.
ghci> quickCheck prop_EquiComponentesConexas2
+++ OK, passed 100 tests.
```

```
prop_EquiComponentesConexas :: Grafo Int -> Bool
prop_EquiComponentesConexas g =
    componentesConexas1 g == componentesConexas2 g
```

```
prop_EquiComponentesConexas2 :: Grafo Int -> Bool
prop_EquiComponentesConexas2 g =
    componentesConexas1 g == componentesConexas3 g
```

Comparemos ahora la eficiencia de las definiciones:

```
ghci> componentesConexas1 grafoNulo
[]
(0.03 secs, 0 bytes)
```

```
ghci> componentesConexas2 grafoNulo
[]
(0.01 secs, 0 bytes)
ghci> componentesConexas3 grafoNulo
[]
(0.01 secs, 0 bytes)
ghci> length (componentesConexas1 (completo 50))
1
(0.23 secs, 0 bytes)
ghci> length (componentesConexas2 (completo 50))
1
(0.16 secs, 0 bytes)
ghci> length (componentesConexas3 (completo 50))
1
(0.08 secs, 0 bytes)
ghci> length (componentesConexas1 (completo 100))
1
(2.17 secs, 205,079,912 bytes)
ghci> length (componentesConexas2 (completo 100))
1
(2.85 secs, 0 bytes)
ghci> length (componentesConexas3 (completo 100))
1
(1.19 secs, 0 bytes)
ghci> length (componentesConexas1 (completo 150))
1
(12.95 secs, 742,916,792 bytes)
ghci> length (componentesConexas2 (completo 150))
1
(20.48 secs, 0 bytes)
ghci> length (componentesConexas3 (completo 150))
1
(9.64 secs, 0 bytes)
```

Con grafos completos de gran tamaño, `Haskell` tarda más tiempo en comprobar que es completo que en calcular directamente sus clases de equivalencia y para grafos nulos apenas hay diferencia en el coste entre la primera y la segunda definición. Por otra parte, la tercera definición es la más eficiente en todos los casos y, será por tanto la que utilizaremos a lo largo del trabajo.

```
componentesConexas :: Ord a => Grafo a -> [[a]]
componentesConexas = componentesConexas3
```

La función `(numeroComponentes g)` devuelve el número de componentes conexas del grafo `g`.

```
-- Ejemplos
-- >>> numeroComponentes (creaGrafo [1..5] [(1,2),(2,3),(4,5)])
```

```
-- 2
-- >>> numeroComponentes (creaGrafo [1..4] [(1,2),(2,2),(3,4)])
-- 2
numeroComponentes :: Ord a => Grafo a -> Int
numeroComponentes g
    | null (aristas g) = orden g
    | otherwise        = length (componentesConexas g)
```

Nota 3.8.1. Hemos comprobado para la definición anterior que comprobar si un grafo es completo es muy costoso, luego, no conviene añadirlo como patrón.

3.8.3. Grafos conexos

Definición 3.8.3. *Dado un grafo, diremos que es* ***conexo*** *si la relación tiene una única clase de equivalencia en él; es decir, si el grafo tiene una única componente conexa.*

La función `(esConexo g)` se verifica si el grafo `g` es conexo.

```
-- Ejemplos
-- >>> esConexo (creaGrafo [1..3] [(1,2),(2,3)])
-- True
-- >>> esConexo (creaGrafo [1..5] [(1,2),(2,3),(4,5)])
-- False
esConexo :: Ord a => Grafo a -> Bool
esConexo g = length (componentesConexas g) == 1
```

```
esConexo2 :: Ord a => Grafo a -> Bool
esConexo2 g | esGrafoNulo g = False
            | esUnitario (vertices g) = True
            | otherwise = aux (vertices g) [] []
            where aux [] [] [] = False
                  aux [] xs _ = True
                  aux (v:vs) [] [] | null vs = True
                                   | null (a v) = False
                                   | otherwise =
                                       aux (vs \\ a v) (i v (a v)) (a v)
                  aux vs xs ys | null ((ug [a v | v <- xs]) \\ xs) = False
                               | otherwise =
                                   aux (vs \\ (ug [a v | v <- xs]))
                                       (u (ug [a v | v <- xs]) xs)
                                       ((ug [a v | v <- xs]) \\ xs)
                  a  = adyacentes g
                  ug = unionGeneral
                  i = inserta
                  u = unionConjuntos
```

La comprobación de la equivalencia de las definiciones con `QuickCheck` es:

```
ghci> quickCheck prop_EquiEsConexo
+++ OK, passed 100 tests.
```

```
prop_EquiEsConexo :: Grafo Int -> Bool
prop_EquiEsConexo g =
    esConexo g == esConexo2 g
```

Vamos a comparar ahora su eficiencia:

```
ghci> let g1 = grafoCiclo 100
(0.00 secs, 0 bytes)
ghci> let g = creaGrafo (vertices g1) (aristas g1    [(2,3),(98,99)])
(0.00 secs, 0 bytes)
ghci> esConexo g
False
(1.17 secs, 151,632,168 bytes)
ghci> esConexo2 g
False
(0.01 secs, 0 bytes)
ghci> esConexo (grafoCiclo 500)
True
(73.41 secs, 10,448,270,712 bytes)
ghci> esConexo2 (grafoCiclo 500)
True
(189.39 secs, 18,468,849,096 bytes)
ghci> esConexo (grafoCiclo 500)
True
(73.41 secs, 10,448,270,712 bytes)
ghci> esConexo2 (grafoCiclo 500)
True
(189.39 secs, 18,468,849,096 bytes)
```

En principio no nos proporciona ninguna ventaja la segunda definición, así que trabajaremos con la primera que hemos dado, ya que es mucho más sencilla.

Teorema 3.8.4. *Sea G un grafo, $G = (V, A)$ es conexo si y solamente si $\forall u, v \in V$ existe un camino entre u y v.*

Vamos a comprobar el resultado con QuickCheck.

```
ghci> quickCheck prop_caracterizaGrafoConexo
+++ OK, passed 100 tests.
```

```
prop_caracterizaGrafoConexo :: Grafo Int -> Property
prop_caracterizaGrafoConexo g =
    not (esGrafoNulo g) ==>
    esConexo g == and [estanConectados g u v
                       | u <- vertices g, v <- vertices g]
```

3.8.4. Excentricidad

Definición 3.8.5. *Sean $G = (V, A)$ un grafo y $v \in V$. Se define la* ***excentricidad*** *de v como el máximo de las distancias entre v y el resto de vértices de G. La denotaremos por $e(v)$.*

La función `(excentricidad g v)` devuelve la excentricidad del vértice `v` en el grafo `g`.

```
-- | Ejemplos
-- >>> let g = creaGrafo [1..3] [(1,2),(2,3),(3,3)]
-- >>> excentricidad g 1
-- Just 2
-- >>> excentricidad g 2
-- Just 1
-- >>> excentricidad (creaGrafo [1..3] [(1,2),(3,3)]) 1
-- Nothing
excentricidad :: Ord a => Grafo a -> a -> Maybe Int
excentricidad g u
  | esGrafoNulo g             = Nothing
  | orden g == 1              = Just 0
  | Nothing `elem` distancias = Nothing
  | otherwise                 = maximum distancias
  where distancias = [distancia g u v | v <- vertices g \\ [u]]
```

La función `(excentricidades g)` devuelve la lista ordenada de las excentricidades de los vértices del grafo `g`.

```
-- | Ejemplos
-- >>> excentricidades (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- [Just 1,Just 2,Just 2]
-- >>> excentricidades (creaGrafo [1..3] [(1,2),(3,3)])
-- [Nothing,Nothing,Nothing]
excentricidades :: Ord a => Grafo a -> [Maybe Int]
excentricidades g = sort (map (excentricidad g) (vertices g))
```

3.8.5. Diámetro

Definición 3.8.6. *Sea $G = (V, A)$ un grafo. Se define el* ***diámetro*** *de G como el máximo de las distancias entre los vértices en V. Lo denotaremos por $d(G)$.*

La función `(diametro g)` devuelve el diámetro del grafo `g`.

```
-- | Ejemplos
-- >>> diametro (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- Just 2
-- >>> diametro (creaGrafo [1..3] [(1,2),(3,3)])
-- Nothing
diametro :: Ord a => Grafo a -> Maybe Int
```

```
diametro g
  | esGrafoNulo g             = Nothing
  | Nothing `elem` distancias = Nothing
  | otherwise                 = maximum distancias
  where vs         = vertices g
        distancias = [distancia g u v | u <- vs, v <- vs, u <= v]
```

3.8.6. Radio

Definición 3.8.7. *Sean $G = (V, A)$ un grafo y $v \in V$. Se define el* ***radio*** *de G como el mínimo de las excentricidades de sus vértices. Lo denotaremos por $r(G)$.*

La función `(radio g)` devuelve el radio del grafo `g`.

```
-- | Ejemplos
-- >>> radio (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- Just 1
-- >>> radio (creaGrafo [1..3] [(1,2),(3,3)])
-- Nothing
radio :: Ord a => Grafo a -> Maybe Int
radio g
  | esGrafoNulo g     = Nothing
  | Nothing `elem` ds = Nothing
  | otherwise         = minimum ds
  where ds = [excentricidad g v | v <- vertices g]
```

3.8.7. Centro

Definición 3.8.8. *Sean $G = (V, A)$ un grafo. Llamamos* ***centro*** *del grafo G al conjunto de vértices de excentricidad mínima. A estos vértices se les denomina* ***vértices centrales****.*

La función `(centro g)` devuelve el centro del grafo `g`. Por ejemplo,

```
-- | Ejemplos
-- >>> centro (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- [2]
-- >>> centro (creaGrafo [1..3] [(1,2),(1,3),(2,3)])
-- [1,2,3]
-- >>> centro (creaGrafo [1..3] [(1,2),(3,3)])
-- [1,2,3]
centro :: Ord a => Grafo a -> [a]
centro g = [v | v <- vertices g, excentricidad g v == r]
  where r = radio g
```

3.8.8. Grosor

Definición 3.8.9. *Sean $G = (V, A)$ un grafo. Se llama* ***grosor*** *o* ***cintura*** *del grafo G al mínimo de las longitudes de los ciclos de G. Si el grafo no posee ciclos (es decir, es un grafo acíclico), se dice que su cintura es infinita.*

La función `(grosor g)` devuelve el grosor del grafo `g`.

```
-- | Ejemplos
-- >>> grosor (creaGrafo [1,2,3] [(1,2),(2,3)])
-- Nothing
-- >>> grosor (creaGrafo [1,2,3] [(1,1),(1,2),(2,3),(3,4)])
-- Just 0
-- >>> grosor grafoPetersen
-- Just 5
-- >>> grosor grafoMoebiusCantor
-- Just 6
-- >>> grosor grafoHeawood
-- Just 6
-- >>> grosor grafoMcGee
-- Just 7
-- >>> grosor grafoTutteCoxeter
-- Just 8
grosor :: Ord a => Grafo a -> Maybe Int
grosor g
  | esAciclico g = Nothing
  | otherwise    = Just (minimum [longitudCamino (head yss)
                                  | x <- vertices g
                                  , let yss = todosCiclos g x
                                  , not (null yss)])
```

Propiedades del grosor de los grafos

Teorema 3.8.10. *El grosor del grafo ciclo de orden n, C_n, es ∞, si $n < 3$ y 3 en caso contrario.*

La propiedad se expresa por

```
prop_grosor_grafoCiclo :: Int -> Bool
prop_grosor_grafoCiclo n =
  grosor (grafoCiclo n) == if n < 3
                           then Nothing
                           else Just n
```

Su comprobación para $n \leq 30$ es

```
ghci> all prop_grosor_grafoCiclo [1..30]
True
```

Teorema 3.8.11. *El grosor del grafo amistad de orden n es 3, para todo n.*

```
prop_grosor_grafoAmistad :: Int -> Bool
prop_grosor_grafoAmistad n =
  grosor (grafoAmistad n) == Just 3
```

Su comprobación para $n \leq 30$ es

```
ghci> all prop_grosor_grafoAmistad [1..30]
True
```

Teorema 3.8.12. *El grosor del grafo completo de orden n, K_n, es ∞ si $n < 3$ y 3 en caso contrario.*

La propiedad se expresa por

```
prop_grosor_completo :: Int -> Bool
prop_grosor_completo n =
  grosor (completo n) == if n < 3
                         then Nothing
                         else Just 3
```

Su comprobación para $n \leq 30$ es

```
ghci> all prop_grosor_completo [1..30]
True
```

Teorema 3.8.13. *El grosor del grafo bipartito completo de orden, $K_{m,n}$, es ∞ si $m = 1$ ó $n = 1$ y es 4 en caso contrario. y 3 en caso contrario.*

La propiedad se expresa por

```
prop_grosor_bipartitoCompleto :: Int -> Int -> Bool
prop_grosor_bipartitoCompleto m n =
  grosor (bipartitoCompleto m n) == if m == 1 || n == 1
                                    then Nothing
                                    else Just 4
```

Su comprobación para $1 \leq m \leq n \leq 15$ es

```
ghci> and [prop_grosor_bipartitoCompleto m n | m <- [1..15], n <- [m..15]]
True
```

Teorema 3.8.14. *El grosor del grafo rueda de orden n, W_n es ∞ si $n < 3$ y 3 en caso contrario.*

La propiedad se expresa por

```
prop_grosor_grafoRueda :: Int -> Bool
prop_grosor_grafoRueda n =
  grosor (grafoRueda n) == if n < 3
                           then Nothing
                           else Just 3
```

Su comprobación para $1 \leq n \leq 30$ es

```
ghci> all prop_grosor_grafoRueda [1..30]
True
```

3.8.9. Propiedades e invariantes por isomorfismos

Teorema 3.8.15. *Sean $G = (V, A)$ y $G' = (V', A')$ grafos isomorfos con $\phi : V \to V'$ un isomorfismo. Entonces, dados $u, v \in V$, $u \sim v$ si y solamente si $\phi(u) \sim \phi(v)$.*

La comprobación del teorema con QuickCheck es:

```
ghci> quickCheck prop_ConexionIsomorfismo1
+++ OK, passed 100 tests.
```

```
prop_ConexionIsomorfismo1 :: Grafo Int -> Grafo Int -> Bool
prop_ConexionIsomorfismo1 g h =
  not (isomorfos g h) ||
  and [ec g u v == ec h (imagen phi u) (imagen phi v)
      | u <- vs
      , v <- vs
      , phi <- isomorfismos g h]
      where vs = vertices g
            ec = estanConectados
```

Teorema 3.8.16. *Sean $G = (V, A)$ y $G' = (V', A')$ grafos isomorfos con $\phi : V \to V'$ un isomorfismo. Entonces, ϕ lleva cada componente conexa de G en una componente conexa de G'.*

La comprobación del teorema con QuickCheck es:

```
ghci> quickCheck prop_ConexionIsomorfismo2
+++ OK, passed 100 tests.
```

```
prop_ConexionIsomorfismo2 :: Grafo Int -> Grafo Int -> Bool
prop_ConexionIsomorfismo2 g h =
  not(isomorfos g h) ||
  and [conjuntosIguales cch (aux f) | f <- isomorfismos g h]
      where cch = componentesConexas h
            ccg = componentesConexas g
            aux f = map (sort . imagenConjunto f) ccg
```

Teorema 3.8.17. *Sean $G = (V, A)$ y $G' = (V', A')$ grafos isomorfos con $\phi : V \to V'$ un isomorfismo. Entonces, G y G' tienen el mismo número de componentes conexas.*

La comprobación del teorema con QuickCheck es:

```
ghci> quickCheck prop_ConexionIsomorfismo3
+++ OK, passed 100 tests.
```

```
prop_ConexionIsomorfismo3 :: Grafo Int -> Grafo Int -> Bool
prop_ConexionIsomorfismo3 g h =
  not (isomorfos g h) ||
  numeroComponentes g == numeroComponentes h
```

Teorema 3.8.18. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, se verifica que el diámetro de G es igual al diámetro de G'; es decir, el diámetro de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos diametro)
+++ OK, passed 100 tests.
```

Teorema 3.8.19. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, se verifica que el radio de G es igual al radio de G'; es decir, el radio de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos radio)
+++ OK, passed 100 tests.
```

Teorema 3.8.20. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, se verifica que el grosor de G es igual al grosor de G'; es decir, el grosor de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos grosor)
+++ OK, passed 100 tests.
```

Teorema 3.8.21. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, se verifica que el centro de G es igual al centro de G'; es decir, el centro de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos centro)
+++ OK, passed 100 tests.
```

Teorema 3.8.22. *Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \to V'$ un isomorfismo. Entonces, se verifica que el número de componentes conexas de G es igual al número de las de G'; es decir, el número de componentes conexas de un grafo es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos numeroComponentes)
+++ OK, passed 100 tests.
```

Teorema 3.8.23. *Sean* $G = (V, A)$ *y* $G' = (V', A')$ *dos grafos y* $\phi : V \to V'$ *un isomorfismo. Entonces, se verifica que los valores que toman las excentricidades de los vértices* G *es igual al de los vértices de* G'*; es decir, el conjunto de valores que toman las excentricidades de los vértices es un invariante por isomorfismos.*

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos excentricidades)
+++ OK, passed 100 tests.
```

Capítulo 4

Matrices asociadas a grafos

En este capítulo se definirán algunas representaciones de grafos como matrices, se comprobarán sus propiedades y volveremos a definir algunos conceptos del capítulo 3 ahora utilizando matrices.
Para elaborar el contenido del capítulo utilizaré el segundo tema del trabajo fin de grado de Fco. Javier Franco Galvín: *Aspectos algebraicos en teoría de grafos* ([7]). En el trabajo, el autor introduce los conceptos de *matriz de adyacencia*, *matriz de incidencia* y otras matrices asociadas a grafos.

4.1. Generador de grafos simples

En esta sección, presentaremos el generador de grafos que nos permitirá generar grafos simples como listas de aristas arbitrariamente y usarlos como ejemplos o para comprobar propiedades en la sección de matrices asociadas a grafos.
`(grafoSimple)` es un generador de grafos simples de hasta 10 vértices. Por ejemplo,

```
ghci> sample grafoSimple
G [1,2,3,4,5,6,7,8] [(1,4),(1,5),(1,6),(1,7),(1,8),(2,3),
   (2,4),(2,5),(2,7),(3,5),(3,7),(4,5),(4,6),(5,6),(6,7)]
G [1,2,3,4,5,6,7,8,9] [(1,4),(1,8),(2,3),(2,5),(2,9),
   (3,4),(3,5),(3,6),(3,8),(3,9),(4,6),(4,8),(4,9),
   (5,8),(5,9),(6,9),(7,8)]
G [1,2,3,4,5,6] [(1,2),(1,3),(2,4),(2,5),(2,6),(3,5),(5,6)]
G [1,2,3,4,5,6,7] [(1,2),(1,3),(1,4),(2,4),(2,5),(2,6),
   (3,6),(4,5),(4,6),(6,7)]
G [1,2,3,4] [(1,3),(1,4),(2,3),(3,4)]
G [1,2,3] [(2,3)]
G [1,2] [(1,2)]
G [1,2,3,4,5,6] [(1,6),(2,4),(2,6),(3,5),(3,6),(5,6)]
G [1,2,3,4,5,6,7,8,9] [(1,2),(1,4),(1,5),(1,6),(1,7),(2,4),
   (3,4),(3,7),(3,8),(3,9),(4,5),(4,7),(4,9),(5,6),(5,7),
   (5,8),(5,9),(6,7),(6,9),(7,8),(7,9)]
```

```
G [1,2,3,4,5,6,7,8,9] [(1,3),(1,4),(1,5),(1,6),(2,3),(2,4),
   (2,9),(3,5),(3,9),(4,5),(4,7),(4,8),(4,9),(5,7),(5,8),
   (5,9),(6,7),(6,8),(6,9),(7,9)]
G [1] []
```

```
grafoSimple :: Gen (Grafo Int)
grafoSimple = do
  n  <- choose (0,10)
  as <- sublistOf [(x,y) | x <- [1..n], y <- [x+1..n]]
  return (creaGrafo [1..n] as)
```

4.2. Matrices de adyacencia

4.2.1. Definición y propiedades

Usaremos la función `(imprimeMatriz p)` para imprimir por pantalla la matriz `p` con una fila por línea.

```
-- | Ejemplo
-- >>> imprimeMatriz (fromLists [[1,3,4],[3,5,7],[4,7,9]])
-- [1,3,4]
-- [3,5,7]
-- [4,7,9]
imprimeMatriz :: Show a => Matrix a -> IO ()
imprimeMatriz p =
  mapM_ print (toLists p)
```

Definición 4.2.1. *Sea* $G = (V, A)$ *un grafo simple con* $V = \{v_1, \ldots, v_n\}$. *Se define su matriz de adyacencia como* $\mathcal{A}_{n\times n} = (a_{i,j})$ *donde el elemento* $a_{i,j}$ *toma el valor* 1 *si existe alguna arista en* A *uniendo los vértices* v_i *y* v_j *y* 0 *en caso contrario.*

Nota 4.2.1. La matriz de adyacencia depende del etiquetado del grafo.

La función `(matrizAdyacencia g)` es la matriz de adyacencia del grafo `g`.

```
-- | Ejemplo,
-- >>> imprimeMatriz (matrizAdyacencia (grafoCiclo 4))
-- [0,1,0,1]
-- [1,0,1,0]
-- [0,1,0,1]
-- [1,0,1,0]
-- >>> imprimeMatriz (matrizAdyacencia (completo 4))
-- [0,1,1,1]
-- [1,0,1,1]
-- [1,1,0,1]
-- [1,1,1,0]
```

```
matrizAdyacencia :: Grafo Int -> Matrix Int
matrizAdyacencia g = matrix n n f
  where n = orden g
        f (i,j) | (i,j) `aristaEn` g = 1
                | otherwise          = 0
```

4.2.2. Propiedades básicas de las matrices

La función `(esSimetrica p)` se verifica si la matriz `p` es simétrica.

```
-- ejemplo,
-- >>> esSimetrica (fromLists [[1,3,4],[3,5,7],[4,7,9]])
-- True
-- >>> esSimetrica (fromLists [[1,3,4],[3,5,7],[4,9,7]])
-- False
esSimetrica :: Eq a => Matrix a -> Bool
esSimetrica p =
  transpose p == p
```

La función `(potencia p n)` devuelvela potencia n–ésima de la matriz cuadrada `p`.

```
-- Ejemplo
-- >>> potencia (fromLists [[1,3,4],[3,5,7],[4,7,9]]) 3
-- (  408  735  975 )
-- (  735 1323 1755 )
-- (  975 1755 2328 )
potencia :: Num a => Matrix a -> Int -> Matrix a
potencia p 1 = p
potencia p n = if (odd n)
               then (m p (potencia (m p p) (div (n-1) 2)))
               else (potencia (m p p) (div (n-1) 2))
    where m = multStd2
```

4.2.3. Propiedades de las matrices de adyacencia

Teorema 4.2.2. *Para todo n, la matriz de adyacencia del grafo ciclo de orden n, C_n es simétrica.*

```
ghci> all prop_simetricaAdyacenciaCompleto [1..30]
True
```

```
prop_simetricaAdyacenciaCompleto :: Int -> Bool
prop_simetricaAdyacenciaCompleto n =
  esSimetrica (matrizAdyacencia (completo n))
```

La función `(tamañoM g)` devuelve el tamaño del grafo simple `g`, calculado usando su matriz de adyacencia.

```
-- | Ejemplos
-- >>> tamañoM (grafoCiclo 4)
-- 4
-- >>> tamañoM grafoPetersen
-- 15
tamañoM :: Grafo Int -> Int
tamañoM g = sum (toList (matrizAdyacencia g)) `div` 2
```

Teorema 4.2.3. *El tamaño de un grafo simple es la mitad de la suma de los elementos de su matriz de adyacencia.*

La comprobación del teorema con QuickCheck es:

```
ghci> quickCheck prop_tamañoMatriz
+++ OK, passed 100 tests.
```

```
prop_tamañoMatriz :: Property
prop_tamañoMatriz =
  forAll grafoSimple
         (\g -> tamaño g == tamañoM g)
```

Teorema 4.2.4. *Si un grafo $G = (V, A)$ tiene un vértice aislado v_i, tanto la fila como la columna i–ésima de su matriz de adyacencia estarán formadas por ceros.*

```
esAisladoM :: Grafo Int -> Int -> Bool
esAisladoM g v = if (all (==0)
                         (((toLists (ma g)) !! (v-1))
                          ++((toLists (t (ma g))) !! (v-1))))
                 then True
                 else False
        where  ma = matrizAdyacencia
               t = transpose
```

La comprobación del teorema es:

```
ghci> quickCheck prop_esAisladoMatriz
+++ OK, passed 100 tests.
```

```
prop_esAisladoMatriz :: Property
prop_esAisladoMatriz =
  forAll grafoSimple
         (\g -> and [esAislado g v == esAisladoM g v
                     | v <- vertices g])
```

Teorema 4.2.5. *Sea un grafo $G = (V, A)$ simple no dirigido con $V = \{v_1, \ldots, v_n\}$. La suma de los elementos de la fila (o columna) i–ésima de su matriz de adyacencia coincide con el grado del vértice v_i. Es decir,*

$$\delta(v_i) = \sum_{j=1}^{n} a_{i,j} = \sum_{j=1}^{n} a_{j,i}$$

```
gradoM :: Grafo Int -> Int -> Int
gradoM g v = sum ((toLists (ma g)) !! (v-1))
    where  ma = matrizAdyacencia
```

La comprobación del teorema es:

```
ghci> quickCheck prop_gradoMatriz
+++ OK, passed 100 tests.
```

```
prop_gradoMatriz :: Property
prop_gradoMatriz =
  forAll grafoSimple
         (\g -> and [grado g v == gradoM g v | v <- vertices g])
```

Teorema 4.2.6. *Sea* $G = (V, A)$ *un grafo bipartito de conjuntos de vértices disjuntos* $V_1 = \{v_1, \ldots, v_k\}$ *y* $V_2 = \{v_{k+1}, \ldots, v_n\}$, *tal que* $V = V_1 \cup V_2$ *y sólo existen aristas que conectan vértices de* V_1 *con vértices de* V_2. *Entonces, con este etiquetado de* V, *la matriz de adyacencia de* G *tiene la forma:*

$$A = \left(\begin{array}{c|c} \theta & B \\ \hline B^t & \theta \end{array}\right)$$

```
prop_BipartitoMatriz :: Grafo Int -> Property
prop_BipartitoMatriz g =
    esBipartito g ==>
    all (==0) [ getElem u v m | [u,v] <- (f (fromJust p))]
        where f (xs,ys) = filter p (subsequences xs ++ subsequences ys)
                    where p zs = length zs == 2
              p = conjuntosVerticesDisjuntos g
              m = matrizAdyacencia g
```

4.2.4. Caminos y arcos

En esta sección, se presentará la información que contiene la matriz de adyacencia de un grafo en relación a los caminos que se encuentran en él.

Teorema 4.2.7. *Sea* $G = (V, A)$ *un grafo con* $V = \{v_1, \ldots, v_n\}$ *y matriz de adyacencia* $A = (a_{i,j})$. *Entonces,* $\forall k \geq 0$ *el elemento* (i, j) *de* A^k, *que denotaremos por* $a^k_{i,j}$, *es el número de caminos de longitud* k *desde el vértice* v_i *al vértice* v_j.

```
numeroCaminosDeLongitudM :: Grafo Int -> Int -> Int -> Int -> Int
numeroCaminosDeLongitudM g u v k = getElem u v mk
    where  ma = matrizAdyacencia g
           n = orden g
           mk = foldr (multStd2) (identity n)  (take k (repeat ma))
```

La comprobación del teorema para $k \leq 6$ es:

```
ghci> quickCheck prop_numeroCaminosMatriz
+++ OK, passed 100 tests.
```

```
prop_numeroCaminosMatriz :: Grafo Int -> Gen Bool
prop_numeroCaminosMatriz g = do
    k <- choose (0,6)
    let vs = vertices g
    return (and [ numeroCaminosDeLongitud g u v k ==
                  numeroCaminosDeLongitudM g u v k
                  |(u,v) <- productoCartesiano vs vs, u < v])
```

De este teorema se deducen las siguientes propiedades:

Teorema 4.2.8. *Sea G un grafo simple. Siguiendo la notación del teorema anterior se tiene que:*

1. $a_{i,i}^{(2)} = \delta(v_i)$.
2. $a_{i,i}^{(3)}$ *es el número de triángulos que contienen al vértice* v_i.

```
gradoCaminosM :: Grafo Int -> Int -> Int
gradoCaminosM g v = getElem v v m2
    where m = matrizAdyacencia g
          m2 = multStd2 m m
```

```
numeroTriangulosM :: Grafo Int -> Int -> Int
numeroTriangulosM g v = getElem v v (potencia m 3)
    where m = matrizAdyacencia g
```

La comprobación con QuickCheck es:

```
ghci> quickCheck prop_GradoCaminosMatriz
+++ OK, passed 100 tests.
ghci> quickCheck prop_TriangulosMatriz
+++ OK, passed 100 tests.
```

```
prop_GradoCaminosMatriz :: Grafo Int -> Bool
prop_GradoCaminosMatriz g =
    and [ grado g v == gradoCaminosM g v | v <- vertices g]
```

```
prop_TriangulosMatriz :: Property
prop_TriangulosMatriz =
    forAll grafoSimple
      (\g -> and [length (triangulos g v) ==
                  numeroTriangulosM g v | v <- vertices g])
```

Capítulo 5

Apéndices

5.1. Sistemas utilizados

El desarrollo de mi Trabajo de Fin de Grado requería de una infraestructura técnica que he tenido que trabajar antes de comenzar a desarrollar el contenido. A continuación, voy a nombrar y comentar los sistemas y paquetes que he utilizado a lo largo del proyecto.

- **Ubuntu como sistema operativo.** El primer paso fue instalar *Ubuntu* en mi ordenador portátil personal. Para ello, seguí las recomendaciones de mi compañero Eduardo Paluzo, quien ya lo había hecho antes.
  Primero, me descargué la imagen del sistema *Ubuntu 16.04 LTS* (para procesador de 64 bits) desde la página de descargas de Ubuntu [1] y también la herramienta LinuxLive USB Creator [2] que transformaría mi pendrive en una unidad USB Booteable cargada con la imagen de Ubuntu. Una vez tuve la unidad USB preparada, procedí a instalar el nuevo sistema: apagué el dispositivo y al encenderlo entré en el `Boot Menu` de la `BIOS` del portátil para arrancar desde el Pendrive en vez de hacerlo desde el disco duro. Automáticamente, comenzó la instalación de *Ubuntu* y solo tuve que seguir las instrucciones del asistente para montar Ubuntu manteniendo además *Windows 10*, que era el sistema operativo con el que había estado trabajando hasta ese momento.
  El resultado fue un poco agridulce, pues la instalación de Ubuntu se había realizado con éxito, sin embargo, al intentar arrancar *Windows* desde la nueva `GRUB`, me daba un error al cargar la imagen del sistema. Después de buscar el error que me aparecía en varios foros, encontré una solución a mi problema: deshabilité el `Security Boot` desde la `BIOS` y pude volver a arrancar *Windows 10* con normalidad.

[1] `http://www.ubuntu.com/download/desktop`
[2] `http://www.linuxliveusb.com/`

- **LaTeX como sistema de composición de textos.** La distribución de LaTeX,*Tex Live*, como la mayoría de software que he utilizado, la descargué utilizando el *Gestor de Paquetes Synaptic*. Anteriormente, sólo había utilizado *TexMaker* como editor de LaTeX así que fue el primero que descargué. Más tarde, mi tutor José Antonio me sugirió que mejor descargara el paquete *AUCTex*, pues me permitiría trabajar con archivos TeX desde el editor *Emacs*, así lo hice y es el que he utilizado para redactar el trabajo. Además de los que me recomendaba el gestor, solo he tenido que descargarme el paquete *spanish* de *babel* para poder componer el trabajo, pues el paquete *Tikz*, que he utilizado para representar los grafos, venía incluido en las sugerencias de *Synaptic*.
- **Haskell como lenguaje de programación.** Ya había trabajado anteriormente con este lenguaje en el grado y sabía que sólo tenía que descargarme la plataforma de *Haskell* y podría trabajar con el editor *Emacs*. Seguí las indicaciones que se dan a los estudiantes de primer curso en la página del Dpto. de Ciencias de la Computación e Inteligencia Artificial [3] y me descargué los paquetes *haskell-platform* y *haskell-mode* desde el *Gestor de Paquetes Synaptic*. La versión de la plataforma de *Haskell* con la que he trabajado es la *2014.2.0.0.debian2* y la del compilador *GHC* la *7.10.3-7*.
- **Dropbox como sistema de almacenamiento compartido.** Ya había trabajado con *Dropbox* en el pasado, así que crear una carpeta compartida con mis tutores no fue ningún problema; sin embargo, al estar *Dropbox* sujeto a software no libre, no me resultó tan sencillo instalarlo en mi nuevo sistema. En primer lugar, intenté hacerlo directamente desde *Ubuntu Software*, que intentó instalar *Dropbox Nautilus* y abrió dos instalaciones en paralelo. Se quedó colgado el ordenador, así que maté los procesos de instalación activos, reinicié el sistema y me descargué directamente el paquete de instalación desde la página de descargas de Dropbox [4] y lo ejecuté desde la terminal.
- **Git como sistema remoto de control de versiones.** Era la primera vez que trabajaba con *Git* y me costó bastante adaptarme. Mi tutor José Antonio y mi compañero Eduardo Paluzo me ayudaron mucho e hicieron el proceso de adaptación mucho más ágil. El manual que Eduardo ha redactado y presenta en su Trabajo Fin de Grado me ha sido muy útil; es una pequeña guía que cualquier interesado en el empleo de *Git* puede utilizar como introducción. La instalación no resultó complicada: a través del *Gestor de Paquetes Synaptic* me descargué el paquete *elpa-magit* y todos los demás paquetes necesarios para trabajar con *Git* en mi portátil usando el editor *Emacs*. Eduardo me ayudó a crear el repositorio *MDenHaskell* en la plataforma *GitHub* y clonarlo en mi ordenador personal. Dicho repositorio contiene el

[3] http://www.cs.us.es/~jalonso/cursos/i1m-15/sistemas.php
[4] https://www.dropbox.com/es/install?os=lnx

total del trabajo y se almacena de forma pública. La versión de *Magit* que utilizo es la *2.5.0-2*

5.2. Mapa de decisiones de diseño en conjuntos

Una vez comenzado el trabajo y habiendo hecho ya varias secciones acerca de la Teoría de Grafos, se puso de manifiesto la conveniencia de crear nuevos capítulos introduciendo conceptos de la Teoría de Conjuntos y las relaciones que se pueden presentar entre sus elementos.
Haskell tiene una librería específica para conjuntos: `Data.Set`, sin embargo, no es tan intuitiva ni tan eficiente como la de listas (`Data.List`). A lo largo del proyecto he creado un módulo que permite trabajar con conjuntos utilizando tanto su representación como listas ordenadas sin elementos repetidos como su representación como listas sin elementos repetidos. Dicho módulo se llama `Conjuntos` y lo presento en el primer capítulo del Trabajo .

5.3. Mapa de decisiones de diseño en grafos

Al comienzo del proyecto, la idea era que las primeras representaciones con las que trabajara fueran las de *grafos como vectores de adyacencia* y *grafos como matrices de adyacencia* que se utilizan en Informática en el primer curso del Grado, con las que ya había trabajado y estaba familiarizada.
Las definiciones de Informática están pensadas para grafos ponderados (dirigidos o no según se eligiera), mientras que en Matemática Discreta apenas se usan grafos dirigidos o ponderados; por tanto, el primer cambio en la representación utilizada fue simplificar las definiciones de modo que solo trabajáramos con grafos no dirigidos y no ponderados, pero manteniendo las estructuras vectorial y matricial que mantenían la eficiencia.
Las representaciones que utilizan *arrays* en *Haskell* son muy restrictivas, pues solo admiten vectores y matrices que se puedan indexar, lo que hace muy complicados todos los algoritmos que impliquen algún cambio en los vértices de los grafos y, además, no permite trabajar con todos los tipos de vértices que pudiéramos desear.
Decidimos volver a cambiar la representación, y esta vez nos decantamos por la representación de *grafos como listas de aristas*, perdiendo en eficiencia pero ganando mucho en flexibilidad de escritura.
Por último mis tutores sugirieron dar una representación de los grafos mediante sus matrices de adyacencia y comprobar la equivalencia de las definiciones que ya había hecho en otros módulos con las dadas para esta representación. Para ello he

trabajado con la librería `Data.Matrix` de *Haskell*.

Bibliografía

[1] J. Alonso. Temas de programación funcional. Technical report, Univ. de Sevilla, 2015.

[2] P. Burzyn. Complejidad computacional en problemas de modificación de aristas en grafos. Technical report, Univ. de Buenos Aires.

[3] K. Claessen and J. Hughes. QuickCheck:A Lightweight Tool for Random Testing of Haskell Programs. Technical report, Chalmers University of Technology, 2000.

[4] M. Cárdenas. Matemática discreta. Technical report, Univ. de Sevilla, 2015.

[5] D. de Matemáticas. El regalo de Cantor. Technical report, IES Matarraña de Valderrobres, 2009.

[6] D. de Álgebra. Álgebra Básica, tema 1: Conjuntos. Technical report, Univ. de Sevilla, 2015.

[7] F. Franco. Aspectos algebraicos en Teoría de Grafos. Technical report, Univ. de Sevilla, 2016.

[8] F. Rabhi and G. Lapalme. *Algorithms: A functional programming approach*. Addison-Wesley Longman Publishing Co., Inc., Boston, MA, USA, 1st edition, 1999.

[9] P. Vasconcelos. Property Testing using QuickCheck. Technical report, Universidade do Porto, 2016.

[10] Wikipedia. Set (mathematics). Technical report, Wikipedia, La Enciclopedia libre., 2016.

[11] Wikipedia. Anexo:Galería de grafos. Technical report, Wikipedia, La Enciclopedia libre., 2016.

[12] Wikipedia. Teoría de la complejidad computacional. Technical report, Wikipedia, La Enciclopedia libre., 2017.

Índice alfabético

Grafo, 14, 48
adyacentes, 49
antiImagenRelacion, 34
aristaEn, 49
aristasCamino, 84
aristas, 49
automorfismos, 83
bipartitoCompleto, 54
biyecciones, 43
cardinal, 18, 27
centro, 99
clasesEquivalencia, 39
combinaciones, 21, 30
complementario, 18, 27
completo, 54
componentesConexas1, 93
componentesConexas2, 94
componentesConexas3, 94
conjuntosIguales', 26
conjuntosIguales, 17
conjuntosVerticesDisjuntos, 55
conservaAdyacencia, 74
creaGrafo, 48
diametro, 98
distancia, 89
dominio, 34
eliminaArista, 70
eliminaLazos, 70
eliminaVertice, 70
elimina, 15, 24
entorno, 65
esAisladoM, 108
esAislado, 65
esAntisimetrica, 37
esAutomorfismo, 82
esBipartito, 56
esCamino, 83
esCerrado, 90
esCircuito, 91
esCompleto, 73
esConexo, 96
esFuncional, 35
esFuncion, 39
esGeodesica, 90
esInyectiva, 41
esIsomorfismo, 77
esLazo, 65
esReflexiva, 36
esRegular, 66
esRelacionEquivalencia, 38
esRelacionHomogenea, 35
esRelacionOrden, 39
esRelacion, 33
esSimetrica, 37, 107
esSimple, 66
esSobreyectiva, 41, 42
esSubconjuntoPropio, 17, 26
esSubconjunto, 16, 25
esSubgrafoMax, 68
esSubgrafoPropio, 68
esSubgrafo, 68
esTransitiva, 37
esUnitario, 19, 28
esVacio, 14, 23

estaRelacionado, 36
estanConectados, 89
excentricidades, 98
excentricidad, 97
funciones, 40
generaGrafo, 50
gradoM, 109, 110
grado, 65
grafoAmistad, 53
grafoCiclo, 52
grafoCirculante, 57
grafoComplementario, 73
grafoEstrella, 56
grafoHeawood, 60
grafoMcGee, 60
grafoMoebiusCantor, 63
grafoNulo, 51
grafoPetersenGen, 58
grafoPetersen, 62
grafoRueda, 57
grafoSimple, 106
grafoThomson, 59
grafoTutteCoxeter, 61
grosor, 99
imagenConjunto, 41
imagenInversa, 43
imagenRelacion, 34
imagen, 40
imprimeMatriz, 106
inserta, 14, 23
interseccion', 29
interseccion, 20
inversa, 43
isomorfismos, 78
isomorfos, 78
longitudCamino, 84
matrizAdyacencia, 106
minimoElemento, 15, 24
morfismos, 75
numeroArcosDeLongitud, 88
numeroCaminosDeLongitud, 85
numeroCaminosDeLongitugM, 109
numeroComponentesConexas, 95
numeroTriangulosM, 110
orden, 64
parDeVertices, 87
pertenece, 15, 24
potencia, 107
productoCartesiano', 30
productoCartesiano, 21
prop_HavelHakimi, 69
prop_LemaApretonDeManos, 69
prop_completos, 72
prop_BipartitoMatriz, 109
prop_ConectadosRelEqui, 93
prop_ConexionIsomorfismo1, 101
prop_ConexionIsomorfismo2, 102
prop_ConexionIsomorfismo3, 102
prop_GradoMatriz, 110
prop_TriangulosMatriz, 110
prop_esAisladoMatriz, 108
prop_gradoMatriz, 109
prop_numeroCaminosMatriz, 110
prop_simetricaAdyacenciaCompleto, 107
prop_tamañoMatriz, 108
prop_todosArcosBA, 88
radio, 98
rango, 34
secuenciaGrados, 67
secuenciaGrafica, 67
sinRepetidos, 15, 25
sonIncidentes, 64
sumaArista, 71
sumaGrafos, 72, 73
sumaVertice, 71
tamañoM, 107
tamaño, 64
todosArcosBA, 87

todosArcos, 86
todosCaminos, 84
todosCiclos, 92
triangulos, 90
unionConjuntos, 19, 28
unionGeneral, 20, 29
vacio, 14, 23
valenciaMax, 66
valenciaMin, 66
variaciones, 22, 31
verticesCamino, 84
vertices, 49

Lista de tareas pendientes