

Matemática discreta en Haskell

María Dolores Valverde Rodríguez

Grupo de Lógica Computacional

Dpto. de Ciencias de la Computación e Inteligencia Artificial

Universidad de Sevilla

Sevilla, 21 de junio de 2016 (Versión de 20 de agosto de 2016)

Esta obra está bajo una licencia Reconocimiento–NoComercial–CompartirIgual 2.5 Spain de Creative Commons.

Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

Bajo las condiciones siguientes:



Reconocimiento. En cualquier explotación de la obra autorizada por la licencia hará falta reconocer la autoría.



No comercial. La explotación de la obra queda limitada a usos no comerciales.



Compartir bajo la misma licencia. La explotación autorizada incluye la creación de obras derivadas siempre que mantengan la misma licencia al ser divulgadas.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/2.5/es/> o envíe una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

Índice general

Introducción	7
1 Conjuntos	9
1.1 El TAD de los conjuntos	10
1.1.1 Conjuntos como listas ordenadas sin repetición	11
1.2 Definiciones y operaciones	14
1.2.1 Conjunto unitario	14
1.2.2 Subconjuntos	15
1.2.3 Igualdad de conjuntos	15
1.2.4 Subconjuntos propios	16
1.2.5 Complementario de un conjunto	16
1.2.6 Cardinal de un conjunto	17
1.2.7 Unión de conjuntos	17
1.2.8 Intersección de conjuntos	18
1.2.9 Producto cartesiano	19
1.2.10 Combinaciones	19
1.2.11 Variaciones con repetición	20
2 Conjuntos	21
2.1 Definiciones y propiedades	22
2.1.1 Pertenencia a un conjunto	23
2.1.2 Conjunto vacío	23
2.1.3 Conjunto unitario	24
2.1.4 Subconjuntos	24
2.1.5 Igualdad de conjuntos	24
2.1.6 Subconjuntos propios	25
2.1.7 Complementario de un conjunto	25
2.1.8 Cardinal de un conjunto	26
2.1.9 Unión de conjuntos	26
2.1.10 Intersección de conjuntos	27
2.1.11 Producto cartesiano	27

2.1.12	Combinaciones	28
2.1.13	Variaciones con repetición	28
3	Relaciones y funciones	29
3.1	Relaciones	29
3.1.1	Relación binaria	29
3.1.2	Imagen por una relación	29
3.1.3	Dominio de una relación	30
3.1.4	Rango de una relación	30
3.1.5	Antiimagen por una relación	31
3.1.6	Relación funcional	31
3.2	Relaciones homogéneas	31
3.2.1	Relaciones reflexivas	32
3.2.2	Relaciones simétricas	33
3.2.3	Relaciones antisimétricas	33
3.2.4	Relaciones transitivas	34
3.2.5	Relaciones de equivalencia	34
3.2.6	Relaciones de orden	35
3.2.7	Clases de equivalencia	35
3.3	Funciones	36
3.3.1	Imagen por una función	37
3.3.2	Funciones inyectivas	37
3.3.3	Funciones sobreyectivas	38
3.3.4	Funciones biyectivas	38
3.3.5	Inversa de una función	39
4	Introducción a la teoría de grafos	41
4.1	Definición de grafo	42
4.2	El TAD de los grafos	43
4.2.1	Grafos como listas de aristas	44
4.3	Generadores de grafos	46
4.4	Ejemplos de grafos	48
4.4.1	Grafo nulo	48
4.4.2	Grafo ciclo	49
4.4.3	Grafo de la amistad	49
4.4.4	Grafo completo	50
4.4.5	Grafo bipartito	51
4.4.6	Grafo estrella	52
4.4.7	Grafo rueda	52

4.4.8	Grafo circulante	53
4.4.9	Grafo de Petersen generalizado	54
4.4.10	Otros grafos importantes	54
4.5	Definiciones y propiedades	59
4.5.1	Definiciones de grafos	60
4.5.2	Propiedades de grafos	65
4.5.3	Operaciones y propiedades sobre grafos	66
4.6	Morfismos de grafos	70
4.6.1	Morfismos	70
4.6.2	Isomorfismos	71
4.6.3	Automorfismos	77
4.7	Conectividad de grafos	78
4.7.1	Caminos	78
4.7.2	Conectividad	86
4.7.3	Clases de equivalencia	86
4.7.4	Propiedades del grosor de los grafos	90
4.7.5	Propiedades de los isomorfismos	92
4.8	Sistemas utilizados	93
4.9	Mapa de decisiones de diseño	94
Bibliografía		96
Índice de definiciones		97
Lista de tareas pendientes		100

Introducción

El objetivo del trabajo es la implementación de algoritmos de matemática discreta en Haskell. Los puntos de partida son

- los temas de la asignatura “Matemática discreta” ([3]),
- los temas de la asignatura “Informática” ([1]),
- el capítulo 7 del libro “Algorithms: A functional programming approach” ([6]) y
- el artículo “Graph theory” ([9]) de la Wikipedia.

Capítulo 1

Conjuntos

El concepto de *conjunto* aparece en todos los campos de las Matemáticas, pero, ¿qué debe entenderse por él? La *Teoría de conjuntos* fue introducida por Georg Cantor (1845-1917); desde 1869, Cantor ejerció como profesor en la Universidad de Halle y entre 1879 y 1884 publicó una serie de seis artículos en el *Mathematische Annalen*, en los que hizo una introducción básica a la teoría de conjuntos. En su *Beiträge zur Begründung der transfiniten Mengenlehre*, Cantor dio la siguiente definición de conjunto:

§ 1

The Conception of Power or Cardinal Number

BY an “aggregate” (*Menge*) we are to understand any collection into a whole (*Zusammenfassung zu einem Ganzen*) M of definite and separate objects m of our intuition or our thought. These objects are called the “elements” of M .

Figura 1.1: Fragmento del texto traducido al inglés en el que Cantor da la definición de conjunto

«Debemos entender por “conjunto” (*Menge*) cualquier colección vista como un todo (*Zusammenfassung zu einem Ganzen*), M , de objetos separados y bien definidos, m , de nuestra intuición o pensamiento. Estos objetos son los “elementos” de M »

Felix Hausdorff, en 1914, dice: «un conjunto es una reunión de cosas que constituyen una totalidad; es decir, una nueva cosa», y añade: «esto puede difícilmente ser una definición, pero sirve como demostración expresiva del concepto de conjunto a través de conjuntos sencillos como el conjunto de habitantes de una ciudad o el de átomos de Hidrógeno del Sol».

Un conjunto así definido no tiene que estar compuesto necesariamente de elementos homogéneos y además, da lugar a cuestiones filosóficas como si podemos llamar

conjunto a aquel que no posee ningún elemento. Matemáticamente, conviene aceptar solo elementos que compartan alguna propiedad y definir el *conjunto vacío* como aquel que no tiene elemento alguno.

El gran mérito de Cantor fue considerar conjuntos *transfinitos* (que tiene infinitos elementos), concepto inaudito hasta avanzado el siglo XIX, hablar del *cardinal* de un conjunto como el número de sus elementos y hablar de *conjuntos equivalentes* cuando puede establecerse una biyección entre ellos; ideas ya apuntadas por Bolzano, quien se centró demasiado en el aspecto filosófico, sin llegar a formalizar sus ideas.

A lo largo de la sección, haremos una pequeña introducción a la Teoría de Conjuntos, presentando formalmente sus conceptos más importantes. A la hora de elaborar el contenido se han utilizado los siguientes recursos bibliográficos:

- el primer tema de la asignatura “Álgebra básica” ([5]),
- los temas de la asignatura “Informática” ([1]),
- el artículo de la Wikipedia “Set (mathematics)” ([8] y
- * el artículo “El regalo de Cantor” ([4]).

1.1. El TAD de los conjuntos

En la presente sección, se definen las operaciones básicas necesarias para trabajar con conjuntos en un lenguaje funcional. En nuestro caso, el lenguaje que utilizaremos será Haskell. Daremos la signatura del Tipo Abstracto de Dato (TAD) de los conjuntos y daremos algunos ejemplos de posibles representaciones de conjuntos con las que podríamos trabajar.

A continuación, presentamos las operaciones definidas en el TAD de los conjuntos:

```
vacio      :: Conj a
inserta    :: Eq a => a -> Conj a -> Conj a
elimina    :: Eq a => a -> Conj a -> Conj a
pertenece  :: Eq a => Conj a -> a -> Bool
esVacio    :: Conj a -> Bool
minimoElemento :: Ord a => Conj a -> a
```

donde,

- (*vacio*) es el conjunto vacío.
- (*inserta x c*) es el conjunto obtenido añadiendo el elemento *x* al conjunto *c*.
- (*elimina x c*) es el conjunto obtenido eliminando el elemento *x* del conjunto *c*.
- (*pertenece x c*) se verifica si *x* pertenece al conjunto *c*.
- (*esVacio c*) se verifica si *c* no tiene ningún elemento.

- `(minimoElemento c)` devuelve el mínimo elemento del conjunto `c`.

Hemos de tener en cuenta que a la hora de crear un nuevo tipo de dato con el que representar a los conjuntos, este debe ser compatible con la entidad matemática que representa.

Estas son algunas de las posibles representaciones con las que podríamos trabajar con conjuntos en Haskell:

- En primer lugar, podemos trabajar con la librería `Data.Set`; en este caso, la implementación del tipo de dato `Set` está basado en árboles binarios balanceados.
- Por otra parte, podemos definir un nuevo tipo `Conj xs` con el que trabajar con conjuntos como listas no ordenadas con duplicados, como listas no ordenadas sin duplicados o como listas ordenadas sin duplicados.
- Otra opción sería trabajar directamente con conjuntos como listas; para ello, debemos ignorar las repeticiones y el orden con vista a la igualdad de conjuntos.
- Los conjuntos que sólo contienen números (de tipo `Int`) entre 0 y $n - 1$, se pueden representar como números binarios con n bits donde el bit i ($0 \leq i < n$) es 1 si el número i pertenece al conjunto.

Nota 1.1.1. Las funciones que aparecen en la especificación del TAD no dependen de la representación que elijamos.

1.1.1. Conjuntos como listas ordenadas sin repetición

A lo largo del trabajo, utilizaré la representación de conjuntos como listas ordenadas y sin duplicados, pensando en el trabajo que realizaré en secciones venideras.

En el módulo `ConjuntosConListasOrdenadasSinRepeticion` se definen las funciones del TAD de los conjuntos dando su representación como listas ordenadas sin repetición.

```
module ConjuntosConListasOrdenadasSinRepeticion (
  Conj
  , vacio          -- Conj a
  , inserta        -- Ord a => a -> Conj a -> Conj a
  , elimina        -- Ord a => a -> Conj a -> Conj a
  , pertenece      -- Ord a => Conj a -> a -> Bool
  , esVacio        -- Conj a -> Bool
  , minimoElemento -- Ord a => Conj a -> a
  , listaAconjunto
) where
```

En las definiciones del presente módulo se usarán algunas funciones de la librería `Data.List`

Vamos a definir un nuevo tipo de dato (`Conj a`), que representa a los conjuntos como listas ordenadas sin repetición.

```
data Conj a = Cj [a]
    deriving Eq
```

```
instance (Show a) => Show (Conj a) where
    showsPrec _ (Cj s) cad = showConj s cad

showConj []      cad = showString "{}" cad
showConj (x:xs) cad = showChar '{' (shows x (showl xs cad))
    where showl []      cad = showChar '}' cad
          showl (x:xs) cad = showChar ',' (shows x (showl xs cad))
```

Las funciones básicas que definiremos a partir de este tipo coincidirán con las indicadas en el TAD de los conjuntos.

- (`vacio`) es el conjunto vacío.

```
-- | Ejemplo
-- >>> vacio
-- {}
vacio :: Conj a
vacio = Cj []
```

- (`inserta x c`) es el conjunto obtenido añadiendo el elemento `x` al conjunto `c`.

```
-- | Ejemplo
-- >>> inserta 5 vacio
-- {5}
-- >>> foldr inserta vacio [2,2,1,1,2,4,2]
-- {1,2,4}
inserta :: Ord a => a -> Conj a -> Conj a
inserta x (Cj s) = Cj (agrega x s)
    where
        agrega x []           = [x]
        agrega x s@(y:ys) | x > y = y : (agrega x ys)
                          | x < y = x : s
                          | otherwise = s
```

El valor de (`listaAconjunto xs`) es el conjunto cuyos elementos son los de la lista `xs`.

```
-- | Ejemplo
-- >>> listaAconjunto [2,2,1,1,2,4,2]
-- {1,2,4}
listaAconjunto :: Ord a => [a] -> Conj a
listaAconjunto = foldr inserta vacio
```

- (esVacio c) se verifica si c es el conjunto vacío.

```
-- | Ejemplos
-- >>> esVacio (listaAconjunto [2,5,1,3,7,5,3,2,1,9,0])
-- False
-- >>> esVacio vacio
-- True
esVacio :: Conj a -> Bool
esVacio (Cj xs) = null xs
```

- (pertenece x c) se verifica si x es un elemento del conjunto c.

```
-- | Ejemplos
-- >>> let c1 = listaAconjunto [2,5,1,3,7,5,3,2,1,9,0]
-- >>> pertenece c1 3
-- True
-- >>> pertenece c1 4
-- False
pertenece :: Ord a => Conj a -> a -> Bool
pertenece (Cj ys) x = elem x (takeWhile (<= x) ys)
```

- (elimina x c) es el conjunto obtenido eliminando el elemento x del conjunto c.

```
-- | Ejemplos
-- >>> let c1 = listaAconjunto [2,5,1,3,7,5,3,2,1,9,0]
-- >>> elimina 3 c1
-- {0,1,2,5,7,9}
-- >>> elimina 4 c1
-- {0,1,2,3,5,7,9}
elimina :: Ord a => a -> Conj a -> Conj a
elimina x (Cj s) = Cj (elimina x s) where
    elimina x [] = []
    elimina x s@(y:ys) | x > y = y : elimina x ys
                       | x < y = s
                       | otherwise = ys
```

- (minimoElemento c) devuelve el mínimo elemento del conjunto c.

```
-- | Ejemplos
-- >>> minimoElemento (listaAconjunto [2,5,1,3,7,5,3,2,1,9,0])
-- 0
-- >>> minimoElemento (listaAconjunto (['a'..'e'] ++ ['A'..'E']))
-- 'A'
minimoElemento :: Ord a => Conj a -> a
minimoElemento (Cj (x:xs)) = x
```

Comentario 1: Ampliar el módulo para definir todas las funciones de Conjuntos.lhs

1.2. Definiciones y operaciones

Definición 1.2.1. Llamaremos *conjunto* a una colección de objetos, que llamaremos *elementos*, distintos entre sí y que comparten una propiedad. Para que un conjunto esté bien definido debe ser posible discernir si un objeto arbitrario está o no en él.

Si el elemento a pertenece al conjunto A , escribiremos $a \in A$. En caso contrario escribiremos $a \notin A$.

Nota 1.2.1. En Haskell, para poder discernir si un objeto arbitrario pertenece a un conjunto se necesita que su tipo pertenezca a la clase `Eq`.

Nota 1.2.2. Al trabajar con la representación de conjuntos como listas en Haskell, hemos de cuidar que los ejemplos con los que trabajemos no tengan elementos repetidos. La función `(nub xs)` de la librería `Data.List` elimina los elementos repetidos de una lista.

Los conjuntos pueden definirse de manera explícita, citando todos sus elementos entre llaves, de manera implícita, dando una o varias características que determinen si un objeto dado está o no en el conjunto. Por ejemplo, los conjuntos $\{1, 2, 3, 4\}$ y $\{x \in \mathbb{N} \mid 1 \leq x \leq 4\}$ son el mismo, definido de forma explícita e implícita respectivamente.

Nota 1.2.3. La definición implícita es necesaria cuando el conjunto en cuestión tiene una cantidad infinita de elementos. En general, los conjuntos se notarán con letras mayúsculas: A, B, \dots y los elementos con letras minúsculas: a, b, \dots .

Cuando trabajamos con conjuntos concretos, siempre existe un contexto donde esos conjuntos existen. Por ejemplo, si $A = \{-1, 1, 2, 3, 4, 5\}$ y $B = \{x \mid x \in \mathbb{N} \text{ es par}\}$ el contexto donde podemos considerar A y B es el conjunto de los números enteros, \mathbb{Z} . En general, a este conjunto se le denomina *conjunto universal*. De una forma algo más precisa, podemos dar la siguiente definición:

Definición 1.2.2. El *conjunto universal*, que notaremos por U , es un conjunto del que son subconjuntos todos los posibles conjuntos que originan el problema que tratamos.

1.2.1. Conjunto unitario

Definición 1.2.3. Un conjunto con un único elemento se denomina *unitario*.

Nota 1.2.4. Notemos que, si $X = \{x\}$ es un conjunto unitario, debemos distinguir entre el conjunto X y el elemento x .

La función `(esUnitario c)` se verifica si el conjunto c es unitario.

```
ghci> let c1 = foldr inserta vacio (take 10 (repeat 5))
ghci> let c2 = foldr inserta vacio "valverde"
```

```
ghci> let c3 = foldr inserta vacio "coco"
esUnitario c1 == True
esUnitario c2 == False
esUnitario c2 == False
```

```
esUnitario :: Ord a => Conj a -> Bool
esUnitario c | esVacio c = False
              | otherwise = esVacio (elimina (minimoElemento c) c)
```

1.2.2. Subconjuntos

Definición 1.2.4. *Dados dos conjuntos A y B , si todo elemento de A es a su vez elemento de B diremos que A es un subconjunto de B y lo notaremos $A \subseteq B$. En caso contrario se notará $A \not\subseteq B$.*

La función (`esSubconjunto c1 c2`) se verifica si $c1$ es un subconjunto de $c2$.

```
ghci> let c1 = foldr inserta vacio [4,2]
ghci> let c2 = foldr inserta vacio [3,2,4]
ghci> let c3 = foldr inserta vacio [4,2,1]
ghci> let c4 = foldr inserta vacio [1,2,4]
c1 'esSubconjunto' c2 == True
c1 'esSubconjunto' vacio == False
vacio 'esSubconjunto' c2 == True
c3 'esSubconjunto' c4 == True
c2 'esSubconjunto' c1 == False
```

```
esSubconjunto :: Ord a => Conj a -> Conj a -> Bool
esSubconjunto c1 c2
  | esVacio c1 = True
  | pertenece c2 (min c1) = esSubconjunto (elimina (min c1) c1) c2
  | otherwise = False
  where min = minimoElemento
```

1.2.3. Igualdad de conjuntos

Definición 1.2.5. *Dados dos conjuntos A y B , diremos que son **iguales** si tienen los mismos elementos; es decir, si se verifica que $A \subseteq B$ y $B \subseteq A$. Lo notaremos $A = B$.*

La función (`conjuntosIguales c1 c2`) se verifica si los conjuntos xs y ys son iguales. Por ejemplo,

```
ghci> let c1 = foldr inserta vacio [4,2]
ghci> let c2 = foldr inserta vacio [3,2,4]
ghci> let c3 = foldr inserta vacio [4,2,1]
ghci> let c4 = foldr inserta vacio [1,2,4]
ghci> let c5 = foldr inserta vacio [4,4,4,4,4,4,2]
conjuntosIguales c1 c2 == False
conjuntosIguales c3 c4 == True
conjuntosIguales c1 c5 == True
```

```
conjuntosIguales :: Ord a => Conj a -> Conj a -> Bool
conjuntosIguales c1 c2 =
    esSubconjunto c1 c2 && esSubconjunto c2 c1
```

1.2.4. Subconjuntos propios

Definición 1.2.6. Los subconjuntos de A distintos del \emptyset y del mismo A se denominan **subconjuntos propios** de A .

La función (`esSubconjuntoPropio c1 c2`) se verifica si $c1$ es un subconjunto propio de $c2$. Por ejemplo,

```
[4,2]    'esSubconjuntoPropio' [3,2,4] == True
[4,2,1]  'esSubconjuntoPropio' [1,2,4] == False
```

```
esSubconjuntoPropio :: Ord a => Conj a -> Conj a -> Bool
esSubconjuntoPropio c1 c2
    | esVacio c1 = False
    | conjuntosIguales c1 c2 = False
    | otherwise = esSubconjunto c1 c2
```

1.2.5. Complementario de un conjunto

Definición 1.2.7. Dado un conjunto A , se define el **complementario** de A , que notaremos por \overline{A} como:

$$\overline{A} = \{x | x \in U, x \notin A\}$$

La función (`complementario u c`) devuelve el complementario del conjunto c y en el universo u . Por ejemplo,


```
ghci> let u  = foldr inserta vacio [1..9]
ghci> let c1 = foldr inserta vacio [3,2,5,7]
ghci> let c2 = foldr inserta vacio [1,4,6,8,9]
complementario u c1      ==  1,4,6,8,9
complementario u u       ==
complementario u vacio   ==  1,2,3,4,5,6,7,8,9
complementario u c2      ==  2,3,5,7
```

```
complementario :: Ord a => Conj a -> Conj a -> Conj a
complementario u c
  | esVacio c = u
  | otherwise =
    complementario (elimina (min c) u) (elimina (min c) c)
    where min = minimoElemento
```

1.2.6. Cardinal de un conjunto

Definición 1.2.8. Dado un conjunto finito A , denominaremos **cardinal** de A al número de elementos que tiene y lo notaremos $|A|$.

La función (`cardinal xs`) devuelve el cardinal del conjunto `xs`. Por ejemplo,

```
cardinal vacio                ==  0
cardinal (foldr inserta vacio [1..10]) == 10
cardinal (foldr inserta vacio "chocolate") == 7
```

```
cardinal :: Ord a => Conj a -> Int
cardinal c | esVacio c = 0
           | otherwise = 1 + cardinal (elimina (min c) c)
           where min = minimoElemento
```

1.2.7. Unión de conjuntos

Definición 1.2.9. Dados dos conjuntos A y B se define la *unión* de A y B , notado $A \cup B$, como el conjunto formado por aquellos elementos que pertenecen al menos a uno de los dos conjuntos, A ó B ; es decir,

$$A \cup B = \{x | x \in A \vee x \in B\}$$

La función (`unionConjuntos c1 c2`) devuelve la unión de los conjuntos `xs` y `ys`. Por ejemplo,

```
ghci> let c1 = foldr inserta vacio [1,3..9]
ghci> let c2 = foldr inserta vacio [2,4..9]
ghci> let c3 = foldr inserta vacio "centri"
ghci> let c4 = foldr inserta vacio "fugado"
ghci> unionConjuntos c1 c2
1,3,5,7,9,2,4,6,8
ghci> unionConjuntos c3 c4
'a','c','d','e','f','g','i','n','o','r','t','u'
```

```
unionConjuntos :: Ord a => Conj a -> Conj a -> Conj a
unionConjuntos c1 c2
  | esVacio c1 = c2
  | esVacio c2 = c1
  | otherwise =
    unionConjuntos (elimina (min c1) c1) (inserta (min c1) c2)
    where min = minimoElemento
```

1.2.8. Intersección de conjuntos

Definición 1.2.10. *Dados dos conjuntos A y B se define la intersección de A y B , notado $A \cap B$, como el conjunto formado por aquellos elementos que pertenecen a cada uno de los dos conjuntos, A y B , es decir,*

$$A \cap B = \{x | x \in A \wedge x \in B\}$$

La función (interseccion c1 c2) devuelve la intersección de los conjuntos c1 y c2. Por ejemplo,

```
ghci> let c1 = foldr inserta vacio [1,3..20]
ghci> let c2 = foldr inserta vacio [2,4..20]
ghci> let c3 = foldr inserta vacio [2,4..30]
ghci> let c4 = foldr inserta vacio [4,8..30]
ghci> let c5 = foldr inserta vacio "noche"
ghci> let c6 = foldr inserta vacio "dia"
interseccion c1 c2 ==
interseccion c3 c4 == 4,8,12,16,20,24,28
interseccion c5 c6 ==
```

```
interseccion :: Ord a => Conj a -> Conj a -> Conj a
interseccion c1 c2
  | esVacio c1 || esVacio c2 = vacio
  | m1 < m2 = interseccion (elimina m1 c1) c2
  | m1 > m2 = interseccion c1 (elimina (m2) c2)
  | otherwise =
    inserta m1 (interseccion (elimina m1 c1) (elimina m2 c2))
```

```
where m1 = minimoElemento c1
      m2 = minimoElemento c2
```

1.2.9. Producto cartesiano

Definición 1.2.11. El *producto cartesiano*¹ de dos conjuntos A y B es una operación sobre ellos que resulta en un nuevo conjunto $A \times B$ que contiene a todos los pares ordenados tales que la primera componente pertenece a A y la segunda pertenece a B ; es decir, $A \times B = \{(a,b) | a \in A, b \in B\}$.

La función (`productoCartesiano c1 c2`) devuelve el producto cartesiano de x s e y s. Por ejemplo,

```
ghci> let c1 = foldr inserta vacio [3,1]
ghci> let c2 = foldr inserta vacio [2,4,7]
ghci> productoCartesiano c1 c2
(1,2),(1,4),(1,7),(3,2),(3,4),(3,7)
ghci> productoCartesiano c2 c1
(2,1),(2,3),(4,1),(4,3),(7,1),(7,3)
```

```
productoCartesiano :: (Ord a, Ord b) => Conj a -> Conj b -> Conj (a,b)
productoCartesiano c1 c2
  | esVacio c1 || esVacio c2 = vacio
  | otherwise =
    u (productoUnitario m1 c2) (productoCartesiano (elimina m1 c1) c2)
  where u = unionConjuntos
        m1 = minimoElemento c1
        m2 = minimoElemento c2

productoUnitario :: (Ord a, Ord b) => a -> Conj b -> Conj (a,b)
productoUnitario a c
  | esVacio c = vacio
  | otherwise =
    inserta (a, min c) (productoUnitario a (elimina (min c) c))
  where min = minimoElemento
```

1.2.10. Combinaciones

Definición 1.2.12. Las *combinaciones* de un conjunto S tomados en grupos de n son todos los subconjuntos de S con n elementos.

¹https://en.wikipedia.org/wiki/Cartesian_product

La función (`combinaciones n xs`) devuelve las combinaciones de los elementos de `xs` en listas de `n` elementos. Por ejemplo,

```
combinaciones 3 ['a'..'d'] == ["abc","abd","acd","bcd"]
combinaciones 2 [2,4..8]   == [[2,4],[2,6],[2,8],[4,6],[4,8],[6,8]]
```

```
combinaciones :: Integer -> [a] -> [[a]]
combinaciones 0 _      = [[]]
combinaciones _ []     = []
combinaciones k (x:xs) =
  [x:ys | ys <- combinaciones (k-1) xs] ++ combinaciones k xs
```

1.2.11. Variaciones con repetición

Definición 1.2.13. Las *variaciones con repetición* de m elementos tomados en grupos de n es el número de diferentes n -tuplas de un conjunto de m elementos.

La función (`variacionesR n xs`) devuelve las variaciones con con repetición de los elementos de `xs` en listas de `n` elementos. Por ejemplo,

```
ghci> variacionesR 3 ['a','b']
["aaa","aab","aba","abb","baa","bab","bba","bbb"]
ghci> variacionesR 2 [2,4..8]
[[2,2],[2,4],[2,6],[2,8],[4,2],[4,4],[4,6],[4,8],
 [6,2],[6,4],[6,6],[6,8],[8,2],[8,4],[8,6],[8,8]]
```

```
variacionesR :: Int -> [a] -> [[a]]
variacionesR _ [] = [[]]
variacionesR 0 _ = [[]]
variacionesR k us =
  [u:vs | u <- us, vs <- variacionesR (k-1) us]
```

Capítulo 2

Conjuntos

El concepto de *conjunto* aparece en todos los campos de las Matemáticas, pero, ¿qué debe entenderse por él? La *Teoría de conjuntos* fue introducida por Georg Cantor (1845-1917); desde 1869, Cantor ejerció como profesor en la Universidad de Halle y entre 1879 y 1884 publicó una serie de seis artículos en el *Mathematische Annalen*, en los que hizo una introducción básica a la teoría de conjuntos. En su *Beiträge zur Begründung der transfiniten Mengenlehre*, Cantor dio la siguiente definición de conjunto:

§ 1

The Conception of Power or Cardinal Number

BY an “aggregate” (*Menge*) we are to understand any collection into a whole (*Zusammenfassung zu einem Ganzen*) M of definite and separate objects m of our intuition or our thought. These objects are called the “elements” of M .

Figura 2.1: Fragmento del texto traducido al inglés en el que Cantor da la definición de conjunto

«Debemos entender por “conjunto” (*Menge*) cualquier colección vista como un todo (*Zusammenfassung zu einem Ganzen*), M , de objetos separados y bien definidos, m , de nuestra intuición o pensamiento. Estos objetos son los “elementos” de M »

Felix Hausdorff, en 1914, dice: «un conjunto es una reunión de cosas que constituyen una totalidad; es decir, una nueva cosa», y añade: «esto puede difícilmente ser una definición, pero sirve como demostración expresiva del concepto de conjunto a través de conjuntos sencillos como el conjunto de habitantes de una ciudad o el de átomos de Hidrógeno del Sol».

Un conjunto así definido no tiene que estar compuesto necesariamente de elementos homogéneos y además, da lugar a cuestiones filosóficas como si podemos llamar

conjunto a aquel que no posee ningún elemento. Matemáticamente, conviene aceptar solo elementos que compartan alguna propiedad y definir el *conjunto vacío* como aquel que no tiene elemento alguno.

El gran mérito de Cantor fue considerar conjuntos *transfinitos* (que tiene infinitos elementos), concepto inaudito hasta avanzado el siglo XIX, hablar del *cardinal* de un conjunto como el número de sus elementos y hablar de *conjuntos equivalentes* cuando puede establecerse una biyección entre ellos; ideas ya apuntadas por Bolzano, quien se centró demasiado en el aspecto filosófico, sin llegar a formalizar sus ideas.

A lo largo de la sección, haremos una pequeña introducción a la Teoría de Conjuntos, presentando formalmente sus conceptos más importantes. A la hora de elaborar el contenido se han utilizado los siguientes recursos bibliográficos:

- el primer tema de la asignatura “Álgebra básica” ([5]),
- los temas de la asignatura “Informática” ([1])
- el artículo de la Wikipedia “Set (mathematics)” ([8]) y
- el artículo “El regalo de Cantor” ([4])

2.1. Definiciones y propiedades

Definición 2.1.1. Llamaremos *conjunto* a una colección de objetos, que llamaremos *elementos*, distintos entre sí y que comparten una propiedad. Para que un conjunto esté bien definido debe ser posible discernir si un objeto arbitrario está o no en él.

Nota 2.1.1. En Haskell, para poder discernir si un objeto arbitrario pertenece a un conjunto, hay que restringirse a tipos de la clase Eq.

Nota 2.1.2. Al trabajar con la representación de conjuntos como listas en Haskell, hemos de cuidar que los ejemplos con los que trabajemos no tengan elementos repetidos. La función (`nub xs`) de la librería `Data.List` elimina los elementos repetidos de una lista. Además, deberemos indicar de qué tipo algebraico serán los elementos de un conjunto (necesariamente contenido en la clase de los comparables por igualdad (es decir, la clase Eq)).

Los conjuntos pueden definirse de manera explícita, citando todos sus elementos entre llaves, de manera implícita, dando una o varias características que determinen si un objeto dado está o no en el conjunto. Por ejemplo, los conjuntos $\{1, 2, 3, 4\}$ y $\{x \in \mathbb{N} \mid 1 \leq x \leq 4\}$ son el mismo, definido de forma explícita e implícita respectivamente.

Nota 2.1.3. La definición implícita es necesaria cuando el conjunto en cuestión tiene una cantidad infinita de elementos. En general, los conjuntos se notarán con letras mayúsculas: A, B, \dots y los elementos con letras minúsculas: a, b, \dots .

Cuando trabajamos con conjuntos concretos, siempre existe un contexto donde esos conjuntos existen. Por ejemplo, si $A = \{-1, 1, 2, 3, 4, 5\}$ y $B = \{x | x \in \mathbb{N} \text{ es par}\}$ el contexto donde podemos considerar A y B es el conjunto de los números enteros, \mathbb{Z} . En general, a este conjunto se le denomina *conjunto universal*. De una forma algo más precisa, podemos dar la siguiente definición:

Definición 2.1.2. El *conjunto universal*, que notaremos por U , es un conjunto del que son subconjuntos todos los posibles conjuntos que originan el problema que tratamos.

2.1.1. Pertenencia a un conjunto

Si el elemento a pertenece al conjunto A , escribiremos $a \in A$. En caso contrario escribiremos $a \notin A$.

La función `(pertenece x xs)` se verifica si x pertenece al conjunto xs .

```
-- | Ejemplos
-- >>> 9 'pertenece' [1..6]
-- False
-- >>> 'c' 'pertenece' "Roca"
-- True
pertenece :: Eq a => a -> [a] -> Bool
pertenece = elem
```

2.1.2. Conjunto vacío

Definición 2.1.3. El conjunto que carece de elementos se denomina *conjunto vacío* y se denota por \emptyset .

La función `conjuntoVacio` devuelve el conjunto vacío y la función `(esVacio xs)` se verifica si el conjunto xs es vacío.

```
-- | Ejemplos
-- >>> esVacio [1..6]
-- False
-- >>> esVacio conjuntoVacio
-- True
conjuntoVacio :: [a]
conjuntoVacio = []

esVacio :: [a] -> Bool
esVacio = null
```

2.1.3. Conjunto unitario

Definición 2.1.4. *Un conjunto con un único elemento se denomina **unitario**.*

Nota 2.1.4. Notemos que, si $X = \{x\}$ es un conjunto unitario, debemos distinguir entre el conjunto X y el elemento x .

La función (`esUnitario xs`) se verifica si el conjunto `xs` es unitario.

```
-- | Ejemplos
-- >>> esUnitario [5]
-- True
-- >>> esUnitario [5,3]
-- False
esUnitario :: Eq a => [a] -> Bool
esUnitario xs = length xs == 1
```

2.1.4. Subconjuntos

Definición 2.1.5. *Dados dos conjuntos A y B , si todo elemento de A es a su vez elemento de B diremos que A es un subconjunto de B y lo notaremos $A \subseteq B$. En caso contrario se notará $A \not\subseteq B$.*

La función (`esSubconjunto xs ys`) se verifica si `xs` es un subconjunto de `ys`.

```
-- | Ejemplos
-- >>> [4,2] 'esSubconjunto' [3,2,4]
-- True
-- >>> [1,2] 'esSubconjunto' conjuntoVacio
-- False
-- >>> conjuntoVacio 'esSubconjunto' [1,2]
-- True
-- >>> [4,2,1] 'esSubconjunto' [1,2,4]
-- True
-- >>> [3,2,4] 'esSubconjunto' [5,2]
-- False
esSubconjunto :: Eq a => [a] -> [a] -> Bool
esSubconjunto xs ys = all ('pertenece' ys) xs
```

2.1.5. Igualdad de conjuntos

Definición 2.1.6. *Dados dos conjuntos A y B , diremos que son **iguales** si tienen los mismos elementos; es decir, si se verifica que $A \subseteq B$ y $B \subseteq A$. Lo notaremos $A = B$.*

La función (`conjuntosIguales xs ys`) se verifica si los conjuntos `xs` y `ys` son iguales.

```
-- | Ejemplos
-- >>> conjuntosIguales [4,2] [4,2,4]
-- True
-- >>> conjuntosIguales [4,2,3] [4,3,2]
-- True
-- >>> conjuntosIguales [5,2] [3,2,4]
-- False
conjuntosIguales :: Eq a => [a] -> [a] -> Bool
conjuntosIguales xs ys =
    esSubconjunto xs ys && esSubconjunto ys xs
```

2.1.6. Subconjuntos propios

Definición 2.1.7. Los subconjuntos de A distintos del \emptyset y del mismo A se denominan **subconjuntos propios** de A .

La función (`esSubconjuntoPropio xs ys`) se verifica si `xs` es un subconjunto propio de `ys`.

```
-- | Ejemplos
-- >>> [4,2] 'esSubconjuntoPropio' [3,2,4]
-- True
-- >>> [4,2,1] 'esSubconjuntoPropio' [1,2,4]
-- False
esSubconjuntoPropio :: Eq a => [a] -> [a] -> Bool
esSubconjuntoPropio xs ys =
    xs 'esSubconjunto' ys && not (conjuntosIguales xs ys)
```

2.1.7. Complementario de un conjunto

Definición 2.1.8. Dado un conjunto A , se define el **complementario** de A , que notaremos por \overline{A} como:

$$\overline{A} = \{x | x \in U, x \notin A\}$$

La función (`complementario xs ys`) devuelve el complementario de `ys` respecto de `xs`.

```
-- | Ejemplo
-- >>> complementario [1..9] [3,2,5,7]
-- [1,4,6,8,9]
complementario :: Eq a => [a] -> [a] -> [a]
complementario = (\\)
```

2.1.8. Cardinal de un conjunto

Definición 2.1.9. Dado un conjunto finito A , denominaremos *cardinal* de A al número de elementos que tiene y lo notaremos $|A|$.

La función (`cardinal xs`) devuelve el cardinal del conjunto `xs`.

```
-- | Ejemplos
-- >>> cardinal conjuntoVacio
-- 0
-- >>> cardinal [1..10]
-- 10
cardinal :: Eq a => [a] -> Int
cardinal = length
```

2.1.9. Unión de conjuntos

Definición 2.1.10. Dados dos conjuntos A y B se define la *unión* de A y B , notado $A \cup B$, como el conjunto formado por aquellos elementos que pertenecen al menos a uno de los dos conjuntos, A ó B ; es decir,

$$A \cup B = \{x | x \in A \vee x \in B\}$$

La función (`unionConjuntos xs ys`) devuelve la unión de los conjuntos `xs` y `ys`.

```
-- | Ejemplos
-- >>> unionConjuntos [1,3..9] [2,4..9]
-- [1,3,5,7,9,2,4,6,8]
-- >>> unionConjuntos "centri" "fugado"
-- "centrifugado"
unionConjuntos :: Eq a => [a] -> [a] -> [a]
unionConjuntos = union
```

Nota 2.1.5. Para ahorrar en escritura, en el futuro utilizaremos la función (`union xs ys`) definida en el módulo `Data.List`, equivalente a (`unionConjuntos xs ys`)

Definición 2.1.11. Dada una familia de conjuntos $\{A_i\}_i$ con $i \in I$, se define la *unión general* de los conjuntos A_i notado $\bigcup_{i \in I} A_i$, como el conjunto formado por aquellos elementos que pertenecen al menos a uno de los conjuntos de la familia; es decir,

$$\bigcup_{i \in I} A_i = \{x | x \in A_i \text{ con } i \in I\}$$

La función (`unionGeneral xss`) devuelve la unión general de la familia de conjuntos de la lista `xss`.

```
-- | Ejemplos
-- >>> unionGeneral [[1,4..15],[2,5..15],[3,6..15]]
-- [1,4,7,10,13,2,5,8,11,14,3,6,9,12,15]
-- >>> unionGeneral ["m","u","r","c","i","e","l","a","g","o"]
-- "murcielago"
unionGeneral :: Eq a => [[a]] -> [a]
unionGeneral = foldr union []
```

2.1.10. Intersección de conjuntos

Definición 2.1.12. *Dados dos conjuntos A y B se define la intersección de A y B , notado $A \cap B$, como el conjunto formado por aquellos elementos que pertenecen a cada uno de los dos conjuntos, A y B , es decir,*

$$A \cap B = \{x | x \in A \wedge x \in B\}$$

La función (interseccion xs ys) devuelve la intersección de los conjuntos xs y ys.

```
-- | Ejemplos
-- >>> interseccion [1,3..20] [2,4..20]
-- []
-- >>> interseccion [2,4..30] [4,8..30]
-- [4,8,12,16,20,24,28]
-- >>> interseccion "noche" "dia"
-- ""
interseccion :: Eq a => [a] -> [a] -> [a]
interseccion = intersect
```

2.1.11. Producto cartesiano

Definición 2.1.13. El **producto cartesiano**¹ de dos conjuntos A y B es una operación sobre ellos que resulta en un nuevo conjunto $A \times B$ que contiene a todos los pares ordenados tales que la primera componente pertenece a A y la segunda pertenece a B ; es decir, $A \times B = \{(a,b) | a \in A, b \in B\}$.

La función (productoCartesiano xs ys) devuelve el producto cartesiano de xs e ys.

```
-- | Ejemplo
-- >>> productoCartesiano [3,1] [2,4,7]
```

¹https://en.wikipedia.org/wiki/Cartesian_product

```
-- [(3,2),(3,4),(3,7),(1,2),(1,4),(1,7)]
productoCartesiano :: [a] -> [b] -> [(a,b)]
productoCartesiano xs ys =
  [(x,y) | x <- xs, y <- ys]
```

2.1.12. Combinaciones

Definición 2.1.14. Las *combinaciones* de un conjunto S tomados en grupos de n son todos los subconjuntos de S con n elementos.

La función `(combinaciones n xs)` devuelve las combinaciones de los elementos de `xs` en listas de `n` elementos.

```
-- | Ejemplos
-- >>> combinaciones 3 ['a'..'d']
-- ["abc","abd","acd","bcd"]
-- >>> combinaciones 2 [2,4..8]
-- [[2,4],[2,6],[2,8],[4,6],[4,8],[6,8]]
combinaciones :: Integer -> [a] -> [[a]]
combinaciones 0 _ = [[]]
combinaciones _ [] = []
combinaciones k (x:xs) =
  [x:ys | ys <- combinaciones (k-1) xs] ++ combinaciones k xs
```

2.1.13. Variaciones con repetición

Definición 2.1.15. Las *variaciones con repetición* de m elementos tomados en grupos de n es el número de diferentes n -tuplas de un conjunto de m elementos.

La función `(variacionesR n xs)` devuelve las variaciones con con repetición de los elementos de `xs` en listas de `n` elementos.

```
-- | Ejemplos
-- >>> variacionesR 3 ['a','b']
-- ["aaa","aab","aba","abb","baa","bab","bba","bbb"]
-- >>> variacionesR 2 [2,3,5]
-- [[2,2],[2,3],[2,5],[3,2],[3,3],[3,5],[5,2],[5,3],[5,5]]
variacionesR :: Int -> [a] -> [[a]]
variacionesR _ [] = [[]]
variacionesR 0 _ = [[]]
variacionesR k us =
  [u:vs | u <- us, vs <- variacionesR (k-1) us]
```

Capítulo 3

Relaciones y funciones

3.1. Relaciones

Las relaciones que existen entre personas, números, conjuntos y muchas otras entidades pueden formalizarse en la idea de relación binaria. En esta sección se define y desarrolla este concepto, particularizando en el caso de las relaciones homogéneas y en el de las funciones.

3.1.1. Relación binaria

Definición 3.1.1. Una *relación binaria*¹ (o *correspondencia*) entre dos conjuntos A y B es un subconjunto del producto cartesiano $A \times B$.

La función `(esRelacion xs ys r)` se verifica si r es una relación binaria de xs en ys . Por ejemplo,

```
-- | Ejemplos
-- >>> esRelacion [3,1] [2,4,7] [(3,4),(1,2)]
-- True
-- >>> esRelacion [3,1] [2,4,7] [(3,1),(1,2)]
-- False
esRelacion :: (Eq a, Eq b) => [a] -> [b] -> [(a,b)] -> Bool
esRelacion xs ys r =
  r `esSubconjunto` productoCartesiano xs ys
```

3.1.2. Imagen por una relación

Definición 3.1.2. Si R es una relación binaria, la *imagen del elemento* x en la relación R es el conjunto de los valores correspondientes a x en R .

¹https://en.wikipedia.org/wiki/Binary_relation

La función (`imagenRelacion r x`) es la imagen de x en la relación r .

```
-- | Ejemplos
-- >>> imagenRelacion [(1,3),(2,5),(1,4)] 1
-- [3,4]
-- >>> imagenRelacion [(1,3),(2,5),(1,4)] 2
-- [5]
-- >>> imagenRelacion [(1,3),(2,5),(1,4)] 3
-- []
imagenRelacion :: (Eq a, Eq b) => [(a,b)] -> a -> [b]
imagenRelacion r x =
  nub [y | (z,y) <- r, z == x]
```

3.1.3. Dominio de una relación

Definición 3.1.3. Dada una relación binaria R , su **dominio** es el conjunto que contiene a todos los valores que se toman en la relación R .

La función (`dominio r`) devuelve el dominio de la relación r .

```
-- | Ejemplo
-- >>> dominio [(3,2),(5,1),(3,4)]
-- [3,5]
dominio :: Eq a => [(a,b)] -> [a]
dominio r = nub (map fst r)
```

3.1.4. Rango de una relación

Definición 3.1.4. El **rango** de una relación binaria R es el conjunto de las imágenes de mediante R .

La función (`rango r`) devuelve el rango de la relación binaria r .

```
-- | Ejemplo
-- >>> rango [(3,2),(5,2),(3,4)]
-- [2,4]
rango :: Eq b => [(a,b)] -> [b]
rango r = nub (map snd r)
```

3.1.5. Antiimagen por una relación

Definición 3.1.5. La *antiimagen del elemento y por una relación r* es el conjunto de los elementos cuya imagen es y .

La `(antiImagenRelacion r y)` es la antiimagen del elemento y en la relación binaria r .

```
-- | Ejemplo
-- >>> antiImagenRelacion [(1,3),(2,3),(7,4)] 3
-- [1,2]
antiImagenRelacion :: (Eq a, Eq b) => [(a,b)] -> b -> [a]
antiImagenRelacion r y =
  nub [x | (x,z) <- r, z == y]
```

3.1.6. Relación funcional

Definición 3.1.6. Dada una relación binaria R , se dice **funcional** si todos los elementos de su dominio tienen una única imagen en R .

La función `(esFuncional r)` se verifica si la relación r es funcional.

```
-- | Ejemplos
-- >>> esFuncional [(3,2),(5,1),(7,9)]
-- True
-- >>> esFuncional [(3,2),(5,1),(3,4)]
-- False
-- >>> esFuncional [(3,2),(5,1),(3,2)]
-- True
esFuncional :: (Eq a, Eq b) => [(a,b)] -> Bool
esFuncional r =
  and [esUnitario (imagenRelacion r x) | x <- dominio r]
```

3.2. Relaciones homogéneas

Para elaborar la presente sección, se han consultado los apuntes de “Álgebra básica” ([5]), asignatura del primer curso del Grado en Matemáticas.

Definición 3.2.1. Una relación binaria entre dos conjuntos A y B se dice que es **homogénea** si los conjuntos son iguales; es decir, si $A = B$. Si el par $(x,y) \in A \times A$ está en la relación homogénea R , diremos que x está R -relacionado con y , o relacionado con y por R . Esto se notará frecuentemente xRy (nótese que el orden es importante).

La función (`esRelacionHomogenea xs r`) se verifica si r es una relación binaria homogénea en el conjunto xs .

```
-- | Ejemplos
-- >>> esRelacionHomogenea [1..4] [(1,2),(2,4),(3,4),(4,1)]
-- True
-- >>> esRelacionHomogenea [1..4] [(1,2),(2,5),(3,4),(4,1)]
-- False
-- >>> esRelacionHomogenea [1..4] [(1,2),(3,4),(4,1)]
-- True
esRelacionHomogenea :: Eq a => [a] -> [(a,a)] -> Bool
esRelacionHomogenea xs = esRelacion xs xs
```

Nota 3.2.1. El segundo argumento que recibe la función ha de ser una lista de pares con ambas componentes del mismo tipo.

La función (`estaRelacionado r x y`) se verifica si x está relacionado con y en la relación homogénea r .

```
-- | Ejemplos
-- >>> estaRelacionado [(1,3),(2,5),(4,6)] 2 5
-- True
-- >>> estaRelacionado [(1,3),(2,5),(4,6)] 2 3
-- False
estaRelacionado :: Eq a => [(a,a)] -> a -> a -> Bool
estaRelacionado r x y = (x,y) `elem` r
```

3.2.1. Relaciones reflexivas

Definición 3.2.2. Sea R una relación binaria homogénea en el conjunto A . Diremos que R es *reflexiva* cuando todos los elementos de A están relacionados por R consigo mismos; es decir, cuando $\forall x \in A$ se tiene que xRx .

La función (`esReflexiva xs r`) se verifica si la relación r en xs es reflexiva.

```
-- | Ejemplos
-- >>> esReflexiva [1,2] [(1,1),(1,2),(2,2)]
-- True
-- >>> esReflexiva [1,2] [(1,1),(1,2)]
-- False
esReflexiva :: Eq a => [a] -> [(a,a)] -> Bool
esReflexiva xs r = zip xs xs `esSubconjunto` r
```

Nota 3.2.2. En el conjunto \mathbb{N} , las relaciones caracterizadas por:

- $xRy \iff x \leq y$,
- $xSy \iff x - y$ es par,
- $xTy \iff x$ divide a y ,

son relaciones binarias homogéneas reflexivas.

3.2.2. Relaciones simétricas

Definición 3.2.3. Diremos que una relación homogénea R es **simétrica** cuando

$$\forall (x, y) \in R \longrightarrow (y, x) \in R$$

La función (`esSimetrica r`) se verifica si la relación r es simétrica.

```
-- | Ejemplos
-- >>> esSimetrica [(1,1),(1,2),(2,1)]
-- True
-- >>> esSimetrica [(1,1),(1,2),(2,2)]
-- False
esSimetrica :: Eq a => [(a,a)] -> Bool
esSimetrica r = [(y,x) | (x,y) <- r] 'esSubconjunto' r
```

Nota 3.2.3. En el conjunto \mathbb{N} , la relación caracterizada por $xSy \iff x - y$ es par, es una relación binaria homogénea simétrica.

3.2.3. Relaciones antisimétricas

Definición 3.2.4. Sea R una relación binaria homogénea en el conjunto A . Diremos que R es **antisimétrica** cuando

$$\forall (x, y) [(x, y) \in R \wedge (y, x) \in R \longrightarrow x = y]$$

La función (`esAntisimetrica r`) se verifica si la relación r es antisimétrica.

```
-- | Ejemplos
-- >>> esAntisimetrica [(1,2),(3,1)]
-- True
-- >>> esAntisimetrica [(1,2),(2,1)]
-- False
esAntisimetrica :: Eq a => [(a,a)] -> Bool
esAntisimetrica r =
  and [x == y | (x,y) <- r, (y,x) 'elem' r]
```

Nota 3.2.4. En el conjunto \mathbb{N} , las relaciones caracterizadas por:

- $xRy \longleftrightarrow x \leq y$,
- $xTy \longleftrightarrow x$ divide a y ,
- $xRy \longleftrightarrow x < y$,

son relaciones binarias homogéneas antisimétricas.

3.2.4. Relaciones transitivas

Definición 3.2.5. Sea R una relación binaria homogénea en el conjunto A . Diremos que R es *transitiva* cuando $\forall (x, y), (y, z) \in R$ se tiene que xRy e $yRz \longrightarrow xRz$.

La función (`esTransitiva r`) se verifica si la relación r es transitiva.

```
-- | Ejemplos
-- >>> esTransitiva [(1,2),(2,3),(1,3)]
-- True
-- >>> esTransitiva [(1,2),(2,3)]
-- False
esTransitiva :: Eq a => [(a,a)] -> Bool
esTransitiva r =
  [(x,z) | (x,y) <- r, (w,z) <- r, y == w] 'esSubconjunto' r
```

Nota 3.2.5. En el conjunto \mathbb{N} , las relaciones caracterizadas por:

- $xRy \longleftrightarrow x \leq y$,
- $xSy \longleftrightarrow x - y$ es par,
- $xTy \longleftrightarrow x$ divide a y ,
- $xRy \longleftrightarrow x < y$,

son relaciones binarias homogéneas transitivas.

3.2.5. Relaciones de equivalencia

Definición 3.2.6. Las relaciones homogéneas que son a la vez reflexivas, simétricas y transitivas se denominan *relaciones de equivalencia*.

La función (`esRelacionEquivalencia xs r`) se verifica si r es una relación de equivalencia en xs .

```
-- | Ejemplos
-- >>> esRelacionEquivalencia [1..3] [(1,1),(2,2),(3,3),(1,2),(2,1)]
-- True
-- >>> esRelacionEquivalencia [1..3] [(2,2),(3,3),(1,2),(2,1)]
-- False
-- >>> esRelacionEquivalencia [1..3] [(1,1),(2,2),(3,3),(1,2)]
-- False
esRelacionEquivalencia :: Eq a => [a] -> [(a,a)] -> Bool
```

```
esRelacionEquivalencia xs r =
  esReflexiva xs r    &&
  esSimetrica r       &&
  esTransitiva r
```

Nota 3.2.6. En el conjunto \mathbb{N} , la relación caracterizada por $xSy \iff x - y$ es par, es una relación de equivalencia.

3.2.6. Relaciones de orden

Definición 3.2.7. Las relaciones homogéneas que son a la vez reflexivas, antisimétricas y transitivas se denominan *relaciones de orden*.

La función (`esRelacionOrden xs r`) se verifica si r es una relación de orden en xs .

```
-- | Ejemplo
-- >>> esRelacionOrden [1..3] [(1,1),(1,2),(1,3),(2,2),(2,3),(3,3)]
-- True
esRelacionOrden :: Eq a => [a] -> [(a,a)] -> Bool
esRelacionOrden xs r =
  esReflexiva xs r    &&
  esAntisimetrica r   &&
  esTransitiva r
```

Nota 3.2.7. En el conjunto \mathbb{N} , las relaciones caracterizadas por:

- $xRy \iff x \leq y$,
- $xTy \iff x$ divide a y ,

son relaciones de orden.

3.2.7. Clases de equivalencia

Definición 3.2.8. Si R es una relación de equivalencia en A , denominamos *clase de equivalencia* de un elemento $x \in A$ al conjunto de todos los elementos de A relacionados con x ; es decir, $\bar{x} = R(x) = \{y \in A \mid xRy\}$ donde la primera notación se usa si la relación con la que se está tratando se sobreentiende, y la segunda si no es así.

La función (`clasesEquivalencia xs r`) devuelve las clases de la relación de equivalencia r en xs .

```
-- | Ejemplo
-- >>> let r = [(x,y) | x <- [1..5], y <- [1..5], even (x-y)]
-- >>> clasesEquivalencia [1..5] r
-- [[1,3,5],[2,4]]
clasesEquivalencia :: Eq a => [a] -> [(a,a)] -> [[a]]
```

```

clasesEquivalencia _ [] = []
clasesEquivalencia [] _ = []
clasesEquivalencia (x:xs) r = (x:c) : clasesEquivalencia (xs \\ c) r
  where c = filter (estaRelacionado r x) xs

```

3.3. Funciones

Definición 3.3.1. Dada una relación F entre A y B , se dirá que es una *función* si es una relación binaria, es funcional y todos los elementos de A están en el dominio.

La función (`esFuncion xs ys f`) se verifica si f es una función de xs en ys .

```

-- | Ejemplos
-- >>> esFuncion [3,1] [2,4,7] [(1,7),(3,2)]
-- True
-- >>> esFuncion [3,1] [2,4,7] [(1,7)]
-- False
-- >>> esFuncion [3,1] [2,4,7] [(1,7),(3,2),(1,4)]
-- False
esFuncion :: (Eq a, Eq b) => [a] -> [b] -> [(a,b)] -> Bool
esFuncion xs ys f =
  esRelacion xs ys f &&
  xs `esSubconjunto` dominio f &&
  esFuncional f

```

Nota 3.3.1. A lo largo de la sección representaremos a las funciones como listas de pares.

```

type Funcion a b = [(a,b)]

```

La función (`funciones xs ys`) devuelve todas las posibles funciones del conjunto xs en ys .

```

-- | Ejemplos
-- >>> pp $ funciones [1,2] [3,4]
-- [(1, 3),(2, 3)],[(1, 3),(2, 4)],[(1, 4),(2, 3)],
-- [(1, 4),(2, 4)]
-- >>> pp $ funciones [1,2] [3,4,5]
-- [(1, 3),(2, 3)],[(1, 3),(2, 4)],[(1, 3),(2, 5)],
-- [(1, 4),(2, 3)],[(1, 4),(2, 4)],[(1, 4),(2, 5)],
-- [(1, 5),(2, 3)],[(1, 5),(2, 4)],[(1, 5),(2, 5)]
-- >>> pp $ funciones [0,1,2] [3,4]
-- [(0, 3),(1, 3),(2, 3)],[(0, 3),(1, 3),(2, 4)],

```

```
-- [(0, 3),(1, 4),(2, 3)],[(0, 3),(1, 4),(2, 4)],
-- [(0, 4),(1, 3),(2, 3)],[(0, 4),(1, 3),(2, 4)],
-- [(0, 4),(1, 4),(2, 3)],[(0, 4),(1, 4),(2, 4)]]
funciones :: [a] -> [b] -> [Funcion a b]
funciones xs ys =
  [zip xs zs | zs <- variacionesR (length xs) ys]
```

3.3.1. Imagen por una función

Definición 3.3.2. Si f es una función entre A y B y x es un elemento del conjunto A , la *imagen del elemento x por la función f* es el valor asociado a x por la función f .

La función `(imagen f x)` es la imagen del elemento x en la función f .

```
-- | Ejemplos
-- >>> imagen [(1,7),(3,2)] 1
-- 7
-- >>> imagen [(1,7),(3,2)] 3
-- 2
imagen :: (Eq a, Eq b) => Funcion a b -> a -> b
imagen f x = head (imagenRelacion f x)
```

La función `(imagenConjunto f xs)` es la imagen del conjunto xs en la función f .

```
-- | Ejemplos
-- >>> imagenConjunto [(1,7),(3,2),(4,3)] [1,4]
-- [7,3]
-- >>> imagenConjunto [(1,7),(3,2)] [3,1]
-- [2,7]
imagenConjunto :: (Eq a, Eq b) => Funcion a b -> [a] -> [b]
imagenConjunto f xs = nub (map (imagen f) xs)
```

3.3.2. Funciones inyectivas

Definición 3.3.3. Diremos que una función f entre dos conjuntos es *inyectiva*² si a elementos distintos del dominio le corresponden elementos distintos de la imagen; es decir, si $\forall a, b \in \text{dominio}(f)$ tales que $a \neq b$, $f(a) \neq f(b)$.

La función `(esInyectiva fs)` se verifica si la función fs es inyectiva.

²https://en.wikipedia.org/wiki/Injective_function

```
-- | Ejemplos
-- >>> esInyectiva [(1,4),(2,5),(3,6)]
-- True
-- >>> esInyectiva [(1,4),(2,5),(3,4)]
-- False
-- >>> esInyectiva [(1,4),(2,5),(3,6),(3,6)]
-- True
esInyectiva :: (Eq a, Eq b) => Funcion a b -> Bool
esInyectiva f =
  all esUnitario [antiImagenRelacion f y | y <- rango f]
```

3.3.3. Funciones sobreyectivas

Definición 3.3.4. Diremos que una función f entre dos conjuntos A y B es **sobreyectiva**³ si todos los elementos de B son imagen de algún elemento de A .

La función (`esSobreyectiva xs ys f`) se verifica si la función f es sobreyectiva. A la hora de definirla, estamos contando con que f es una función entre xs y ys .

```
-- | Ejemplos
-- >>> esSobreyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,6)]
-- True
-- >>> esSobreyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,4)]
-- False
-- >>> esSobreyectiva [1,2,3] [4,5,6] [(1,4),(2,4),(3,6),(3,6)]
-- False
esSobreyectiva :: (Eq a, Eq b) => [a] -> [b] -> Funcion a b -> Bool
esSobreyectiva _ ys f = ys `esSubconjunto` rango f
```

3.3.4. Funciones biyectivas

Definición 3.3.5. Diremos que una función f entre dos conjuntos A y B es **biyectiva**⁴ si cada elementos de B es imagen de un único elemento de A .

La función (`esBiyectiva xs ys f`) se verifica si la función f es biyectiva.

```
-- | Ejemplos
-- >>> esBiyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,6),(3,6)]
-- True
-- >>> esBiyectiva [1,2,3] [4,5,6] [(1,4),(2,5),(3,4)]
```

³https://en.wikipedia.org/wiki/Surjective_function

⁴https://en.wikipedia.org/wiki/Bijjective_function

```
-- False
-- >>> esBiyectiva [1,2,3] [4,5,6,7] [(1,4),(2,5),(3,6)]
-- False
esBiyectiva :: (Eq a, Eq b) => [a] -> [b] -> Funcion a b -> Bool
esBiyectiva xs ys f =
  esInyectiva f && esSobreyectiva xs ys f
```

La funciones `biyecciones1 xs ys` y `biyecciones2 xs ys` devuelven la lista de todas las biyecciones entre los conjuntos `xs` y `ys`. La primera lo hace filtrando las funciones entre los conjuntos que son biyectivas y la segunda lo hace construyendo únicamente las funciones biyectivas entre los conjuntos, con el consecuente ahorro computacional.

```
ghci> length (biyecciones1 [1..7] ['a'..'g'])
5040
(16.75 secs, 4,146,744,104 bytes)
ghci> length (biyecciones2 [1..7] ['a'..'g'])
5040
(0.02 secs, 0 bytes)
ghci> length (biyecciones1 [1..6] ['a'..'g'])
0
(2.53 secs, 592,625,824 bytes)
ghci> length (biyecciones2 [1..6] ['a'..'g'])
0
(0.01 secs, 0 bytes)
```

```
biyecciones1 :: (Eq a, Eq b) => [a] -> [b] -> [Funcion a b]
biyecciones1 xs ys =
  filter (esBiyectiva xs ys) (funciones xs ys)

biyecciones2 :: (Eq a, Eq b) => [a] -> [b] -> [Funcion a b]
biyecciones2 xs ys
  | length xs /= length ys = []
  | otherwise               = [zip xs zs | zs <- permutations ys]
```

Nota 3.3.2. En lo que sigue trabajaremos con la función `biyecciones2` así que la definiremos como `biyecciones`.

```
biyecciones :: (Eq a, Eq b) => [a] -> [b] -> [Funcion a b]
biyecciones = biyecciones2
```

3.3.5. Inversa de una función

Definición 3.3.6. Si f es una función biyectiva entre los conjuntos A y B , definimos la *función inversa*⁵ como la función que a cada elemento de B le hace corresponder el elemento de

⁵https://en.wikipedia.org/wiki/Inverse_function

A del que es imagen en B.

El valor de `(inversa f)` es la función inversa de `f`.

```
-- | Ejemplos
-- >>> inversa [(1,4),(2,5),(3,6)]
-- [(4,1),(5,2),(6,3)]
-- >>> inversa [(1,'f'),(2,'m'),(3,'a')]
-- [('f',1),('m',2),('a',3)]
inversa :: (Eq a, Eq b) => Funcion a b -> Funcion b a
inversa f = [(y,x) | (x,y) <- f]
```

Nota 3.3.3. Para considerar la inversa de una función, esta tiene que ser biyectiva. Luego `(inversa f)` asigna a cada elemento del conjunto imagen (que en este caso coincide con la imagen) uno y solo uno del conjunto de salida.

La función `(imagenInversa f y)` devuelve el elemento del conjunto de salida de la función `f` tal que su imagen es `y`.

```
-- | Ejemplos
-- >>> imagenInversa [(1,4),(2,5),(3,6)] 5
-- 2
-- >>> imagenInversa [(1,'f'),(2,'m'),(3,'a')] 'a'
-- 3
imagenInversa :: (Eq a, Eq b) => Funcion a b -> b -> a
imagenInversa f = imagen (inversa f)
```


Capítulo 4

Introducción a la teoría de grafos

Se dice que la Teoría de Grafos tiene su origen en 1736, cuando Euler dio una solución al problema (hasta entonces no resuelto) de los siete puentes de Königsberg: ¿existe un camino que atravesase cada uno de los puentes exactamente una vez?

Para probar que no era posible, Euler sustituyó cada región por un nodo y cada puente por una arista, creando el primer grafo que fuera modelo de un problema matemático.




Figura 4.1: Dibujo de los puentes de Königsberg Figura 4.2: Modelo de los puentes de Königsberg

Desde entonces, se ha ido desarrollando esta metodología hasta convertirse en los últimos años en una herramienta importante en áreas del conocimiento muy variadas como, por ejemplo: la Investigación Operativa, la Computación, la Ingeniería Eléctrica, la Geografía y la Química. Es por ello que, además, se ha erigido como una nueva disciplina matemática, que generalmente asociada a las ramas de Topología y Álgebra.

La utilidad de los grafos se basa en su gran poder de abstracción y una representación muy clara de cualquier relación, lo que facilita enormemente tanto la fase de modelado como la de resolución de cualquier problema. Gracias a la Teoría de Grafos se han desarrollado una gran variedad de algoritmos y métodos de decisión que podemos implementar a través de lenguajes funcionales y permiten automatizar la resolución de muchos problemas, a menudo tediosos de resolver a mano.

Comentario 2: Pendiente de ampliar la introducción conforme se vaya escribiendo los módulos.

Contenido

3.1	Relaciones	29
3.1.1	Relación binaria	29
3.1.2	Imagen por una relación	29
3.1.3	Dominio de una relación	30
3.1.4	Rango de una relación	30
3.1.5	Antiimagen por una relación	31
3.1.6	Relación funcional	31
3.2	Relaciones homogéneas	31
3.2.1	Relaciones reflexivas	32
3.2.2	Relaciones simétricas	33
3.2.3	Relaciones antisimétricas	33
3.2.4	Relaciones transitivas	34
3.2.5	Relaciones de equivalencia	34
3.2.6	Relaciones de orden	35
3.2.7	Clases de equivalencia	35
3.3	Funciones	36
3.3.1	Imagen por una función	37
3.3.2	Funciones inyectivas	37
3.3.3	Funciones sobreyectivas	38
3.3.4	Funciones biyectivas	38
3.3.5	Inversa de una función	39

4.1. Definición de grafo

En primer lugar, vamos a introducir terminología básica en el desarrollo de la Teoría de Grafos.

Definición 4.1.1. Un **grafo** G es un par (V, A) , donde V es el conjunto cuyos elementos llamamos **vértices** (o **nodos**) y A es un conjunto cuyos elementos llamamos **aristas**.

Definición 4.1.2. Una **arista** de un grafo $G = (V, A)$, es un conjunto de dos elementos de V . Es decir, para dos vértices v, v' de G , (v, v') y (v', v) representa la misma arista.

Definición 4.1.3. Dado un grafo $G = (V, A)$, diremos que un vértice $v \in V$ es **adyacente** a $v' \in V$ si $(v', v) \in A$.

Definición 4.1.4. Si en un grafo dirigido se permiten aristas repetidas, lo llamaremos **multi-grafo**. Si no se permiten, lo llamaremos **grafo regular**.

Nota 4.1.1. Denotaremos por $|V|$ al número de vértices y por $|A|$ al número de aristas del grafo (V, A) .

Ejemplo 4.1.2. Sea $G = (V, A)$ un grafo con $V = \{a, b, c, d\}$ y $A = \{(a, b), (a, c), (b, d), (d, d)\}$. En este grafo, los vértices a, d son adyacentes a b .



4.2. El TAD de los grafos

En esta sección, nos planteamos la tarea de implementar las definiciones presentadas anteriormente en un lenguaje funcional. En nuestro caso, el lenguaje que utilizaremos será Haskell. Definiremos el Tipo Abstracto de Dato (TAD) de los grafos y daremos algunos ejemplos de posibles representaciones de grafos con las que podremos trabajar.

Si consideramos un grafo finito cualquiera $G = (V, A)$, podemos ordenar el conjunto de los vértices y representarlo como $V = \{v_1, \dots, v_n\}$ con $n = |V|$.

En primer lugar, necesitaremos crear un tipo (`Grafo`) cuya definición sea compatible con la entidad matemática que representa y que nos permita definir las operaciones que necesitamos para trabajar con los grafos. Estas operaciones son:

```
creaGrafo  -- [a] -> [(a,a)] -> Grafo a
vertices   -- Grafo a -> [a]
adyacentes -- Grafo a -> a -> [a]
aristaEn   -- (a,a) -> Grafo a -> Bool
aristas    -- Grafo a -> [(a,a)]
```

donde:


- `(creaGrafo vs as)` es un grafo tal que el conjunto de sus vértices es `vs` y el de sus aristas es `as`.

- `(vertices g)` es la lista de todos los vértices del grafo `g`.
- `(adyacentes g v)` es la lista de los vértices adyacentes al vértice `v` en el grafo `g`.
- `(aristaEn a g)` se verifica si `a` es una arista del grafo `g`.
- `(aristas g)` es la lista de las aristas del grafo `g`.

Nota 4.2.1. Las funciones que aparecen en la especificación del TAD no dependen de la representación que elijamos.

Ejemplo 4.2.2. Veamos un ejemplo de creación de grafo y su representación gráfica

```
creaGrafo [1..5] [(1,2),(1,3),(1,5),(2,4),
                  (2,5),(3,4),(3,5),(4,5)]
```



4.2.1. Grafos como listas de aristas

En el módulo `GrafoConListaDeAristas` se definen las funciones del TAD de los grafos dando su representación como conjuntos de aristas; es decir, representando a un grafo como dos conjuntos, la primera será la lista ordenada de los vértices y la segunda la lista ordenada de las aristas (en ambas listas se excluye la posibilidad de repeticiones).

Nota 4.2.3. Las ventajas de usar arrays frente a usar listas es que los array tienen acceso constante ($O(1)$) a sus elementos mientras que las listas tienen acceso lineal ($O(n)$) y que la actualización de un elemento en un array no supone espacio extra. Sin embargo, los arrays son representaciones muy rígidas: cualquier modificación en su estructura, como cambiar su tamaño, supone un gran coste computacional pues se tendría que crear de nuevo el array y, además, sus índices deben pertenecer a la clase de los objetos indexables (`Ix`), luego perdemos mucha flexibilidad en la representación.

```
{-# LANGUAGE DeriveGeneric #-}

module GrafoConListaDeAristas
  ( Grafo
  , creaGrafo -- [a] -> [(a,a)] -> Grafo a
  , vertices  -- Grafo a -> [a]
  , adyacentes -- Grafo a -> a -> [a]
```

```
, aristaEn    -- (a,a) -> Grafo a -> Bool
, aristas     -- Grafo a -> [(a,a)]
) where
```

En las definiciones del presente módulo se usarán las funciones `nub` y `sort` de la librería `Data.List`

Vamos a definir un nuevo tipo de dato (`Grafo a`), que representará un grafo a partir de la lista de sus vértices (donde los vértices son de tipo `a`) y de aristas (que son pares de vértices).

```
data Grafo a = G [a] [(a,a)]
    deriving (Eq, Show, Generic)

instance (Ord a) => Out (Grafo a)
```


Las funciones básicas que definiremos a partir de este tipo coincidirán con las indicadas en el TAD de los grafos.

- (`creaGrafo vs as`) es el grafo cuyo conjunto de vértices es `cs` y el de sus aristas es `as`.

```
-- | Ejemplo
-- >>> creaGrafo [1..5] [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
-- G [1,2,3,4,5] [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
creaGrafo :: Ord a => [a] -> [(a,a)] -> Grafo a
creaGrafo vs as =
    G (sort vs) (nub (sort [parOrdenado a | a <- as]))

parOrdenado :: Ord a => (a,a) -> (a,a)
parOrdenado (x,y) | x <= y    = (x,y)
                  | otherwise = (y,x)
```

Ejemplo 4.2.4. `ejGrafo` es el grafo



Los ejemplos usarán el siguiente grafo

```
ejGrafo :: Grafo Int
ejGrafo = creaGrafo [1..5]
                [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
```

- `(vertices g)` es la lista de los vértices del grafo `g`.

```
-- | Ejemplo
-- >>> vertices ejGrafo
-- [1,2,3,4,5]
vertices :: Grafo a -> [a]
vertices (G vs _) = vs
```

- `(adyacentes g v)` es la lista de los vértices adyacentes al vértice `v` en el grafo `g`.

```
-- | Ejemplos
-- >>> adyacentes ejGrafo 4
-- [1,3,5]
-- >>> adyacentes ejGrafo 3
-- [2,4,5]
adyacentes :: Eq a => Grafo a -> a -> [a]
adyacentes (G _ as) v =
  [u | (u,x) <- as, x == v] 'union'
  [u | (x,u) <- as, x == v]
```

- `(aristaEn a g)` se verifica si `a` es una arista del grafo `g`.

```
-- | Ejemplos
-- >>> (5,1) 'aristaEn' ejGrafo
-- True
-- >>> (3,1) 'aristaEn' ejGrafo
-- False
aristaEn :: Ord a => (a,a) -> Grafo a -> Bool
aristaEn a (G _ as) = parOrdenado a 'elem' as
```

- `(aristas g)` es la lista de las aristas del grafo `g`.

```
-- | Ejemplo
-- >>> aristas ejGrafo
-- [(1,2),(1,4),(1,5),(2,3),(2,5),(3,4),(3,5),(4,5)]
aristas :: Grafo a -> [(a,a)]
aristas (G _ as) = as
```

4.3. Generadores de grafos

En esta sección, presentaremos el generador de grafos que nos permitirá generar grafos como listas de aristas arbitrariamente y usarlos como ejemplos o para comprobar propiedades.

Para aprender a controlar el tamaño de los grafos generados, he consultado las siguientes fuentes:

* [QuickCheck: A Lightweight Tool for Random Testing of Haskell Programs](#)¹ ([2])

* [Property Testing using QuickCheck](#)² ([7])

(generaGrafos n) es un generador de grafos de hasta n vértices. Por ejemplo,

```
ghci> sample (generaGrafo 5)
G [1,2] []
G [1] [(1,1)]
G [] []
G [1,2,3,4] [(2,2)]
G [1,2,3] [(1,1),(1,2),(1,3),(2,2)]
G [1,2,3,4,5] [(1,2),(1,4),(1,5),(2,2),(2,3),(2,4),(3,3),(3,4),(4,4),(4,5)]
G [1] []
G [1,2,3] [(1,2),(2,2),(3,3)]
G [1,2,3,4] [(1,1),(1,4),(2,3),(2,4),(3,3),(3,4),(4,4)]
G [1,2] []
G [1,2,3] [(1,1),(1,2),(2,2),(3,3)]

ghci> sample (generaGrafo 2)
G [1,2] [(1,2),(2,2)]
G [1,2] [(1,1)]
G [1,2] [(1,1),(1,2)]
G [] []
G [1] [(1,1)]
G [1] []
G [1] []
G [] []
G [] []
G [] []
G [1] [(1,1)]
```

```
generaGrafo :: Int -> Gen (Grafo Int)
generaGrafo s = do
  let m = s `mod` 11
  n <- choose (0,m)
  as <- sublistOf [(x,y) | x <- [1..n], y <- [x..n]]
  return (creaGrafo [1..n] as)
```

Nota 4.3.1. Los grafos están contenidos en la clase de los objetos generables aleatoriamente.

```
instance Arbitrary (Grafo Int) where
  arbitrary = sized generaGrafo
```

¹<https://www.eecs.northwestern.edu/~robby/courses/395-495-2009-fall/quick.pdf>

²<https://www.dcc.fc.up.pt/~pbv/aulas/tapf/slides/quickcheck.html>

En el siguiente ejemplo se pueden observar algunos grafos generados

```
ghci> sample (arbitrary :: Gen (Grafo Int))
G [] []
G [1] [(1,1)]
G [1,2] [(1,1),(1,2)]
G [1] [(1,1)]
G [1,2,3,4,5,6,7,8] [(1,4),(1,7),(2,2),(2,3),(2,5),(2,8),(3,5),
    (3,6),(3,8),(4,4),(4,6),(4,7),(4,8),(5,6),(6,6),(6,7),(7,8)]
G [1,2,3,4,5] [(1,1),(1,2),(1,4),(1,5),(2,4),(2,5),(3,4),(5,5)]
G [] []
G [] []
G [1,2,3] [(1,1),(2,2),(2,3)]
G [1,2,3,4] [(1,1),(1,2),(1,4),(2,2),(3,3)]
G [1,2,3,4,5,6,7] [(1,1),(1,5),(1,6),(1,7),(2,2),(2,4),(2,5),
    (2,6),(2,7),(3,3),(3,5),(4,6),(4,7),(5,5),(5,7),(6,6),(7,7)]
```

4.4. Ejemplos de grafos

El objetivo de esta sección es reunir una colección de grafos lo suficientemente extensa y variada como para poder utilizarla como recurso a la hora de comprobar las propiedades y definiciones de funciones que implementaremos más adelante.

En el proceso de recopilación de ejemplos, se ha trabajado con diversas fuentes:

- los apuntes de la asignatura “Matemática discreta” ([3]),
- los temas de la asignatura “Informática” ([1]) y
- el artículo “Graph theory” ([9]) de la Wikipedia.

Nota 4.4.1. Se utilizará la representación de los grafos como listas de aristas.

4.4.1. Grafo nulo

Definición 4.4.1. *Un grafo nulo es un grafo que no tiene ni vértices ni aristas.*

La función (`grafoNulo`) devuelve un grafo nulo.

```
grafoNulo :: Ord a => Grafo a
grafoNulo = creaGrafo [] []
```

La función (`esGrafoNulo g`) se verifica si `g` es un grafo nulo.

```
-- | Ejemplos
-- >>> esGrafoNulo grafoNulo
-- True
-- >>> esGrafoNulo (creaGrafo [] [(1,2)])
```




```
-- False
-- >>> esGrafoNulo (creaGrafo [1,2] [(1,2)])
-- False
esGrafoNulo :: Grafo a -> Bool
esGrafoNulo g =
    null (vertices g) && null (aristas g)
```

4.4.2. Grafo ciclo

Definición 4.4.2. Un **ciclo**,³ de orden n , $C(n)$, es un grafo no dirigido y no ponderado cuyo conjunto de vértices viene dado por $V = \{1, \dots, n\}$ y el de las aristas por $A = \{(0,1), (1,2), \dots, (n-2, n-1), (n-1, 0)\}$

La función (`grafoCiclo n`) nos genera el ciclo de orden n .



```
-- | Ejemplos
-- >>> grafoCiclo 5
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
grafoCiclo :: Int -> Grafo Int
grafoCiclo 0 = grafoNulo
grafoCiclo 1 = creaGrafo [1] []
grafoCiclo n = creaGrafo [1..n]
                ([ (u,u+1) | u <- [1..n-1] ] ++ [(n,1)])
```


4.4.3. Grafo de la amistad

Definición 4.4.3. Un **grafo de la amistad**,⁴ de orden n es un grafo con $2n + 1$ vértices y $3n$ aristas formado uniendo n copias del ciclo C_3 por un vértice común. Lo denotamos por F_n .

La función (`grafoAmistad n`) genera el grafo de la amistad de orden n . Por ejemplo,

³https://es.wikipedia.org/wiki/Grafo_completo

⁴https://es.wikipedia.org/wiki/Grafo_de_la_amistad




```
-- | Ejemplos
-- >>> pp $ grafoAmistad 2
-- G [1,2,3,4,5]
-- [(1, 2),(1, 3),(1, 4),(1, 5),(2, 3),(4, 5)]
-- >>> pp $ grafoAmistad 3
-- G [1,2,3,4,5,6,7]
-- [(1, 2),(1, 3),(1, 4),(1, 5),(1, 6),(1, 7),(2, 3),
--    (4, 5),(6, 7)]
grafoAmistad :: Int -> Grafo Int
grafoAmistad n =
  creaGrafo [1..2*n+1]
    [(1,a) | a <- [2..2*n+1]] ++
    [(a,b) | (a,b) <- zip [2,4..2*n] [3,5..2*n+1]]
```

4.4.4. Grafo completo

Definición 4.4.4. El **grafo completo**,⁵ de orden n , $K(n)$, es un grafo no dirigido cuyo conjunto de vértices viene dado por $V = \{1, \dots, n\}$ y tiene una arista entre cada par de vértices distintos.

La función `(completo n)` nos genera el grafo completo de orden n . Por ejemplo,

⁵https://es.wikipedia.org/wiki/Grafo_completo




```
-- | Ejemplo
-- >>> completo 4
-- G [1,2,3,4] [(1,2),(1,3),(1,4),(2,3),(2,4),(3,4)]
completo :: Int -> Grafo Int
completo n =
  creaGrafo [1..n]
    [(a,b) | a <- [1..n], b <- [1..a-1]]
```

4.4.5. Grafo bipartito

Definición 4.4.5. Un **grafo bipartito**⁶ es un grafo $G = (V, A)$ verificando que el conjunto de sus vértices se puede dividir en dos subconjuntos disjuntos V_1, V_2 tales que $V_1 \cup V_2 = V$ de manera que $\forall u_1, u_2 \in V_1 [(u_1, u_2) \notin A]$ y $\forall v_1, v_2 \in V_2 [(v_1, v_2) \notin A]$.

Un **grafo bipartito completo**⁷ será entonces un grafo bipartito $G = (V_1 \cup V_2, A)$ en el que todos los vértices de una partición están conectados a los de la otra. Si $n = |V_1|, m = |V_2|$ denotamos al grafo bipartito $G = (V_1 \cup V_2, A)$ por $K_{n,m}$.

La función (bipartitoCompleto n m) nos genera el grafo bipartito $K_{n,m}$. Por ejemplo,



```
-- | Ejemplo
-- >>> bipartitoCompleto 2 3
-- G [1,2,3,4,5] [(1,3),(1,4),(1,5),(2,3),(2,4),(2,5)]
bipartitoCompleto :: Int -> Int -> Grafo Int
bipartitoCompleto n m =
  creaGrafo [1..n+m]
    [(a,b) | a <- [1..n], b <- [n+1..n+m]]
```


⁶https://es.wikipedia.org/wiki/Grafo_bipartito

⁷https://es.wikipedia.org/wiki/Grafo_bipartito_completo

4.4.6. Grafo estrella

Definición 4.4.6. Una **estrella**⁸ de orden n es el grafo bipartito completo $K_{1,n}$. Denotaremos a una estrella de orden n por S_n . Una estrella con 3 aristas se conoce en inglés como **claw** (garra o garfio).

La función (`grafoEstrella n`) crea un grafo circulante a partir de su orden n .




```
-- | Ejemplo
-- >>> grafoEstrella 5
-- G [1,2,3,4,5,6] [(1,2),(1,3),(1,4),(1,5),(1,6)]
grafoEstrella :: Int -> Grafo Int
grafoEstrella = bipartitoCompleto 1
```

4.4.7. Grafo rueda

Definición 4.4.7. Un **grafo rueda**⁹ de orden n es un grafo no dirigido y no ponderado con n vértices que se forma conectando un único vértice a todos los vértices de un ciclo C_{n-1} . Lo denotaremos por W_n .

La función (`grafoRueda n`) crea un grafo rueda a partir de su orden n .



```
-- | Ejemplo
-- >>> pp $ grafoRueda 6
-- G [1,2,3,4,5,6]
```

⁸[https://en.wikipedia.org/wiki/Star_\(graph_theory\)\)](https://en.wikipedia.org/wiki/Star_(graph_theory)))


⁹https://es.wikipedia.org/wiki/Grafo_rueda

```
-- [(1, 2),(1, 3),(1, 4),(1, 5),(1, 6),(2, 3),(2, 6),
--   (3, 4),(4, 5),(5, 6)]
grafoRueda :: Int -> Grafo Int
grafoRueda n =
  creaGrafo [1..n]
    [(1,a) | a <- [2..n]] ++
    [(a,b) | (a,b) <- zip (2:[2..n-1]) (3:n:[4..n])])
```

4.4.8. Grafo circulante

Definición 4.4.8. Un **grafo circulante**¹⁰ de orden $n \geq 3$ y saltos $\{s_1, \dots, s_k\}$ es un grafo no dirigido y no ponderado $G = (\{1, \dots, n\}, A)$ en el que cada nodo $\forall i \in V$ es adyacente a los $2k$ nodos $i \pm s_1, \dots, i \pm s_k \pmod n$. Lo denotaremos por $\text{Cir}_n^{s_1, \dots, s_k}$.

La función `(grafoCirculante n ss)` crea un grafo circulante a partir de su orden n y de la lista de sus saltos ss . Por ejemplo,




```
-- | Ejemplo
-- >>> pp $ grafoCirculante 6 [1,2]
-- G [1,2,3,4,5,6]
-- [(1, 2),(1, 3),(1, 5),(1, 6),(2, 3),(2, 4),(2, 6),
--   (3, 4),(3, 5),(4, 5),(4, 6),(5, 6)]
grafoCirculante :: Int -> [Int] -> Grafo Int
grafoCirculante n ss =
  creaGrafo [1..n]
    [(a,b) | a <- [1..n]
              , b <- sort (auxCir a ss n)
              , a < b]
  where auxCir v ss1 k =
    concat [[fun (v+s) k, fun (v-s) k] | s <- ss1]
    fun a b = if mod a b == 0 then b else mod a b
```

¹⁰https://en.wikipedia.org/wiki/Circulant_graph

4.4.9. Grafo de Petersen generalizado

El **grafo de Petersen generalizado**¹¹ que denotaremos $GP_{n,k}$ (con $n \geq 3$ y $1 \leq k \leq (n-1)/2$) es un grafo formado por un grafo circulante $Cir_{\{k\}}^n$ en el interior, rodeado por un ciclo C_n al que está conectado por una arista saliendo de cada vértice, de forma que se creen n polígonos regulares. El grafo $GP_{n,k}$ tiene $2n$ vértices y $3n$ aristas.

La función `(grafoPetersenGen n k)` devuelve el grafo de Petersen generalizado $GP_{n,k}$.




```
-- | Ejemplo
-- >>> pp $ grafoPetersenGen 4 2
-- G [1,2,3,4,5,6,7,8]
-- [(1, 3),(1, 5),(2, 4),(2, 6),(3, 7),(4, 8),(5, 6),
--    (5, 8),(6, 7),(7, 8)]
grafoPetersenGen :: Int -> Int -> Grafo Int
grafoPetersenGen n k =
  creaGrafo [1..2*n]
    (filter p (aristas (grafoCirculante n [k])) ++
     [(x,x+n) | x <- [1..n]] ++
     (n+1,n+2) : (n+1,2*n) : [(x,x+1) | x <- [n+2..2*n-1]])
  where p (a,b) = a < b
```

4.4.10. Otros grafos importantes

Grafo de Thomson

Definición 4.4.9. El grafo bipartito completo $K_{3,3}$ es conocido como el **grafo de Thomson** y, como veremos más adelante, será clave a la hora de analizar propiedades topológicas de los grafos.

¹¹https://en.wikipedia.org/wiki/Generalized_Petersen_graph




La función (`grafoThomson`) genera el grafo de Thomson.

```
-- | Ejemplo
-- >>> pp $ grafoThomson
-- G [1,2,3,4,5,6]
-- [(1, 4),(1, 5),(1, 6),(2, 4),(2, 5),(2, 6),(3, 4),
--    (3, 5),(3, 6)]
grafoThomson :: Grafo Int
grafoThomson = bipartitoCompleto 3 3
```

Grafo de Heawood

El **grafo de Heawood**¹² es un grafo no dirigido, regular con 14 vértices y 21 aristas. Todos sus vértices son incidentes a exactamente 3 aristas; es decir, es un grafo **cúbico**. Tiene importantes propiedades geométricas y topológicas.



La función `grafoHeawood` genera el grafo de Heawood.


```
-- | Ejemplo
-- >>> pp $ grafoHeawood
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14]
```

¹²https://en.wikipedia.org/wiki/Heawood_graph

```
-- [(1, 2),(1, 6),(1, 14),(2, 3),(2, 11),(3, 4),(3, 8),
--   (4, 5),(4, 13),(5, 6),(5, 10),(6, 7),(7, 8),(7, 12),
--   (8, 9),(9, 10),(9, 14),(10, 11),(11, 12),(12, 13),
--   (13, 14)]
grafoHeawood :: Grafo Int
grafoHeawood =
  creaGrafo [1..14]
    (aristas (grafoCiclo 14) ++
      zip [1, 2, 3, 4, 5, 7, 9]
          [6, 11, 8, 13, 10, 12, 14])
```

Grafo de McGee

El **grafo de McGee**¹³ es un grafo no dirigido, cúbico, con 24 vértices y 36 aristas. Tiene importantes propiedades geométricas y topológicas.



La función `grafoMcGee` genera el grafo de McGee.


```
-- | Ejemplo
-- >>> pp $ grafoMcGee
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,
--    20,21,22,23,24]
-- [(1, 2),(1, 13),(1, 24),(2, 3),(2, 9),(3, 4),(3, 20),
--   (4, 5),(4, 16),(5, 6),(5, 12),(6, 7),(6, 23),(7, 8),
--   (7, 19),(8, 9),(8, 15),(9, 10),(10, 11),(10, 22),
--   (11, 12),(11, 18),(12, 13),(13, 14),(14, 15),
```

¹³https://en.wikipedia.org/wiki/McGee_graph


```
--      (14, 21),(15, 16),(16, 17),(17, 18),(17, 24),
--      (18, 19),(19, 20),(20, 21),(21, 22),(22, 23),
--      (23, 24)]
grafoMcGee :: Grafo Int
grafoMcGee =
  creaGrafo [1..24]
    (aristas (grafoCiclo 24) ++
      zip [ 1,2, 3, 4, 5, 6, 7, 8,10,11,14,17]
          [13,9,20,16,12,23,19,15,22,18,21,24]))
```

Grafo Tutte–Coxeter

El **grafo Tutte–Coxeter**¹⁴ es un grafo no dirigido, cúbico, con 30 vértices y 45 aristas. Tiene importantes propiedades geométricas y topológicas.



La función `grafoTutteCoxeter` genera el grafo Tutte–Coxeter.


```
-- | Ejemplo
-- >>> pp $ grafoTutteCoxeter
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,
--     20,21,22,23,24,25,26,27,28,29,30]
-- [(1, 2),(1, 18),(1, 30),(2, 3),(2, 23),(3, 4),
```

¹⁴https://en.wikipedia.org/wiki/Tutte\T1\textendashCoxeter_graph

```
-- (3, 10),(4, 5),(4, 27),(5, 6),(5, 14),(6, 7),(6, 19),
-- (7, 8),(7, 24),(8, 9),(8, 29),(9, 10),(9, 16),
-- (10, 11),(11, 12),(11, 20),(12, 13),(12, 25),
-- (13, 14),(13, 30),(14, 15),(15, 16),(15, 22),
-- (16, 17),(17, 18),(17, 26),(18, 19),(19, 20),
-- (20, 21),(21, 22),(21, 28),(22, 23),(23, 24),
-- (24, 25),(25, 26),(26, 27),(27, 28),(28, 29),
-- (29, 30)]
grafoTutteCoxeter :: Grafo Int
grafoTutteCoxeter =
  creaGrafo [1..30]
    (aristas (grafoCiclo 30) ++
      zip [ 1, 2, 3, 4, 5, 6, 7, 8, 9,11,12,13,15,17,21]
          [18,23,10,27,14,19,24,29,16,20,25,30,22,26,28])
```

Grafo de Petersen

El **grafo de Petersen** ¹⁵ se define como el grafo de Petersen generalizado $GP_{5,2}$; es decir, es un grafo cúbico formado por los vértices de un pentágono, conectados a los vértices de una estrella de cinco puntas en la que cada nodo es adyacente a los nodos que están a un salto 2 de él. Es usado como ejemplo y como contraejemplo en muchos problemas de la Teoría de grafos.




La función `grafoPetersen` devuelve el grafo de Petersen.

```
-- | Ejemplo
-- >>> pp $ grafoPetersen
-- G [1,2,3,4,5,6,7,8,9,10]
-- [(1, 3),(1, 4),(1, 6),(2, 4),(2, 5),(2, 7),(3, 5),
-- (3, 8),(4, 9),(5, 10),(6, 7),(6, 10),(7, 8),(8, 9),
-- (9, 10)]
grafoPetersen :: Grafo Int
grafoPetersen = grafoPetersenGen 5 2
```

¹⁵https://en.wikipedia.org/wiki/Petersen_graph

Grafo de Moëbius–Cantor

El **grafo de Moëbius–Cantor**¹⁶ se define como el grafo de Petersen generalizado $GP_{8,3}$; es decir, es un grafo cúbico formado por los vértices de un octógono, conectados a los vértices de una estrella de ocho puntas en la que cada nodo es adyacente a los nodos que están a un salto 3 de él. Al igual que el grafo de Petersen, tiene importantes propiedades que lo hacen ser ejemplo y contraejemplo de muchos problemas de la Teoría de Grafos.



La función `grafoMoebiusCantor` genera el grafo de Moëbius–Cantor

```
-- | Ejemplo
-- >>> pp $ grafoMoebiusCantor
-- G [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16]
-- [(1, 4),(1, 6),(1, 9),(2, 5),(2, 7),(2, 10),(3, 6),
--   (3, 8),(3, 11),(4, 7),(4, 12),(5, 8),(5, 13),(6, 14),
--   (7, 15),(8, 16),(9, 10),(9, 16),(10, 11),(11, 12),
--   (12, 13),(13, 14),(14, 15),(15, 16)]
grafoMoebiusCantor :: Grafo Int
grafoMoebiusCantor = grafoPetersenGen 8 3
```

4.5. Definiciones y propiedades

Una vez construida una pequeña fuente de ejemplos, estamos en condiciones de implementar las definiciones sobre grafos en Haskell y ver que funcionan correctamente. Además, comprobaremos que se cumplen las propiedades básicas que se han presentado en el tema *Introducción a la teoría de grafos* de “Matemática discreta” ([3]).

Nota 4.5.1. Se utilizará el tipo abstracto de grafos presentados en la sección 4.2 y se utilizarán las librerías `Data.List` y `Test.QuickCheck`.

¹⁶https://en.wikipedia.org/wiki/Moëbius-Kantor_graph

4.5.1. Definiciones de grafos

Definición 4.5.1. El *orden* de un grafo $G = (V, A)$ se define como su número de vértices. Lo denotaremos por $|V(G)|$.

La función (`orden g`) devuelve el orden del grafo g .

```
-- | Ejemplos
-- >>> orden (grafoCiclo 4)
-- 4
-- >>> orden (grafoEstrella 4)
-- 5
orden :: Grafo a -> Int
orden = length . vertices
```

Definición 4.5.2. El *tamaño* de un grafo $G = (V, A)$ se define como su número de aristas. Lo denotaremos por $|A(G)|$.

La función (`tamaño g`) devuelve el orden del grafo g .

```
-- | Ejemplos
-- >>> tamaño (grafoCiclo 4)
-- 4
-- >>> tamaño grafoPetersen
-- 15
tamaño :: Grafo a -> Int
tamaño = length . aristas
```

Definición 4.5.3. Diremos que dos aristas a, a' son *incidentes* si tienen intersección no vacía; es decir, si tienen algún vértice en común.

La función (`sonIncidentes a a'`) se verifica si las aristas a y a' son incidentes.

```
-- | Ejemplos
-- >>> sonIncidentes (1,2) (2,4)
-- True
-- >>> sonIncidentes (1,2) (3,4)
-- False
sonIncidentes :: Eq a => (a,a) -> (a,a) -> Bool
sonIncidentes (u1,u2) (v1,v2) =
  or [u1 == v1, u1 == v2, u2 == v1, u2 == v2]
```

Definición 4.5.4. Diremos que una arista de un grafo G es un *lazo* si va de un vértice en sí mismo.

La función `(lazos g)` devuelve los lazos del grafo `g`.

```
-- | Ejemplos
-- >>> lazos (creaGrafo [1,2] [(1,1),(1,2),(2,2)])
-- [(1,1),(2,2)]
-- >>> lazos (grafoCiclo 5)
-- []
lazos :: Eq a => Grafo a -> [(a,a)]
lazos g = [(u,v) | (u,v) <- aristas g, u == v]
```

La función `(esLazo a)` se verifica si la arista `a` es un lazo.

```
-- | Ejemplos
-- >>> esLazo (4,4)
-- True
-- >>> esLazo (1,2)
-- False
esLazo :: Eq a => (a,a) -> Bool
esLazo (u,v) = u == v
```

Definición 4.5.5. Dado un grafo $G = (V, A)$, fijado un vértice $v \in V$, al conjunto de vértices que son adyacentes a v lo llamaremos **entorno** de v y lo denotaremos por $N(v) = \{u \in V \mid (u, v) \in A\}$.

La función `(entorno g v)` devuelve el entorno del vértice `v` en el grafo `g`.

```
-- | Ejemplo
-- >>> entorno (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 2
-- [1,3]
entorno :: Eq a => Grafo a -> a -> [a]
entorno = adyacentes
```

Definición 4.5.6. Sea $G = (V, A)$ un grafo. El **grado** (o **valencia**) de $v \in V$ es $\text{grad}(v) = |N(v)|$.

La función `(grado g v)` devuelve el grado del vértice `v` en el grafo `g`.

```
-- | Ejemplos
-- >>> grado (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 2
-- 2
-- >>> grado (grafoEstrella 5) 1
-- 5
grado :: Eq a => Grafo a -> a -> Int
grado g v = length (entorno g v)
```

Definición 4.5.7. Un vértice v de un grafo es *aislado* si su grado es 0.

La función (`esAislado g v`) se verifica si el vértice v es aislado en el grafo g .

```
-- | Ejemplos
-- >>> esAislado (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 4
-- True
-- >>> esAislado (creaGrafo [1..5] [(1,2),(1,3),(2,3)]) 3
-- False
esAislado :: Eq a => Grafo a -> a -> Bool
esAislado g v = grado g v == 0
```

Definición 4.5.8. Un grafo es *regular* si todos sus vértices tienen el mismo grado.

La función (`esRegular g`) se verifica si el grafo g es regular.

```
-- | Ejemplos
-- >>> esRegular (creaGrafo [1..3] [(1,2),(1,3),(2,3)])
-- True
-- >>> esRegular (creaGrafo [1..4] [(1,2),(1,3),(2,4)])
-- False
esRegular :: Eq a => Grafo a -> Bool
esRegular g = all (==x) xs
  where (x:xs) = [grado g v | v <- vertices g]
```

Definición 4.5.9. Dado un grafo $G = (V, A)$ llamamos *valencia mínima* o *grado mínimo* de G al valor $\delta(G) = \min\{\text{grad}(v) | v \in V\}$

La función (`valenciaMin g`) devuelve la valencia mínima del grafo g .

```
-- | Ejemplo
-- >>> valenciaMin (creaGrafo [1..4] [(1,2),(1,3),(2,4)])
-- 1
valenciaMin :: Ord a => Grafo a -> Int
valenciaMin g = minimum [grado g v | v <- vertices g]
```

Definición 4.5.10. Dado un grafo $G = (V, A)$ llamamos *valencia máxima* o *grado máximo* de G al valor $\delta(G) = \max\{\text{grad}(v) | v \in V\}$

La función (`valenciaMax g`) devuelve la valencia máxima del grafo g .

```
-- | Ejemplo
-- >>> valenciaMax (creaGrafo [1..4] [(1,2),(1,3),(2,4)])
-- 2
valenciaMax :: Ord a => Grafo a -> Int
valenciaMax g = maximum [grado g v | v <- vertices g]
```

Definición 4.5.11. Se dice que un grafo es simple si no contiene lazos ni aristas repetidas.

La función (`esSimple g`) se verifica si `g` es un grafo simple.

```
-- | Ejemplos
-- >>> esSimple (creaGrafo [1..3] [(1,2),(1,3),(2,3)])
-- True
-- >>> esSimple (creaGrafo [1..3] [(1,1),(1,2),(2,3)])
-- False
esSimple :: Ord a => Grafo a -> Bool
esSimple g =
  and [not ((x,x) 'aristaEn' g) | x <- vertices g]
```

Definición 4.5.12. Sea G un grafo. Llamamos **secuencia de grados** de G a la lista de grados de sus vértices. La secuencia se suele presentar en orden decreciente: $d_1 \geq d_2 \geq \dots \geq d_n$.

La función (`secuenciaGrados g`) devuelve la secuencia de los grados del grafo `g` en orden decreciente.

```
-- | Ejemplo
-- >>> secuenciaGrados (creaGrafo [1..5] [(1,2),(1,3),(1,4),(2,4)])
-- [3,2,2,1,0]
secuenciaGrados :: Eq a => Grafo a -> [Int]
secuenciaGrados g = sortBy (flip compare) [grado g v | v <- vertices g]
```

Nota 4.5.2. ¿Qué listas de n números enteros son secuencias de grafos de n vértices?

- Si $\sum_{i=1}^n d_i$ es impar, no hay ninguno.
- Si $\sum_{i=1}^n d_i$ es par, entonces siempre hay un grafo con esa secuencia de grados (aunque no necesariamente simple).

Definición 4.5.13. Una **secuencia gráfica** es una lista de número enteros no negativos que es la secuencia de grados para algún grafo simple.

La función (`secuenciaGrafica ss`) se verifica si existe algún grafo con la secuencia de grados `ss`.

```
-- | Ejemplos
-- >>> secuenciaGrafica [3,2,2,1,0]
-- True
-- >>> secuenciaGrafica [3,2,2,2]
-- False
secuenciaGrafica :: [Int] -> Bool
secuenciaGrafica ss = even (sum ss) && all p ss
  where p s = s >= 0 && s <= length ss
```

Definición 4.5.14. Dado un grafo $G = (V, A)$, diremos que $G' = (V', A')$ es un **subgrafo** de G si $V' \subseteq V$ y $A' \subseteq A$.

La función `(esSubgrafo g' g)` se verifica si g' es un subgrafo de g .

```
-- | Ejemplos
-- >>> esSubgrafo (bipartitoCompleto 3 2) (bipartitoCompleto 3 3)
-- True
-- >>> esSubgrafo (grafoEstrella 4) (grafoEstrella 5)
-- True
-- >>> esSubgrafo (completo 5) (completo 4)
-- False
-- >>> esSubgrafo (completo 3) (completo 4)
-- True
esSubgrafo :: Ord a => Grafo a -> Grafo a -> Bool
esSubgrafo g' g =
  vertices g' 'esSubconjunto' vertices g &&
  aristas g' 'esSubconjunto' aristas g
```

Definición 4.5.15. Si $G' = (V', A')$ es un subgrafo de $G = (V, A)$ tal que $V' = V$, diremos que G' es un **subgrafo maximal**, **grafo recubridor** o **grafo de expansión** (en inglés, *spanning graph*) de G .

La función `(esSubgrafoMax g' g)` se verifica si g' es un subgrafo maximal de g .

```
-- | Ejemplos
-- >>> esSubgrafoMax (grafoRueda 3) (grafoRueda 4)
-- False
-- >>> esSubgrafoMax (grafoCiclo 4) (grafoRueda 4)
-- True
-- >>> esSubgrafoMax (creaGrafo [1..3] [(1,2)]) (grafoCiclo 3)
-- True
-- >>> esSubgrafoMax (creaGrafo [1..2] [(1,2)]) (grafoCiclo 3)
-- False
esSubgrafoMax :: Ord a => Grafo a -> Grafo a -> Bool
esSubgrafoMax g' g =
  g' 'esSubgrafo' g && conjuntosIguales (vertices g') (vertices g)
```

Definición 4.5.16. Sean $G' = (V', A')$, $G = (V, A)$ dos grafos si $V' \subset V$, o $A' \subset A$, se dice que G' es un **subgrafo propio** de G , y se denota por $G' \subset G$.

La función `(esSubgrafoPropio g' g)` se verifica si g' es un subgrafo propio de g .

```
-- | Ejemplos
-- >>> esSubgrafoPropio (grafoRueda 3) (grafoRueda 4)
```



```
-- True
-- >>> esSubgrafoPropio (grafoRueda 4) (grafoCiclo 5)
-- False
-- >>> esSubgrafoPropio (creaGrafo [1..3] [(1,2)]) (grafoCiclo 3)
-- True
-- >>> esSubgrafoPropio (creaGrafo [1..2] [(1,2)]) (grafoCiclo 3)
-- True
esSubgrafoPropio :: Ord a => Grafo a -> Grafo a -> Bool
esSubgrafoPropio g' g =
  esSubgrafo g' g &&
  (not (conjuntosIguales (vertices g) (vertices g'))) ||
  not (conjuntosIguales (aristas g) (aristas g')))
```

4.5.2. Propiedades de grafos

Teorema 4.5.17 (Lema del apretón de manos). *En todo grafo simple el número de vértices de grado impar es par o cero.*

Vamos a comprobar que se verifica el lema del apretón de manos utilizando la función `prop_LemaApretonDeManos`.

```
ghci> quickCheck prop_LemaApretonDeManos
+++ OK, passed 100 tests.
```

```
prop_LemaApretonDeManos :: Grafo Int -> Bool
prop_LemaApretonDeManos g =
  even (length (filter odd [grado g v | v <- vertices g]))
  where g' = creaGrafo (vertices g)
                    [(x,y) | (x,y) <- aristas g, x /= y]
```

Teorema 4.5.18 (Havel–Hakimi). *Si $n > 1$ y $D = [d_1, \dots, d_n]$ es una lista de enteros, entonces D es secuencia gráfica si y sólo si la secuencia D' obtenida borrando el mayor elemento d_{\max} y restando 1 a los siguientes d_{\max} elementos más grandes es gráfica.*

Vamos a comprobar que se verifica el teorema de Havel–Hakimi utilizando la función `prop_HavelHakimi`.

```
ghci> quickCheck prop_HavelHakimi
+++ OK, passed 100 tests.
```

```
prop_HavelHakimi :: [Int] -> Bool
prop_HavelHakimi [] = True
prop_HavelHakimi (s:ss) =
  not (secuenciaGrafica (s:ss) && not (esVacio ss)) ||
  secuenciaGrafica (map (\x -> x-1) (take s ss) ++ drop s ss)
```

4.5.3. Operaciones y propiedades sobre grafos

Eliminación de una arista

Definición 4.5.19. Sea $G = (V, A)$ un grafo y sea $(u, v) \in A$. Definimos el grafo $G \setminus (u, v)$ como el subgrafo de G , $G' = (V', A')$, con $V' = V$ y $A' = A \setminus \{(u, v)\}$. Esta operación se denomina **eliminar una arista**.

La función `(eliminaArista g a)` elimina la arista a del grafo g .

```
-- | Ejemplos
-- >>> eliminaArista (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) (1,4)
-- G [1,2,3,4] [(1,2),(2,4)]
-- >>> eliminaArista (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) (4,1)
-- G [1,2,3,4] [(1,2),(2,4)]
-- >>> eliminaArista (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) (4,3)
-- G [1,2,3,4] [(1,2),(1,4),(2,4)]
eliminaArista :: Ord a => Grafo a -> (a,a) -> Grafo a
eliminaArista g (a,b) =
    creaGrafo (vertices g)
              (aristas g \\ [(a,b),(b,a)])
```

Eliminación un vértice

Definición 4.5.20. Sea $G = (V, A)$ un grafo y sea $v \in V$. Definimos el grafo $G \setminus v$ como el subgrafo de G , $G' = (V', A')$, con $V' = V \setminus \{v\}$ y $A' = A \setminus \{a \in A | v \text{ es un extremo de } a\}$. Esta operación se denomina **eliminar un vértice**.

La función `(eliminaVertice g v)` elimina el vértice v del grafo g .

```
-- | Ejemplos
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) 1
-- G [2,3,4] [(2,4)]
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) 4
-- G [1,2,3] [(1,2)]
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(2,4)]) 3
-- G [1,2,4] [(1,2),(1,4),(2,4)]
-- >>> eliminaVertice (creaGrafo [1..4] [(1,2),(1,4),(1,3)]) 1
-- G [2,3,4] []
eliminaVertice :: Ord a => Grafo a -> a -> Grafo a
eliminaVertice g v =
    creaGrafo (vertices g \\ [v])
              (as \\ [(a,b) | (a,b) <- as, a == v || b == v])
    where as = aristas g
```

Comentario 3: Añadir la especificación de `eliminaLazos`.

```
-- | Ejemplos
-- >>> eliminaLazos (creaGrafo [1,2] [(1,1),(1,2),(2,2)])
-- G [1,2] [(1,2)]
-- >>> grafoCiclo 5
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
-- >>> eliminaLazos (grafoCiclo 5)
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
eliminaLazos :: Ord a => Grafo a -> Grafo a
eliminaLazos g = creaGrafo (vertices g)
                        [(x,y) | (x,y) <- aristas g, x /= y]
```

Suma de aristas

Definición 4.5.21. Sea $G = (V, A)$ un grafo y sean $u, v \in V$ tales que $(u, v), (v, u) \notin A$. Definimos el grafo $G + (u, v)$ como el grafo $G' = (V, A \cup \{(u, v)\})$. Esta operación se denomina **suma de una arista**.

La función (`sumaArista g a`) suma la arista a al grafo g .

```
-- | Ejemplos
-- >>> grafoCiclo 5
-- G [1,2,3,4,5] [(1,2),(1,5),(2,3),(3,4),(4,5)]
-- >>> sumaArista (grafoCiclo 5) (1,3)
-- G [1,2,3,4,5] [(1,2),(1,3),(1,5),(2,3),(3,4),(4,5)]
sumaArista :: Ord a => Grafo a -> (a,a) -> Grafo a
sumaArista g a =
  creaGrafo (vertices g) (a : aristas g)
```

Suma de vértices

Definición 4.5.22. Sea $G = (V, A)$ un grafo y sea $v \notin V$. Definimos el grafo $G + v$ como el grafo $G' = (V', A')$, donde $V' = V \cup \{v\}$, $A' = A \cup \{(u, v) | u \in V\}$. Esta operación se denomina **suma de un vértice**.

La función (`sumaVertice g a`) suma el vértice a al grafo g .

```
-- | Ejemplo
-- >>> grafoCiclo 3
-- G [1,2,3] [(1,2),(1,3),(2,3)]
-- >>> sumaVertice (grafoCiclo 3) 4
-- G [1,2,3,4] [(1,2),(1,3),(1,4),(2,3),(2,4),(3,4)]
sumaVertice :: Ord a => Grafo a -> a -> Grafo a
sumaVertice g v =
  creaGrafo (v : vs) (aristas g ++ [(u,v) | u <- vs])
  where vs = vertices g
```

Propiedad de los grafos completos

Proposición 4.5.23. *La familia de grafos completos K_n verifica que $K_n = K_{n-1} + n$.*

Vamos a ver que se cumple la propiedad utilizando la función `prop_completos`.

```
ghci> quickCheck prop_completos
+++ OK, passed 100 tests.
```

```
prop_completos :: Int -> Property
prop_completos n = n >= 2 ==>
  completo n == sumaVertice (completo (n-1)) n
```

Comentario 4: A partir de esta propiedad, se puede dar una definición alternativa de K_n (`completo2`) y comprobar su equivalencia con la primera (`completo`).

Suma de grafos

Definición 4.5.24. Sean $G = (V, A), G' = (V', A)$ dos grafos. Definimos el grafo suma de G y G' como el grafo $G + G' = (V \cup V', A \cup A' \cup \{(u, v) | u \in V, v \in V'\})$. Esta operación se denomina **suma de grafos**.

La función (`sumaGrafos g g'`) suma los grafos g y g' .

```
-- | Ejemplo
-- >>> let g1 = creaGrafo [1..3] [(1,1),(1,3),(2,3)]
-- >>> let g2 = creaGrafo [4..6] [(4,6),(5,6)]
-- >>> pp $ sumaGrafos g1 g2
-- G [1,2,3,4,5,6]
-- [(1, 1),(1, 3),(1, 4),(1, 5),(1, 6),(2, 3),(2, 4),
-- (2, 5),(2, 6),(3, 4),(3, 5),(3, 6),(4, 6),(5, 6)]
sumaGrafos :: Ord a => Grafo a -> Grafo a -> Grafo a
sumaGrafos g1 g2 =
  creaGrafo (vs1 `union` vs2)
    (aristas g1 `union`
     aristas g2 `union`
     [(u,v) | u <- vs1, v <- vs2])
  where vs1 = vertices g1
        vs2 = vertices g2
```

Unión de grafos

Definición 4.5.25. Sean $G = (V, A)$, $G' = (V', A)$ dos grafos. Definimos el *grafo unión* de G y G' como el grafo $G \cup H = (V \cup V', A \cup A')$. Esta operación se denomina **unión de grafos**.

La función (`unionGrafos g g'`) une los grafos g y g' . Por ejemplo,

```
-- | Ejemplo
-- >>> let g1 = creaGrafo [1..3] [(1,1),(1,3),(2,3)]
-- >>> let g2 = creaGrafo [4..6] [(4,6),(5,6)]
-- >>> unionGrafos g1 g2
-- G [1,2,3,4,5,6] [(1,1),(1,3),(2,3),(4,6),(5,6)]
unionGrafos :: Ord a => Grafo a -> Grafo a -> Grafo a
unionGrafos g1 g2 =
    creaGrafo (vertices g1 'union' vertices g2)
              (aristas g1 'union' aristas g2)
```

Grafo complementario

Definición 4.5.26. Dado un grafo $G = (V, A)$ se define el *grafo complementario* de G como $\overline{G} = (V, \overline{A})$, donde $\overline{A} = \{(u, v) | u, v \in V, (u, v) \notin A\}$.

La función (`grafoComplementario g`) devuelve el grafo complementario de g .

```
-- | Ejemplo
-- >>> grafoComplementario (creaGrafo [1..3] [(1,1),(1,3),(2,3)])
-- G [1,2,3] [(1,2),(2,2),(3,3)]
grafoComplementario :: Ord a => Grafo a -> Grafo a
grafoComplementario g =
    creaGrafo vs
              [(u,v) | u <- vs, v <- vs, u <= v, not ((u,v) 'aristaEn' g)]
    where vs = vertices g
```

Definición 4.5.27. Dado un grafo, diremos que es **completo** si su complementario sin lazos no tiene aristas.


La función (`esCompleto g`) se verifica si el grafo g es completo.

```
-- Ejemplos
-- >>> esCompleto (completo 4)
-- True
-- >>> esCompleto (grafoCiclo 5)
-- False
esCompleto :: Ord a => Grafo a -> Bool
esCompleto g =
    tamaño (eliminaLazos (grafoComplementario g)) == 0
```

4.6. Morfismos de grafos

Llegados a este punto, es importante resaltar que un grafo se define como una entidad matemática abstracta; es evidente que lo importante de un grafo no son los nombres de sus vértices ni su representación gráfica. La propiedad que caracteriza a un grafo es la forma en que sus vértices están unidos por las aristas.

A priori, los siguientes grafos pueden parecer distintos:



Sin embargo, ambos grafos son grafos de cinco vértices que tienen las mismas relaciones de vecindad entre sus nodos. En esta sección estudiaremos estas relaciones que establecen las aristas entre los vértices de un grafo y presentaremos algoritmos que nos permitan identificar cuándo dos grafos se pueden relacionar mediante aplicaciones entre sus vértices cumpliendo ciertas características.

4.6.1. Morfismos

Definición 4.6.1. Si f es una función entre dos grafos $G = (V, A)$ y $G' = (V', A')$, diremos que **conserva la adyacencia** si $\forall u, v \in V$ se verifica que si $(u, v) \in A$, entonces $(f(u), f(v)) \in A'$.

La función $(\text{conservaAdyacencia } g \text{ h } f)$ se verifica si la función f entre los grafos g y h conserva las adyacencias.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1..4] [(1,2),(2,3),(3,4)]
-- >>> let g2 = creaGrafo [1..4] [(1,2),(2,3),(2,4)]
-- >>> let g3 = creaGrafo [4,6..10] [(4,8),(6,8),(8,10)]
-- >>> conservaAdyacencia g1 g3 [(1,4),(2,6),(3,8),(4,10)]
-- False
-- >>> conservaAdyacencia g2 g3 [(1,4),(2,8),(3,6),(4,10)]
-- True
conservaAdyacencia :: (Ord a, Ord b) =>
    Grafo a -> Grafo b -> Funcion a b -> Bool
conservaAdyacencia g h f =
    and [(imagen f x, imagen f y) `aristaEn` h | (x,y) <- aristas g]
```

Definición 4.6.2. Dados dos grafos simples $G = (V, A)$ y $G' = (V', A')$, un **morfismo** entre G y G' es una función $\phi : V \rightarrow V'$ que conserva las adyacencias.

La función `(esMorfismo g h vvs)` se verifica si la función representada por `vvs` es un morfismo entre los grafos `g` y `h`.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,2),(3,2)]
-- >>> let g2 = creaGrafo [4,5,6] [(4,6),(5,6)]
-- >>> esMorfismo g1 g2 [(1,4),(2,6),(3,5)]
-- True
-- >>> esMorfismo g1 g2 [(1,4),(2,5),(3,6)]
-- False
-- >>> esMorfismo g1 g2 [(1,4),(2,6),(3,5),(7,9)]
-- False
esMorfismo :: (Ord a,Ord b) => Grafo a -> Grafo b ->
              Funcion a b -> Bool
esMorfismo g1 g2 f =
  esFuncion (vertices g1) (vertices g2) f &&
  conservaAdyacencia g1 g2 f
```

La función `(morfismos g h)` devuelve todos los posibles morfismos entre los grafos `g` y `h`.

```
-- | Ejemplos
-- >>> grafoCiclo 3
-- G [1,2,3] [(1,2),(1,3),(2,3)]
-- >>> let g = creaGrafo [4,6] [(4,4),(6,6)]
-- >>> morfismos (grafoCiclo 3) g
-- [[(1,4),(2,4),(3,4)],[(1,6),(2,6),(3,6)]]
-- >>> morfismos g (grafoCiclo 3)
-- []
morfismos :: (Ord a, Ord b) => Grafo a -> Grafo b -> [(a,b)]
morfismos g h =
  [f | f <- funciones (vertices g) (vertices h)
    , conservaAdyacencia g h f]
```

Comentario 5: Introducir el problema de decidir si dos grafos son homomorfos.

4.6.2. Isomorfismos

Definición 4.6.3. *Dados dos grafos simples $G = (V, A)$ y $G' = (V', A')$, un **isomorfismo** entre G y G' es un morfismo biyectivo cuyo inverso es morfismo entre G' y G .*

La función `(esIsomorfismo g h f)` se verifica si la aplicación `f` es un isomorfismo entre los grafos `g` y `h`.

```

-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> esIsomorfismo g1 g2 [(1,2),(2,4),(3,6)]
-- True
-- >>> esIsomorfismo g1 g2 [(1,2),(2,2),(3,6)]
-- False
esIsomorfismo :: (Ord a,Ord b) =>
                Grafo a -> Grafo b -> Funcion a b -> Bool
esIsomorfismo g h f =
  esBiyectiva vs1 vs2 f      &&
  esMorfismo g h f          &&
  esMorfismo h g (inversa f)
  where vs1 = vertices g
        vs2 = vertices h

```

La función `(isomorfismos1 g h)` devuelve todos los isomorfismos posibles entre los grafos `g` y `h`.

```

-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfismos1 g1 g2
-- [(1,2),(2,4),(3,6)]
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfismos1 g3 g4
-- []
-- >>> let g5 = creaGrafo [1,2,3] [(1,1),(2,2),(3,3)]
-- >>> let g6 = creaGrafo [2,4,6] [(2,2),(4,4),(6,6)]
-- >>> pp $ isomorfismos1 g5 g6
-- [(1, 2),(2, 4),(3, 6)],[(1, 4),(2, 2),(3, 6)],
-- [(1, 6),(2, 4),(3, 2)],[(1, 4),(2, 6),(3, 2)],
-- [(1, 6),(2, 2),(3, 4)],[(1, 2),(2, 6),(3, 4)]
isomorfismos1 :: (Ord a,Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos1 g h =
  [f | f <- biyecciones vs1 vs2
      , conservaAdyacencia g h f
      , conservaAdyacencia h g (inversa f)]
  where vs1 = vertices g
        vs2 = vertices h

```

Definición 4.6.4. Dos grafos G y H se dicen *isomorfos* si existe algún isomorfismo entre ellos.

La función `isomorfos1 g h` se verifica si los grafos `g` y `h` son isomorfos.


```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfos1 g1 g2
-- True
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfos1 g3 g4
-- False
isomorfos1 :: (Ord a,Ord b) => Grafo a -> Grafo b -> Bool
isomorfos1 g = not . null . isomorfismos1 g
```

Nota 4.6.1. Al tener Haskell una evaluación perezosa, la función `(isomorfos g h)` no necesita generar todos los isomorfismos entre los grafos `g` y `h`.

Definición 4.6.5. Sea $G = (V, A)$ un grafo. Un *invariante por isomorfismos* de G es una propiedad de G que tiene el mismo valor para todos los grafos que son isomorfos a él.

```
esInvariantePorIsomorfismos ::
  Eq a => (Grafo Int -> a) -> Grafo Int -> Grafo Int -> Bool
esInvariantePorIsomorfismos p g h =
  isomorfos g h --> (p g == p h)
  where (-->) = (<=)
```

Teorema 4.6.6. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que $|V(G)| = |V(G')|$; es decir, el orden de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos orden)
+++ OK, passed 100 tests.
```

Teorema 4.6.7. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que $|A(G)| = |A(G')|$; es decir, el tamaño de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos tamaño)
+++ OK, passed 100 tests.
```

Teorema 4.6.8. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, tienen la misma secuencia de grados; es decir, la secuencia de grados de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos secuenciaGrados)
+++ OK, passed 100 tests.
```

A partir de las propiedades que hemos demostrado de los isomorfismos, vamos a dar otra definición equivalente de las funciones `(isomorfismos1 g h)` y `(isomorfos1 g h)`.

```
-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfismos2 g1 g2
-- [[(1,2),(2,4),(3,6)]]
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfismos2 g3 g4
-- []
-- >>> let g5 = creaGrafo [1,2,3] [(1,1),(2,2),(3,3)]
-- >>> let g6 = creaGrafo [2,4,6] [(2,2),(4,4),(6,6)]
-- >>> pp $ isomorfismos2 g5 g6
-- [[(1, 2),(2, 4),(3, 6)],[(1, 4),(2, 2),(3, 6)],
-- [(1, 6),(2, 4),(3, 2)],[(1, 4),(2, 6),(3, 2)],
-- [(1, 6),(2, 2),(3, 4)],[(1, 2),(2, 6),(3, 4)]]
isomorfismos2 :: (Ord a, Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos2 g h
  | orden g /= orden h = []
  | tamaño g /= tamaño h = []
  | secuenciaGrados g /= secuenciaGrados h = []
  | otherwise = [f | f <- biyecciones vs1 vs2
                      , conservaAdyacencia g h f]
  where vs1 = vertices g
        vs2 = vertices h

-- | Ejemplos
-- >>> let g1 = creaGrafo [1,2,3] [(1,1),(1,2),(2,3)]
-- >>> let g2 = creaGrafo [2,4,6] [(2,2),(2,4),(4,6)]
-- >>> isomorfos2 g1 g2
-- True
-- >>> let g3 = creaGrafo [1,2,3] [(1,1),(1,2),(1,3)]
-- >>> let g4 = creaGrafo [2,4,6] [(2,2),(4,4),(2,6)]
-- >>> isomorfos2 g3 g4
-- False
isomorfos2 :: (Ord a, Ord b) => Grafo a -> Grafo b -> Bool
```

```
isomorfos2 g =
  not. null . isomorfismos2 g
```

```
isomorfismos3 :: (Ord a, Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos3 g h
  | orden g /= orden h = []
  | tamaño g /= tamaño h = []
  | secuenciaGrados g /= secuenciaGrados h = []
  | otherwise = filter (conservaAdyacencia g h) (posibles g h)

verticesPorGrados g =
  [filter (p n) (vertices g) | n <- nub (secuenciaGrados g)]
  where p m v = grado g v == m

aux1 [] _ = []
aux1 (xs:xss) (ys:yss) = (map (zip xs) (permutations ys)):aux1 xss yss

aux2 [] = []
aux2 (xss:[]) = xss
aux2 (xss:yss:[]) = [xs ++ ys | xs <- xss, ys <- yss]
aux2 (xss:yss:xsss) =
  aux2 ([xs ++ ys | xs <- xss, ys <- yss]:xsss)

posibles g h =
  aux2 (aux1 (verticesPorGrados g) (verticesPorGrados h))
```

Vamos a comparar la eficiencia entre ambas definiciones

Nota 4.6.2. La nueva definición de (isomorfismos3 g h) es la más eficiente con grafos "poco regulares", sin embargo, cuando todos los vértices tienen el mismo grado, (isomorfismos2 g h) sigue siendo mejor opción (aunque no hay mucha diferencia en el coste computacional).

Comentario 6: Buscar definiciones más eficientes que isomorfismos3.

```
ghci> let n = 6 in (length (isomorfismos1 (completo n) (completo n)))
720
(0.18 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos2 (completo n) (completo n)))
720
(0.18 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos3 (completo n) (completo n)))
720
(0.18 secs, 26,123,800 bytes)
```

```

ghci> let n = 6 in (length (isomorfismos1 (grafoCiclo n) (grafoCiclo n)))
12
(0.04 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos2 (grafoCiclo n) (grafoCiclo n)))
12
(0.04 secs, 0 bytes)
ghci> let n = 6 in (length (isomorfismos3 (grafoCiclo n) (grafoCiclo n)))
12
(0.04 secs, 0 bytes)

ghci> length (isomorfismos1 (grafoCiclo 6) (completo 7))
0
(0.00 secs, 0 bytes)
ghci> length (isomorfismos2 (grafoCiclo 6) (completo 7))
0
(0.01 secs, 0 bytes)
ghci> length (isomorfismos3 (grafoCiclo 6) (completo 7))
0
(0.00 secs, 0 bytes)

ghci> isomorfos1 (completo 10) (grafoCiclo 10)
False
(51.90 secs, 12,841,861,176 bytes)
ghci> isomorfos2 (completo 10) (grafoCiclo 10)
False
(0.00 secs, 0 bytes)
ghci> isomorfos3 (completo 10) (grafoCiclo 10)
False
(0.00 secs, 0 bytes)

ghci> isomorfos1 (grafoCiclo 10) (grafoRueda 10)
False
(73.90 secs, 16,433,969,976 bytes)
ghci> isomorfos2 (grafoCiclo 100) (grafoRueda 100)
False
(0.00 secs, 0 bytes)
ghci> isomorfos3 (grafoCiclo 100) (grafoRueda 100)
False
(0.00 secs, 0 bytes)

ghci> length (isomorfismos2 (grafoRueda 10) (grafoRueda 10))
18
(101.12 secs, 23,237,139,992 bytes)
ghci> length (isomorfismos3 (grafoRueda 10) (grafoRueda 10))
18
(44.67 secs, 9,021,442,440 bytes)

```

Nota 4.6.3. Cuando los grafos son isomorfos, comprobar que tienen el mismo núme-

ro de vértices, el mismo número de aristas y la misma secuencia gráfica no requiere mucho tiempo ni espacio, dando lugar a costes muy similares entre los dos pares de definiciones. Sin embargo, cuando los grafos tienen el mismo número de vértices y fallan en alguna de las demás propiedades, el resultado es muy costoso para la primera definición mientras que es inmediato con la segunda.

A partir de ahora utilizaremos la función `isomorfismos2` para calcular los isomorfismos entre dos grafos y la función `isomorfos` para determinar si dos grafos son isomorfos, de modo que las renombraremos por `isomorfismos` y `isomorfismos` respectivamente.

```
isomorfismos :: (Ord a, Ord b) => Grafo a -> Grafo b -> [Funcion a b]
isomorfismos = isomorfismos2

isomorfos :: (Ord a, Ord b) => Grafo a -> Grafo b -> Bool
isomorfos =
```

4.6.3. Automorfismos

Definición 4.6.9. Dado un grafo simple $G = (V, A)$, un **automorfismo** de G es un isomorfismo de G en sí mismo.

La función `(esAutomorfismo g f)` se verifica si la aplicación f es un automorfismo de g .

```
-- | Ejemplos
-- >>> let g = creaGrafo [1,2,3] [(1,2),(1,3)]
-- >>> esAutomorfismo g [(1,1),(2,3),(3,2)]
-- True
-- >>> esAutomorfismo g [(1,2),(2,3),(3,1)]
-- False
esAutomorfismo :: Ord a => Grafo a -> Funcion a a -> Bool
esAutomorfismo g = esIsomorfismo g g
```

La función `(automorfismos g)` devuelve la lista de todos los posibles automorfismos en g .

```
-- | Ejemplo
-- >>> automorfismos (creaGrafo [1,2,3] [(1,2),(1,3)])
-- [[(1,1),(2,2),(3,3)],[(1,1),(2,3),(3,2)]]
automorfismos :: Ord a => Grafo a -> [Funcion a a]
automorfismos g = isomorfismos1 g g
```

Nota 4.6.4. Cuando trabajamos con automorfismos, es mejor utilizar en su definición la función `isomorfismos1` en vez de `isomorfismos2`, pues ser isomorfos es una relación reflexiva; es decir, un grafo siempre es isomorfo a sí mismo.

4.7. Conectividad de grafos

Una de las aplicaciones de la teoría de grafos es la determinación de trayectos o recorridos en una red de transporte o de distribución de productos. Así, si cada vértice representa un punto de interés y cada arista representa una conexión entre dos puntos, usando grafos como modelos, podemos simplificar el problema de encontrar la ruta más ventajosa en cada caso.

4.7.1. Caminos

Definición 4.7.1. Sea $G = (V, A)$ un grafo simple y sean $u, v \in V$ dos vértices. Un **camino** entre u y v es una sucesión de vértices de G : $u = v_0, v_1, v_2, \dots, v_{k-1}, v_k = v$ donde $\forall i \in \{0, \dots, k-1\}, (v_i, v_{i+1}) \in A$.

La función `(esCamino g c)` se verifica si la sucesión de vértices `c` es un camino en el grafo `g`.

```
-- | Ejemplo
-- >>> grafoCiclo 5
-- G [(1,2),(1,5),(2,3),(3,4),(4,5)]
-- >>> esCamino (grafoCiclo 5) [1,2,3,4,5,1]
-- True
-- >>> esCamino (grafoCiclo 5) [1,2,4,5,3,1]
-- False
esCamino :: Ord a => Grafo a -> [a] -> Bool
esCamino g c = all ('aristaEn' g) (zip c (tail c))
```

La función `(aristasCamino c)` devuelve la lista de las aristas recorridas en el camino `c`.

```
-- | Ejemplos
-- >>> aristasCamino [1,2,3]
-- [(1,2),(2,3)]
-- >>> aristasCamino [1,2,3,1]
-- [(1,2),(2,3),(1,3)]
-- >>> aristasCamino [1,2,3,2]
-- [(1,2),(2,3),(2,3)]
aristasCamino :: Ord a => [a] -> [(a,a)]
aristasCamino c =
  map parOrdenado (zip c (tail c))
  where parOrdenado (u,v) | u <= v    = (u,v)
                        | otherwise = (v,u)
```

La función `(verticesCamino c)` devuelve la lista de las vertices recorridas en el camino `c`.

```
-- | Ejemplo
-- >>> verticesCamino [1,2,3,1]
-- [1,2,3]
verticesCamino :: Ord a => [a] -> [a]
verticesCamino c = nub c
```

Definición 4.7.2. Sea $G = (V, A)$ un grafo y sean $u, v \in V$. Un camino entre u y v que no repite aristas (quizás vértices) se llama **recorrido**.

La función `(esRecorrido g c)` se verifica si el camino c es un recorrido en el grafo g .

```
-- | Ejemplo
-- >>> esRecorrido (grafoCiclo 4) [2,1,4]
-- True
-- >>> esRecorrido (grafoCiclo 4) [2,1,4,1]
-- False
-- >>> esRecorrido (grafoCiclo 4) [2,1,3]
-- False
esRecorrido :: Ord a => Grafo a -> [a] -> Bool
esRecorrido g c =
  esCamino g c &&
  aristasCamino c == nub (aristasCamino c)
```

Comentario 7: Añadir a la sección de conjuntos la función `(sinRepetidos xs)` que se verifique si xs no tiene elementos repetidos y usarla para definir `esRecorrido`.

Definición 4.7.3. Un camino que no repite vértices (y, por tanto, tampoco aristas) se llama **camino simple**.

La función `(esCaminoSimple g c)` se verifica si el camino c es un camino simple.

```
-- | Ejemplos
-- >>> grafoCiclo 4
-- G [1,2,3,4] [(1,2),(1,4),(2,3),(3,4)]
-- >>> esCaminoSimple (grafoCiclo 4) [2,1,4]
-- True
-- >>> esCaminoSimple (grafoCiclo 4) [1,4,3,2,1]
-- True
-- >>> esCaminoSimple (grafoCiclo 4) [4,3,2,1,2]
-- False
esCaminoSimple :: Ord a => Grafo a -> [a] -> Bool
esCaminoSimple g (v:vs) =
  esRecorrido g (v:vs) &&
  verticesCamino vs == vs
```

Comentario 8: Corregir la definición de `esCaminoSimple` para evitar contraejemplos como el siguiente (`esCaminoSimple (creaGrafo [1,2] [(1,1),(1,2)]) [1,1,2] == True`).

Comentario 9: Simplificar la definición de `esCaminoSimple` usando `sinRepetidos`.

Definición 4.7.4. Se llama *longitud* de un camino al número de veces que se atraviesa una arista en dicho camino.

La función (`longitudCamino c`) devuelve la longitud del camino `c`.

```
-- | Ejemplo
-- >>> longitudCamino [4,2,7]
-- 2
longitudCamino :: [a] -> Int
longitudCamino c = length c - 1
```

La función (`todosCaminoBP g u v`) devuelve una lista con todos los caminos simples posibles en el grafo `g` entre los vértices `u` y `v`, utilizando una algoritmo de búsqueda en profundidad sobre el grafo `g`. Este algoritmo recorre el grafo de izquierda a derecha y de forma al visitar un nodo, explora todos los caminos que pueden continuar por él antes de pasar al siguiente.

```
-- | Ejemplos
-- >>> grafoCiclo 4
-- G [1,2,3,4] [(1,2),(1,4),(2,3),(3,4)]
-- >>> todosCaminoBP (grafoCiclo 4) 1 4
-- [[1,4],[1,2,3,4]]
-- >>> todosCaminoBP (grafoCiclo 4) 4 1
-- [[4,3,2,1],[4,1]]
-- >>> todosCaminoBP (creaGrafo [1..4] [(1,2),(3,4)]) 1 4
-- []
-- >>> todosCaminoBP (creaGrafo [1,2] [(1,1),(1,2)]) 1 1
-- [[1]]
todosCaminoBP :: Ord a => Grafo a -> a -> a -> [[a]]
todosCaminoBP g x y = aux [[y]]
  where aux [] = []
        aux ([]:zss) = aux zss
        aux ((z:zs):zss)
          | z == x = (z:zs) : aux zss
          | otherwise = aux ([v:zs | v <- adyacentes g' z \\\ zs] ++ zss)
        g' = eliminaLazos g
        eliminaLazos h = creaGrafo (vertices h)
                               [(x,y) | (x,y) <- aristas h, x /= y]
```


La función (`todosCaminosBA g u v`) devuelve una lista con todos los caminos simples posibles en el grafo `g` entre los vértices `u` y `v`, utilizando un algoritmo de búsqueda en anchura sobre el grafo `g`. Este algoritmo recorre el grafo por niveles, de forma que el primer camino de la lista es de longitud mínima.

```
-- | Ejemplos
-- >>> grafoCiclo 4
-- G [1,2,3,4] [(1,2),(1,4),(2,3),(3,4)]
-- >>> todosCaminosBA (grafoCiclo 4) 1 4
-- [[1,4],[1,2,3,4]]
-- >>> todosCaminosBA (grafoCiclo 4) 4 1
-- [[4,1],[4,3,2,1]]
-- >>> todosCaminosBA (creaGrafo [1..4] [(1,2),(3,4)]) 1 4
-- []
-- >>> todosCaminosBA (creaGrafo [1,2] [(1,1),(1,2)]) 1 1
-- [[1]]
todosCaminosBA :: Ord a => Grafo a -> a -> a -> [[a]]
todosCaminosBA g x y = aux [[y]]
  where aux [] = []
        aux ([]:zss) = aux zss
        aux ((z:zs):zss)
          | z == x = (z:zs) : aux zss
          | otherwise = aux (zss ++ [v:z:zs | v <- adyacentes g' z \\ zs])
        g' = eliminaLazos g
```

Vamos a comprobar con QuickCheck que el primer elemento de la lista que devuelve la función (`todosCaminosBA g u v`) es de longitud mínima. Para ello, vamos a utilizar la función (`parDeVertices g`) que es un generador de pares de vértices del grafo no nulo `g`. Por ejemplo,

```
ghci> sample (parDeVertices (creaGrafo [1..9] []))
(3,9)
(9,3)
(7,4)
(4,3)
(2,8)
(7,2)
(8,4)
(5,3)
(7,2)
(3,1)
(7,2)
```

```
parDeVertices :: Grafo Int -> Gen (Int,Int)
parDeVertices g = do
```

```
x <- elements vs
y <- elements vs
return (x,y)
where vs = vertices g
```

La propiedad es

```
prop_todosCaminosBA :: Grafo Int -> Property
prop_todosCaminosBA g =
  not (esGrafoNulo g) ==>
  forAll (parDeVertices g)
    (\(x,y) -> let zss = todosCaminosBA g x y
                 in null zss || longitudCamino (head zss) ==
                    minimum (map longitudCamino zss))
```

La comprobación es

```
ghci> quickCheck prop_todosCaminosBA
+++ OK, passed 100 tests:
```

Veamos que la función (todosCaminosBP g u v) no verifica la propiedad.

```
ghci> quickCheck prop_todosCaminosBP
*** Failed! Falsifiable (after 6 tests):
G [1,2,3,4,5] [(1,2),(1,4),(1,5),(2,2),(2,3),(2,4),(4,4),(4,5),(5,5)]
(5,2)
```

```
prop_todosCaminosBP :: Grafo Int -> Property
prop_todosCaminosBP g =
  not (esGrafoNulo g) ==>
  forAll (parDeVertices g)
    (\(x,y) -> let zss = todosCaminosBP g x y
                 in null zss || longitudCamino (head zss) ==
                    minimum (map longitudCamino zss))
```

Definición 4.7.5. Dado un grafo $G = (V, A)$, sean $u, v \in V$. Si existe algún camino entre u y v en el grafo G diremos que están **conectados** y lo denotamos por $u \sim v$.

La función (estanConectados g u v) se verifica si los vértices u y v están conectados en el grafo g .

```
-- | Ejemplos
-- >>> estanConectados (creaGrafo [1..4] [(1,2),(2,4)]) 1 4
-- True
-- >>> estanConectados (creaGrafo [1..4] [(1,2),(3,4)]) 1 4
```

```
-- False
estanConectados :: Ord a => Grafo a -> a -> a -> Bool
estanConectados g u v
  | esGrafoNulo g = False
  | otherwise     = not (null (todosCaminosBA g u v))
```

Nota 4.7.1. La función `(estanConectados g u v)` no necesita calcular todos los caminos entre u y v . Puesto que Haskell utiliza por defecto evaluación perezosa, si existe algún camino entre los dos vértices, basta calcular el primero para saber que la lista de todos los caminos no es vacía

Definición 4.7.6. Se define la *distancia* entre u y v en el grafo G como la longitud del camino más corto que los une. Si u y v no están conectados, decimos que la distancia es infinita.

La función `(distancia g u v)` devuelve la distancia entre los vértices u y v en el grafo g . En caso de que los vértices no estén conectados devuelve el valor `Nothing`.

```
-- | Ejemplos
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 1
-- Just 0
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 2
-- Just 1
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 3
-- Just 2
-- >>> distancia (creaGrafo [1..4] [(1,2),(2,3)]) 1 4
-- Nothing
distancia :: Ord a => Grafo a -> a -> a -> Maybe Int
distancia g u v
  | estanConectados g u v =
      Just (longitudCamino (head (todosCaminosBA g u v)))
  | otherwise = Nothing
```

Definición 4.7.7. Dado $G = (V, A)$ un grafo, sean $u, v \in V$. Un camino entre u y v cuya longitud coincide con la distancia entre los vértices se llama *geodésica* entre u y v .

La función `(esGeodesica g c)` se verifica si el camino c es una geodésica entre sus extremos en el grafo g .

```
-- | Ejemplos
-- >>> let g = creaGrafo [1..4] [(1,2),(1,3),(2,3),(3,4)]
-- >>> esGeodesica g [1,3,4]
-- True
-- >>> esGeodesica g [1,2,3,4]
-- False
esGeodesica :: Ord a => Grafo a -> [a] -> Bool
```

```

esGeodesica g c =
  esRecorrido g c &&
  longitudCamino c == fromJust (distancia g u v)
  where u = head c
        v = last c

```

Comentario 10: Comparar la definición de `esGeodesica` con la siguiente: ¿son equivalentes?, ¿cuál es más cercana a la definición matemática?, ¿cuál es más corta?

```

esGeodesica2 :: Ord a => Grafo a -> [a] -> Bool
esGeodesica2 g c =
  esCamino g c &&
  longitudCamino c == fromJust (distancia g (head c) (last c))

```

Definición 4.7.8. Un camino en un grafo G se dice **cerrado** si sus extremos son iguales.

La función `(esCerrado g c)` se verifica si el camino c es cerrado en el grafo g .

```

-- | Ejemplos
-- >>> let g = creaGrafo [1..4] [(1,2),(1,3),(2,3),(3,4)]
-- >>> esCerrado g [1,2,3,1]
-- True
-- >>> esCerrado g [1,2,3]
-- False
-- >>> esCerrado g [1,2,4,1]
-- False
esCerrado :: (Ord a) => Grafo a -> [a] -> Bool
esCerrado g (v:c) =
  esCamino g c && v == last c

```

Comentario 11: En la definición de `esCerrado` ¿qué pasa con el camino vacío?

Definición 4.7.9. Un recorrido en un grafo G se dice **circuito** si sus extremos son iguales.

La función `(esCircuito g c)` se verifica si la sucesión de vértices c es un circuito en el grafo g .

```

-- | Ejemplos
-- >>> grafoCiclo 4
-- G [1,2,3,4] [(1,2),(1,4),(2,3),(3,4)]
-- >>> esCircuito (grafoCiclo 4) [1,2,3,4,1]
-- True
-- >>> esCircuito (grafoCiclo 4) [1,2,3,4]
-- False

```

```
-- >>> esCircuito (grafoCiclo 4) [1,2,3,4,1,4,1]
-- False
esCircuito :: (Ord a) => Grafo a -> [a] -> Bool
esCircuito g c =
  esRecorrido g c && esCerrado g c
```

Definición 4.7.10. *Un camino simple en un grafo G se dice que es un **ciclo** si sus extremos son iguales.*

La función (`esCiclo g c`) se verifica si el camino `c` es un ciclo en el grafo `g`.

```
-- | Ejemplos
-- >>> esCiclo (grafoCiclo 4) [1,2,1]
-- False
-- >>> esCiclo (grafoCiclo 4) [1,2,3,4,1]
-- True
-- >>> esCiclo (grafoCiclo 4) [1,2,3,4]
-- False
-- >>> esCiclo (grafoCiclo 4) [1,2,3,4,1,4,1]
-- False
esCiclo :: (Ord a) => Grafo a -> [a] -> Bool
esCiclo g c =
  esCaminoSimple g c && esCerrado g c
```

La función (`todosCiclos g v`) devuelve todos los ciclos en el grafo `g` que pasan por el vértice `v`.

```
-- | Ejemplos
-- >>> todosCiclos (grafoCiclo 4) 3
-- [[3,4,1,2,3],[3,2,1,4,3]]
-- >>> todosCiclos (completo 3) 2
-- [[2,3,1,2],[2,1,3,2]]
-- >>> todosCiclos (creaGrafo [1,2,3] [(1,1),(1,2),(1,3),(2,3)]) 1
-- [[1],[1,3,2,1],[1,2,3,1]]
-- >>> todosCiclos (creaGrafo [1,2] [(1,2),(2,2),(2,3)]) 2
-- [[2]]
todosCiclos :: Ord a => Grafo a -> a -> [[a]]
todosCiclos g x = if aristaEn (x,x) g
  then [x] : aux [[x]]
  else aux [[x]]
  where aux [] = []
        aux [v:z] = aux [v:z] | v <- adyacentes g' z \ [x]
        aux ([]:zss) = aux zss
        aux ((z:zs):zss)
          | z == x = (z:zs) : aux zss
          | otherwise = aux (zss ++ [v:z:zs] | v <- adyacentes g' z \ [x])
```

```

                                                    (head zs:(zs \\ [x])))]
g' = eliminaLazos g
eliminaLazos h = creaGrafo (vertices h)
                    [(x,y) | (x,y) <- aristas h, x /= y]

```

Definición 4.7.11. Diremos que un grafo $G = (V, A)$ es **acíclico** si no contiene ningún ciclo; es decir, si $\forall v \in V$ no existe ningún camino simple que comience y termine en v .

La función `(esAciclico g)` se verifica si el grafo g es acíclico.

```

-- | Ejemplo
-- >>> esAciclico (creaGrafo [1..4] [(1,2),(2,4)])
-- True
-- >>> esAciclico (grafoCiclo 5)
-- False
-- >>> esAciclico (grafoEstrella 6)
-- True
esAciclico :: Ord a => Grafo a -> Bool
esAciclico g =
  and [null (todosCiclos g x) | x <- vertices g]

```

Nota 4.7.2. El algoritmo utilizado en la definición de `(todosCiclos g)` es el de búsqueda en anchura. Este algoritmo recorre el grafo por niveles, de forma que el primer camino de la lista es de longitud mínima.

4.7.2. Conectividad

Teorema 4.7.12. Dado un grafo G , la relación $u \sim v$ (estar conectados por un camino) es una relación de equivalencia.

La función `(estarConectadosCamino g)` devuelve la relación entre los vértices del grafo g de estar conectados por un camino en él.

```

-- | Ejemplo
-- >>> estarConectadosCamino (creaGrafo [1..4] [(1,2),(2,4)])
-- [(1,1),(1,2),(1,4),(2,1),(2,2),(2,4),(3,3),(4,1),(4,2),(4,4)]
estarConectadosCamino :: Ord a => Grafo a -> [(a,a)]
estarConectadosCamino g =
  [(u,v) | u <- vs, v <- vs, estanConectados g u v]
  where vs = vertices g

```

A continuación, comprobaremos el resultado con QuickCheck.

```

ghci> quickCheck prop_conectadosRelEqui
+++ OK, passed 100 tests.

```

```
prop_conectadosRelEqui :: Grafo Int -> Bool
prop_conectadosRelEqui g =
  esRelacionEquivalencia (vertices g) (estarConectadosCamino g)
```

Definición 4.7.13. Las clases de equivalencia obtenidas por la relación \sim , estar conectados por un camino en un grafo G , inducen subgrafos en G , los vértices y todas las aristas de los caminos que los conectan, que reciben el nombre de **componentes conexas por caminos** de G .

La función (`componentesConexas g`) devuelve las componentes conexas por caminos del grafo g .

```
-- | Ejemplos
-- >>> componentesConexas (creaGrafo [1..5] [(1,2),(2,3)])
-- [[1,2,3],[4],[5]]
-- >>> componentesConexas (creaGrafo [1..5] [(1,2),(2,3),(4,5)])
-- [[1,2,3],[4,5]]
-- >>> componentesConexas (creaGrafo [1..3] [(1,2),(2,3)])
-- [[1,2,3]]
componentesConexas :: Ord a => Grafo a -> [[a]]
componentesConexas g
  | esGrafoNulo g = []
  | esCompleto g  = [vertices g]
  | otherwise     =
    clasesEquivalencia (vertices g) (estarConectadosCamino g)
```

La función (`numeroComponentes g`) devuelve el número de componentes conexas del grafo g .

```
-- Ejemplos
-- >>> numeroComponentes (creaGrafo [1..5] [(1,2),(2,3),(4,5)])
-- 2
-- >>> numeroComponentes (creaGrafo [1..4] [(1,2),(2,2),(3,4)])
-- 2
numeroComponentes :: Ord a => Grafo a -> Int
numeroComponentes g
  | null (aristas g) = orden g
  | esCompleto g     = 1
  | otherwise        = length (componentesConexas g)
```

Definición 4.7.14. Dado un grafo, diremos que es **conexo** si la relación \sim tiene una única clase de equivalencia en él; es decir, si el grafo tiene una única componente conexas.

La función (`esConexo g`) se verifica si el grafo g es conexo.

```
-- Ejemplos
-- >>> esConexo (creaGrafo [1..3] [(1,2),(2,3)])
-- True
-- >>> esConexo (creaGrafo [1..5] [(1,2),(2,3),(4,5)])
-- False
esConexo :: Ord a => Grafo a -> Bool
esConexo g = length (componentesConexas g) == 1
```

Comentario 12: Buscar definiciones más eficientes de esConexo.

Teorema 4.7.15. Sea G un grafo, $G = (V, A)$ es conexo si y solamente si $\forall u, v \in V$ existe un camino entre u y v .

Vamos a comprobar el resultado con QuickCheck.

```
ghci> quickCheck prop_caracterizaGrafoConexo
+++ OK, passed 100 tests.
```

```
prop_caracterizaGrafoConexo :: Grafo Int -> Property
prop_caracterizaGrafoConexo g =
  not (esGrafoNulo g) ==>
  esConexo g == and [estánConectados g u v
                    | u <- vertices g, v <- vertices g]
```

*Definición 4.7.16. Sean $G = (V, A)$ un grafo y $v \in V$. Se define la **excentricidad** de v como el máximo de las distancias entre v y el resto de vértices de G . La denotaremos por $e(v)$.*

La función (excentricidad g v) devuelve la excentricidad del vértice v en el grafo g.

```
-- | Ejemplos
-- >>> let g = creaGrafo [1..3] [(1,2),(2,3),(3,3)]
-- >>> excentricidad g 1
-- Just 2
-- >>> excentricidad g 2
-- Just 1
-- >>> excentricidad (creaGrafo [1..3] [(1,2),(3,3)]) 1
-- Nothing
excentricidad :: Ord a => Grafo a -> a -> Maybe Int
excentricidad g u
  | esGrafoNulo g           = Nothing
  | orden g == 1            = Just 0
  | Nothing 'elem' distancias = Nothing
  | otherwise               = maximum distancias
  where distancias = [distancia g u v | v <- vertices g \\< [u]]
```


La función (`excentricidades g`) devuelve la lista ordenada de las excentricidades de los vértices del grafo `g`.

```
-- | Ejemplos
-- >>> excentricidades (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- [Just 1,Just 2,Just 2]
-- >>> excentricidad (creaGrafo [1..3] [(1,2),(3,3)])
-- [Nothing,Nothing,Nothing]
excentricidades :: Ord a => Grafo a -> [Maybe Int]
excentricidades g = sort (map (excentricidad g) (vertices g))
```

Definición 4.7.17. Sea $G = (V, A)$ un grafo. Se define el **diámetro** de G como el máximo de las distancias entre los vértices en V . Lo denotaremos por $d(G)$.

La función (`diametro g`) devuelve el diámetro del grafo `g`.

```
-- | Ejemplos
-- >>> diametro (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- Just 2
-- >>> diametro (creaGrafo [1..3] [(1,2),(3,3)])
-- Nothing
diametro :: Ord a => Grafo a -> Maybe Int
diametro g
  | esGrafoNulo g           = Nothing
  | Nothing 'elem' distancias = Nothing
  | otherwise                = maximum distancias
  where vs                  = vertices g
        distancias = [distancia g u v | u <- vs, v <- vs, u <= v]
```

Definición 4.7.18. Sean $G = (V, A)$ un grafo y $v \in V$. Se define el **radio** de G como el mínimo de las excentricidades de sus vértices. Lo denotaremos por $r(G)$.

La función (`radio g`) devuelve el radio del grafo `g`.

```
-- | Ejemplos
-- >>> radio (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- Just 1
-- >>> radio (creaGrafo [1..3] [(1,2),(3,3)])
-- Nothing
radio :: Ord a => Grafo a -> Maybe Int
radio g
  | esGrafoNulo g          = Nothing
  | Nothing 'elem' ds       = Nothing
  | otherwise               = minimum ds
  where ds = [excentricidad g v | v <- vertices g]
```

*Definición 4.7.19. Sean $G = (V, A)$ un grafo. Llamamos **centro** del grafo G al conjunto de vértices de excentricidad mínima. A estos vértices se les denomina **vértices centrales**.*

La función (`centro g`) devuelve el centro del grafo g . Por ejemplo,

```
-- | Ejemplos
-- >>> centro (creaGrafo [1..3] [(1,2),(2,3),(3,3)])
-- [2]
-- >>> centro (creaGrafo [1..3] [(1,2),(1,3),(2,3)])
-- [1,2,3]
-- >>> centro (creaGrafo [1..3] [(1,2),(3,3)])
-- [1,2,3]
centro :: Ord a => Grafo a -> [a]
centro g = [v | v <- vertices g, excentricidad g v == r]
  where r = radio g
```

*Definición 4.7.20. Sean $G = (V, A)$ un grafo. Se llama **grosor** o **cintura** del grafo G al mínimo de las longitudes de los ciclos de G . Si el grafo no posee ciclos (es decir, es un grafo acíclico), se dice que su cintura es infinita.*

La función (`grosor g`) devuelve el grosor del grafo g .

```
-- | Ejemplos
-- >>> grosor (creaGrafo [1,2,3] [(1,2),(2,3)])
-- Nothing
-- >>> grosor (creaGrafo [1,2,3] [(1,1),(1,2),(2,3),(3,4)])
-- Just 0
-- >>> grosor grafoPetersen
-- Just 5
-- >>> grosor grafoMoebiusCantor
-- Just 6
-- >>> grosor grafoHeawood
-- Just 6
-- >>> grosor grafoMcGee
-- Just 7
-- >>> grosor grafoTutteCoxeter
-- Just 8
grosor :: Ord a => Grafo a -> Maybe Int
grosor g
  | esAciclico g = Nothing
  | otherwise    = Just (minimum [longitudCamino (head yss)
                                   | x <- vertices g
                                   , let yss = todosCiclos g x
                                   , not (null yss)])
```

4.7.3. Propiedades del grosor de los grafos

Teorema 4.7.21. El grosor del grafo ciclo de orden n , C_n , es ∞ , si $n < 3$ y 3 en caso contrario.

La propiedad se expresa por

```
prop_grosor_grafoCiclo :: Int -> Bool
prop_grosor_grafoCiclo n =
  grosor (grafoCiclo n) == if n < 3
                           then Nothing
                           else Just n
```

Su comprobación para $n \leq 30$ es

```
ghci> all prop_grosor_grafoCiclo [1..30]
True
```

Teorema 4.7.22. El grosor del grafo amistad de orden n es 3, para todo n .

```
prop_grosor_grafoAmistad :: Int -> Bool
prop_grosor_grafoAmistad n =
  grosor (grafoAmistad n) == Just 3
```

Su comprobación para $n \leq 30$ es

```
ghci> all prop_grosor_grafoAmistad [1..30]
True
```

Teorema 4.7.23. El grosor del grafo completo de orden n , K_n , es ∞ si $n < 3$ y 3 en caso contrario.

La propiedad se expresa por

```
prop_grosor_completo :: Int -> Bool
prop_grosor_completo n =
  grosor (completo n) == if n < 3
                        then Nothing
                        else Just 3
```

Su comprobación para $n \leq 30$ es

```
ghci> all prop_grosor_completo [1..30]
True
```

Teorema 4.7.24. El grosor del grafo bipartito completo de orden, $K_{m,n}$, es ∞ si $m = 1$ ó $n = 1$ y es 4 en caso contrario. y 3 en caso contrario.

La propiedad se expresa por

```
prop_grosor_bipartitoCompleto :: Int -> Int -> Bool
prop_grosor_bipartitoCompleto m n =
  grosor (bipartitoCompleto m n) == if m == 1 || n == 1
                                     then Nothing
                                     else Just 4
```

Su comprobación para $1 \leq m \leq n \leq 15$ es

```
|ghci> and [prop_grosor_bipartitoCompleto m n | m <- [1..15], n <- [m..15]]
|True
```

Teorema 4.7.25. El grosor del grafo rueda de orden n , W_n es ∞ si $n < 3$ y 3 en caso contrario.

La propiedad se expresa por

```
prop_grosor_grafoRueda :: Int -> Bool
prop_grosor_grafoRueda n =
  grosor (grafoRueda n) == if n < 3
                           then Nothing
                           else Just 3
```

Su comprobación para $1 \leq n \leq 30$ es

```
|ghci> all prop_grosor_grafoRueda [1..30]
|True
```

4.7.4. Propiedades e invariantes por isomorfismos

Teorema 4.7.26. Sean $G = (V, A)$ y $G' = (V', A')$ grafos isomorfos con $\phi : V \rightarrow V'$ un isomorfismo. Entonces, dados $u, v \in V$, $u \sim v$ si y solamente si $\phi(u) \sim \phi(v)$.

La comprobación del teorema con QuickCheck es:

```
|ghci> quickCheckWith stdArgs maxDiscardRatio = 20 prop_ConexionIsomorfismo1
|+++ OK, passed 100 tests.
```

```
prop_ConexionIsomorfismo1 :: Grafo Int -> Grafo Int -> Property
prop_ConexionIsomorfismo1 g h =
  isomorfos g h ==>
  and [ec g u v == ec h (imagen phi u) (imagen phi v)
       | u <- vs, v <- vs, phi <- isomorfismos g h]
  where vs = vertices g
        ec = estanConectados
```

Teorema 4.7.27. Sean $G = (V, A)$ y $G' = (V', A')$ grafos isomorfos con $\phi : V \rightarrow V'$ un isomorfismo. Entonces, ϕ lleva cada componente conexa de G en una componente conexa de G' .

La comprobación del teorema con QuickCheck es:

```
ghci> quickCheckWith stdArgs maxDiscardRatio = 15 prop_ConexionIsomorfismo2
+++ OK, passed 100 tests.
```

```
prop_ConexionIsomorfismo2 :: Grafo Int -> Grafo Int -> Property
prop_ConexionIsomorfismo2 g h =
  isomorfos g h ==>
  and [conjuntosIguales cch (aux f) | f <- isomorfismos g h]
    where cch = componentesConexas h
          ccg = componentesConexas g
          aux f = map (sort . imagenConjunto f) ccg
```

Teorema 4.7.28. Sean $G = (V, A)$ y $G' = (V', A')$ grafos isomorfos con $\phi : V \rightarrow V'$ un isomorfismo. Entonces, G y G' tienen el mismo número de componentes conexas.

La comprobación del teorema con QuickCheck es:

```
ghci> quickCheckWith stdArgs maxDiscardRatio = 15 prop_ConexionIsomorfismo3
+++ OK, passed 100 tests.
```

```
prop_ConexionIsomorfismo3 :: Grafo Int -> Grafo Int -> Property
prop_ConexionIsomorfismo3 g h =
  isomorfos g h ==>
  numeroComponentes g == numeroComponentes h
```

Teorema 4.7.29. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que el diámetro de G es igual al diámetro de G' ; es decir, el diámetro de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
ghci> quickCheck (esInvariantePorIsomorfismos diametro)
+++ OK, passed 100 tests.
```

Teorema 4.7.30. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que el radio de G es igual al radio de G' ; es decir, el radio de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
|ghci> quickCheck (esInvariantePorIsomorfismos radio)
|+++ OK, passed 100 tests.
```

Teorema 4.7.31. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que el grosor de G es igual al grosor de G' ; es decir, el grosor de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
|ghci> quickCheck (esInvariantePorIsomorfismos grosor)
|+++ OK, passed 100 tests.
```

Teorema 4.7.32. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que el centro de G es igual al centro de G' ; es decir, el centro de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
|ghci> quickCheck (esInvariantePorIsomorfismos centro)
|+++ OK, passed 100 tests.
```

Teorema 4.7.33. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que el número de componentes conexas de G es igual al número de las de G' ; es decir, el número de componentes conexas de un grafo es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
|ghci> quickCheck (esInvariantePorIsomorfismos numeroComponentes)
|+++ OK, passed 100 tests.
```

Teorema 4.7.34. Sean $G = (V, A)$ y $G' = (V', A')$ dos grafos y $\phi : V \rightarrow V'$ un isomorfismo. Entonces, se verifica que los valores que toman las excentricidades de los vértices G es igual al de los vértices de G' ; es decir, el conjunto de valores que toman las excentricidades de los vértices es un invariante por isomorfismos.

Vamos a comprobar el teorema anterior con QuickCheck.

```
|ghci> quickCheck (esInvariantePorIsomorfismos excentricidades)
|+++ OK, passed 100 tests.
```

4.8. Sistemas utilizados

El desarrollo de mi Trabajo de Fin de Grado requería de una infraestructura técnica que he tenido que trabajar antes de comenzar a desarrollar el contenido. A continuación, voy a nombrar y comentar los sistemas y paquetes que he utilizado a lo largo del proyecto.

- **Ubuntu como sistema operativo.** El primer paso fue instalar *Ubuntu* en mi ordenador portátil personal. Para ello, seguí las recomendaciones de mi compañero Eduardo Paluzo, que ya lo había hecho antes.

Primero, me descargué la imagen del sistema *Ubuntu 16.04 LTS* (para procesador de 64 bits) desde la [página de descargas de Ubuntu](http://www.ubuntu.com/download/desktop)¹⁷ y también la herramienta [Linux-Live USB Creator](http://www.linuxliveusb.com/)¹⁸ que transformaría mi Pendrive en una unidad USB Booteable cargada con la imagen de Ubuntu. Una vez tuve la unidad USB preparada, procedí a instalar el nuevo sistema: apagué el dispositivo y al encenderlo entré en el Boot Menu de la BIOS del portátil para arrancar desde el Pendrive en vez de hacerlo desde el disco duro. Automáticamente, comenzó la instalación de *Ubuntu* y solo tuve que seguir las instrucciones del asistente para montar Ubuntu manteniendo además *Windows 10*, que era el sistema operativo con el que había estado trabajando hasta ese momento.

El resultado fue un poco agridulce, pues la instalación de Ubuntu se había realizado con éxito, sin embargo, al intentar arrancar *Windows* desde la nueva GRUB, me daba un error al cargar la imagen del sistema. Después de buscar el error que me aparecía en varios foros, encontré una solución a mi problema: deshabilité el Security Boot desde la BIOS y pude volver a arrancar *Windows 10* con normalidad.

- **L^AT_EX como sistema de composición de textos.** La distribución de L^AT_EX, *Tex Live*, como la mayoría de software que he utilizado, la descargué utilizando el *Gestor de Paquetes Synaptic*. Anteriormente, sólo había utilizado *TexMaker* como editor de L^AT_EX así que fue el primero que descargué. Más tarde, mi tutor José Antonio me sugirió que mejor descargara el paquete *AUCTex*, pues me permitiría trabajar con archivos T_EX desde el editor *Emacs*, así lo hice y es el que he utilizado para escribir el trabajo. Además de los que me recomendaba el gestor, solo he tenido que descargarme el paquete *spanish* de *babel* para poder desarrollar el Trabajo, pues el paquete *Tikz*, que he utilizado para representar los grafos, venía incluido en las sugerencias de *Synaptic*.
- **Haskell como lenguaje de programación.** Ya había trabajado anteriormente con este lenguaje en el Grado y sabía que sólo tenía que decargarme la plataforma de *Haskell* y podría trabajar con el editor *Emacs*. Seguí las indicaciones que se dan a los

¹⁷<http://www.ubuntu.com/download/desktop>

¹⁸<http://www.linuxliveusb.com/>

estudiantes de primer curso en la [página del Dpto. de Ciencias de la Computación e Inteligencia Artificial](#)¹⁹ y me descargué los paquetes *haskell-platform* y *haskell-mode* desde el *Gestor de Paquetes Synaptic*.

Comentario 13: Indicar la versión de la plataforma y de GHC que estás utilizando.

- **Dropbox como sistema de almacenamiento compartido.** Ya había trabajado con *Dropbox* en el pasado, así que crear una carpeta compartida con mis tutores no fue ningún problema; sin embargo, al estar *Dropbox* sujeto a software no libre, no me resultó tan sencillo instalarlo en mi nuevo sistema. En primer lugar, intenté hacerlo directamente desde *Ubuntu Software*, que intentó instalar *Dropbox Nautilus* y abrió dos instalaciones en paralelo. Se quedó colgado el ordenador, así que maté los procesos de instalación activos, reinicié el sistema y me descargué directamente el paquete de instalación desde la [página de descargas de Dropbox](#)²⁰ para ejecutarlo desde la terminal.

Comentario 14: Queda pendiente ampliar el apéndice conforme vayamos avanzando y surjan nuevos sistemas.

4.9. Mapa de decisiones de diseño

Al comienzo del proyecto, la idea era que las primeras representaciones con las que trabajara fueran las de *grafos como vectores de adyacencia* y *grafos como matrices de adyacencia* que se utilizan en Informática en el primer curso del Grado, con las que ya había trabajado y estaba familiarizada.

Las definiciones de Informática están pensadas para grafos ponderados (dirigidos o no según se eligiera), mientras que en Matemática Discreta apenas se usan grafos dirigidos o ponderados; por tanto, el primer cambio en la representación utilizada fue simplificar las definiciones de modo que solo trabajáramos con grafos no dirigidos y no ponderados, pero manteniendo las estructuras vectorial y matricial que mantenían la eficiencia.

Las representaciones que utilizan *arrays* en *Haskell* son muy restrictivas, pues solo admiten vectores y matrices que se puedan indexar, lo que hace muy complicados todos los algoritmos que impliquen algún cambio en los vértices de los grafos y, además, no permite trabajar con todos los tipos de vértices que pudiéramos desear. Decidimos volver a cambiar la representación, y esta vez nos decantamos por la representación de *grafos como listas de aristas*, perdiendo en eficiencia pero ganando mucho en flexibilidad de escritura.

¹⁹<http://www.cs.us.es/~jalonso/cursos/i1m-15/sistemas.php>

²⁰<https://www.dropbox.com/es/install?os=linux>

Comentario 15: Queda pendiente ampliar el apéndice conforme vayamos avanzando y surjan nuevas representaciones.

Bibliografía

- [1] J. Alonso. [Temas de programación funcional](#). Technical report, Univ. de Sevilla, 2015.
- [2] K. Claessen and J. Hughes. [QuickCheck: A Lightweight Tool for Random Testing of Haskell Programs](#). Technical report, Chalmers University of Technology, 2000.
- [3] M. Cárdenas. [Matemática discreta](#). Technical report, Univ. de Sevilla, 2015.
- [4] D. de Matemáticas. [El regalo de Cantor](#). Technical report, IES Matarraña de Valderrobres, 2009.
- [5] D. de Álgebra. [Álgebra Básica, tema 1: Conjuntos](#). Technical report, Univ. de Sevilla, 2015.
- [6] F. Rabhi and G. Lapalme. *Algorithms: A functional programming approach*. Addison-Wesley Longman Publishing Co., Inc., Boston, MA, USA, 1st edition, 1999.
- [7] P. Vasconcelos. [Property Testing using QuickCheck](#). Technical report, Universidade do Porto, 2016.
- [8] Wikipedia. [Set \(mathematics\)](#). Technical report, Wikipedia, La Enciclopedia libre., 2016.
- [9] Wikipedia. [Anexo:Galería de grafos](#). Technical report, Wikipedia, La Enciclopedia libre., 2016.

Índice alfabético

Grafo, 10, 42
adyacentes, 43
antiImagenRelacion, 29
aristaEn, 43
aristasCamino, 72
aristas, 43
automorfismos, 71
bipartitoCompleto, 48
biyecciones, 37
cardinal, 15, 24
centro, 82
clasesEquivalencia, 33
combinaciones, 18, 25
complementario, 15, 23
completo, 48
componentesConexas, 80
conjuntoVacio, 21
conjuntosIguales, 14, 23
conservaAdyacencia, 64
creaGrafo, 42
diametro, 82
distancia, 76
dominio, 28
eliminaArista, 60
eliminaVertice, 60
elimina, 11
entorno, 55
esAislado, 56
esAntisimetrica, 31
esAutomorfismo, 71
esCamino, 72
esCerrado, 77
esCircuito, 78
esCompleto, 63
esConexo, 81
esFuncional, 29
esFuncion, 34
esGeodesica, 77
esInyectiva, 35
esIsomorfismo, 65
esLazo, 55
esReflexiva, 30
esRegular, 56
esRelacionEquivalencia, 32
esRelacionHomogenea, 30
esRelacionOrden, 33
esRelacion, 27
esSimetrica, 31
esSimple, 57
esSobreyectiva, 36
esSubconjuntoPropio, 14, 23
esSubconjunto, 13, 22
esSubgrafoMax, 58
esSubgrafoPropio, 58
esSubgrafo, 58
esTransitiva, 32
esVacio, 11, 21
estaRelacionado, 30
estanConectados, 76
excentricidad, 81
funciones, 34
generaGrafo, 45
grado, 55
grafoAmistad, 47

grafoCiclo, 46
grafoCirculante, 50
grafoComplementario, 63
grafoEstrella, 49
grafoMoebiusCantor, 53
grafoNulo, 46
grafoPetersenGen, 51
grafoPetersen, 52
grafoRueda, 49
grafoThomson, 52
grosor, 83
imagenInversa, 38
imagenRelacion, 28
imagen, 35
inserta, 10
interseccion, 16, 24
inversa, 37
isomorfismos, 66
isomorfos, 66
longitudCamino, 74
minimoElemento, 11
morfismos, 65
orden, 54
pertenece, 11
productoCartesiano, 17, 25
prop_HavelHakimi, 59
prop_LemaApretonDeManos, 59
prop_completos, 62
prop_ConectadosRelEqui, 80
prop_todosCaminosBA, 76
prop_todosCiclos, 79
radio, 82
rango, 28
secuenciaGrados, 57
secuenciaGrafica, 57
sonIncidentes, 54
sumaArista, 61
sumaGrafos, 62, 63
sumaVertice, 61
tamaño, 54
todosCaminos, 74
unionConjuntos, 16, 24
unitario, 13, 22
vacio, 10
valenciaMax, 56
valenciaMin, 56
variaciones, 18, 26
verticesCamino, 72
vertices, 43
textttparDeVertices, 75
todosCiclos, 78

Lista de tareas pendientes

■ Comentario 1: Ampliar el módulo para definir todas las funciones de Conjuntos.lhs	13
■ Comentario 2: Pendiente de ampliar la introducción conforme se vaya escribiendo los módulos.	41
■ Comentario 3: Añadir la especificación de eliminaLazos.	66
■ Comentario 4: A partir de esta propiedad, se puede dar una definición alternativa de K_n (completo2) y comprobar su equivalencia con la primera (completo).	68
■ Comentario 5: Introducir el problema de decidir si dos grafos son homomorfos.	71
■ Comentario 6: Buscar definiciones más eficientes que isomorfismos3.	75
■ Comentario 7: Añadir a la sección de conjuntos la función (sinRepetidos xs) que se verifique si xs no tiene elementos repetidos y usarla para definir esRecorrido.	79
■ Comentario 8: Corregir la definición de esCaminoSimple para evitar contraejemplos como el siguiente (esCaminoSimple (creaGrafo [1,2] [(1,1),(1,2)]) [1,1,2] == True).	80
■ Comentario 9: Simplificar la definición de esCaminoSimple usando sinRepetidos.	80
■ Comentario 10: Comprobar que con todosCaminosBP no se cumple la propiedad.	82
■ Comentario 11: Comparar la definición de esGeodesica con la siguiente: ¿son equivalentes?, ¿cuál es más cercana a la definición matemática?, ¿cuál es más corta?	83
■ Comentario 12: En la definición de esCerrado ¿qué pasa con el camino vacío?)	84
■ Comentario 13: Buscar definiciones más eficientes de esConexo.	87
■ Comentario 14: Indicar la versión de la plataforma y de GHC que estás utilizando.	94
■ Comentario 15: Queda pendiente ampliar el apéndice conforme vayamos avanzando y surjan nuevos sistemas.	94
■ Comentario 16: Queda pendiente ampliar el apéndice conforme vayamos avanzando y surjan nuevas representaciones.	94